
Введение

Рефал Плюс представляет собой один из диалектов языка про-
граммирования Рефал.

Рефал (алгоритмический язык рекурсивных функций) был
создан Валентином Федоровичем Турчиным в качестве языка,
предназначенного для описания семантики других алгоритмиче-
ских языков [Тур 1966], [Тур 1971]. Впоследствии, после появле-
ния достаточно эффективных реализаций [КРТ 1972], [БзР 1977],
[Ром 1987а] Рефал нашел применения в таких областях как ком-
пьютерная алгебра, конструирование компиляторов и интерпре-
таторов, искусственный интеллект и др.

Основным типом данных в Рефале являются объектные выра-
жения, которые представляют собой произвольные деревья, изоб-
ражаемые в линейной записи как последовательности символов
и скобок, правильно построенные относительно скобок. Символы
представляют собой элементарные объекты (такие как литеры,
слова, числа и ссылки на объекты).

Основным средством для анализа структуры объектных выра-
жений и для доступа к их компонентам является сопоставление
с образцом. Образцы в Рефале могут содержать символы, скоб-
ки и переменные. Если объектное выражение имеет структуру,
соответствующую образцу, переменные, входящие в образец, по-
лучают в качестве значений фрагменты объектного выражения.
Значения переменных впоследствии могут использоваться для по-
строения новых объектных выражений.

Рефал-программа представляет собой набор определений
функций. Каждая функция получает в качестве аргумента неко-

торое объектное выражение и вырабатывает в качестве результата
некоторое объектное выражение. Функции могут произвольным
образом вызывать друг друга. Основным средством организации
управления в программах является рекурсия, т. е. такая организа-
ция вызовов функций, при которой некоторые функции вызывают
сами себя (либо непосредственно, либо через другие функции).

Рефал Плюс возник в результате обобщения опыта, накоп-
ленного при разработке, реализации и использовании Базисного
Рефала [БзР 1977], Рефала-2 [КлР 1986], [КлР 1987], [Ром 1987а],
Рефала-4 [Ром 1987б], [Ром 1987в], Рефала-5 [Тур 1989], и языков
FLAC [Кис 1987] и RL [Ром 1987г], [Ром 1988а], [Ром 1988б].

По сравнению с другими диалектами Рефала, Рефал Плюс об-
ладает следующими особенностями.

• Более развитая система модульности.

Каждый модуль разделен на две части: интерфейс модуля и
реализацию модуля. Интерфейс модуля содержит ту часть мо-
дуля, которая “видна” из других модулей, а реализация модуля
содержит те части, которые скрыты от других модулей.

Интерфейс модуля может содержать любые объявления, что
дает возможность экспортировать из модуля не только объявле-
ния функций, но и объявления констант, каналов ввода-вывода,
ящиков и таблиц. При экспорте функций экспортируются не толь-
ко их имена, но и описания форматов их аргументов и результа-
тов, что дает возможность проверять корректность их вызовов
еще на стадии компиляции.

При компиляции модуля используются только интерфейсы
других модулей, но не их реализации, что позволяет проводить
компиляцию модуля даже в том случае, если еще отсутствуют ре-
ализации других модулей. Кроме того, внесение изменений в ре-
ализацию модуля не требует перекомпиляции зависящих от него
модулей. Таким образом, допускаются любые (в том числе и цик-
лические) связи между модулями.

• Статические объявления динамических объектов.

Все объекты, которые могут порождаться динамически, во
время исполнения программы (каналы ввода-вывода, ящики, век-
торы и таблицы) могут объявляться и статически. В этом случае
им присваиваются символические имена, которые используются в
дальнейшем в тексте программы.

• Объявления функций

Каждая функция в Рефале Плюс может быть объявлена от-
катной или безоткатной. Если функция — откатная, она может
выдавать в качестве результата особое значение “неуспех”. Напри-
мер, все функции-предикаты выдают в качестве результата либо
пустое выражение, либо “неуспех”. Если же функция — безоткат-
ная, она никогда не выдает “неуспех”.

Предыдущие диалекты Рефала позволяли определять только
безоткатные функции.

Другая особенность Рефала Плюс — возможность опреде-
лять функции, получающие несколько аргументов и выдающие
несколько результатов. Количество и типы аргументов функции
мы будем называть ее арностью (arity), а количество и типы ре-
зультатов — ее коарностью (co-arity).

Арность и коарность функции определяются заданием ее вход-
ного и выходного формата. Эти форматы представляют собой об-
разцы, которые содержат символы, скобки и переменные. Вход-
ной формат налагает синтаксические ограничения на внешний
вид вызовов функции, а выходной формат — на контексты, в ко-
торых может стоять вызов функции. Если из входного формата
удалить символы и скобки, получаем последовательность пере-
менных, описывающих арность функции. Если же то же самое
сделать с выходным форматом функции, получаем ее коарность.

Явное объявление форматов функций позволяет обнаружи-
вать многие ошибки еще на стадии компиляции и дает возмож-
ность уменьшить накладные расходы при исполнении вызовов
функций.

В предыдущих диалектах Рефала считалось, что каждая
функция получает ровно один аргумент и выдает ровно один ре-
зультат.

• Перехват неуспехов и аварий.

Вычисление любой конструкции в Рефале Плюс может завер-
шаться либо успехом, либо неуспехом, либо аварией.

Если вычисление закончилось успехом, результатом являет-
ся некоторое объектное выражение (которое всегда соответствует
некоторому заранее ожидаемому формату результата и в реализа-
ции может фактически представляться несколькими выражения-
ми).

Если вычисление закончилось неуспехом, результатом явля-
ется сигнал неуспеха и, может быть, некоторая дополнительная
информация о месте, в котором он возник.

Если вычисление закончилось аварией, результатом является
сигнал аварии и сообщение об ошибке — некоторое объектное вы-
ражение.

Рефал Плюс содержит конструкции, которые позволяют пере-
хватывать и анализировать неуспехи и аварии.

• Ввод-вывод объектных выражений.

Рефал Плюс предоставляет набор функций, позволяющих вы-
водить и вводить цепочки литер, а также изображения объектных
выражений. При этом происходит автоматическое превращение
литер слов и чисел в их изображения и обратно.

• Работа с ящиками, векторами и таблицами.

Рефал Плюс дает возможность работать с динамически созда-
ваемыми объектами: ящиками, векторами, строками и таблицами.
Средства работы с ящиками те же, что и в Рефале-2, в то время
как векторы, строки и таблицы являются особенностью Рефала
Плюс.

Каждый ящик представляет собой объект, который может со-
держать произвольное объектное выражение.

Каждый вектор представляет собой объект, который может со-
держать произвольную конечную последовательность объектных
выражений.

Каждая строка представляет собой объект, который может со-
держать произвольную конечную последовательность литер.

Доступ к ящикам, векторам и строкам производится через
символы-ссылки, указывающие на эти объекты. Имеются функ-
ции для создания новых ящиков, векторов и строк и для извлече-
ния и изменения их содержимого. Возможно выборочное извле-
чение и изменение отдельных компонент векторов и строк.

Каждая таблица представляет собой объект, который содер-
жит конечный набор ключей, с каждым из которых связано его
значение. И ключи, и таблицы являются некоторыми объектны-
ми выражениями. Доступ к таблице производится через символы-
ссылки, указывающие на эту таблицу. Имеются функции для со-
здания и копирования таблиц, для получения по ключу его зна-
чения и для получения всех ключей, содержащихся в таблице.
По существу, каждая таблица представляет собой изображение
функции с конечной областью определения.

• “Векторное” представление объектных выражений.

Реализация Рефала Плюс основана на новом, “векторном”
представлении объектных выражений [АбР 1988], позволяющем
свести копирование выражения к копированию пары указателей
на его концы.

Дешевизна операции копирования позволяет применять функ-
циональный стиль программирования на Рефале, в то время как
при использовании предыдущих реализаций приходилось тща-
тельно избегать копирования, что вынуждало, по существу, при-
держиваться императивного стиля.

Освобождение памяти, занятой уже ненужными выражения-
ми, производится с помощью сборки мусора.

Глава 1

Программирование на
Рефале Плюс

В данной главе мы даем неформальный обзор основных изоб-
разительных средств языка Рефал Плюс, приводим примеры про-
грамм и обсуждаем некоторые приемы программирования на Ре-
фале Плюс. Полное описание Рефала Плюс содержится в главе
II (“Синтаксис и семантика Рефала Плюс”), куда и следует об-
ращаться за недостающими тонкостями и подробностями. Если в
примерах программ встречаются вызовы неизвестных функций,
следует обращаться за их описанием к главе III (“Библиотека
функций”) и к разделу “Алфавитный указатель функций”.

1.1 Первый пример

По сложившейся традиции, начнем с рассмотрения простей-
шей программы на Рефале Плюс:

$use StdIO; // Импорт функций ввода-вывода
// из модуля StdIO

$func Main = e; // Объявление формата функции:
// на входе - пусто,
// на выходе - что угодно.

Main // Определение главной функции

= <PrintLn "Hello!">; // Печать и перевод строки

Эта программа состоит из трех директив. Первая директива

$use StdIO;

говорит о том, что в программе будут использоваться библиотеч-
ные функции ввода-вывода, которые должны быть импортирова-
ны из модуля StdIO. Вторая директива описывает формат функ-
ции Main: что она может получать на входе, и что она может выда-
вать в качестве результата. Третья директива является определе-
нием функции Main. По соглашению, принятому в Рефале Плюс,
работа программы всегда начинается с вызова функции Main.

Функция Main всегда имеет пустой аргумент. В данной про-
грамме она вызывает библиотечную функцию PrintLn с аргумен-
том "Hello!", в результате чего на стандартное устройство выво-
да выводится цепочка литер

Hello!

а вслед за ней — литера “конец строки”, после чего работа про-
граммы заканчивается.

Литеры // обозначают начало комментария, т. е. все, что сле-
дует за ними до конца строки — игнорируется.

1.2 Структуры данных

1.2.1 Объектные выражения

Данные, обрабатываемые программами, написанными на язы-
ке Рефал Плюс, являются так называемыми объектными выра-
жениями.

Ниже, в качестве примера, приведены 3 объектных выражения

"Петр" "Иванович" 33 "года"
("Маша" 17) ("Клава" 24) ("Эльвира" 6)
("мой" "дом") "имеет" ("большие" ("светлые" "окна"))

Первое, что бросается в глаза, это использование скобок. Если
в этих примерах изменить количество или расположение скобок,
то изменится структура и, видимо, подразумеваемый смысл этих
выражений.

Кроме скобок вышеприведенные выражения содержат симво-
лы. Вот примеры символов:

"John" "Джон" "джоН" "ку-ку" 1988 -99999999999999

В общем случае объектные выражения состоят из символов и
скобок. Всякое объектное выражение — это последовательность
объектных термов (которая может быть пустой). Объектный
терм — это либо символ, либо объектное выражение, заключенное
в круглые скобки “(” и “)”. Таким образом, всякое объектное вы-
ражение — это последовательность символов и скобок, в которой
скобки “правильно расставлены”.

В памяти компьютера выражения хранятся в виде струк-
туризованных объектов, однако при выполнении операций вво-
да/вывода (печать, запись в файл или чтение из файла и т. п.)
они должны быть представлены в виде линейных последователь-
ностей литер (печатных знаков).

Рефал-система обеспечивает ввод и вывод объектных выра-
жений и все необходимые при этом преобразования. При вводе
выражений входной поток литер разбивается на лексемы. Каж-
дая лексема изображает символ или скобку. Лексемы отделяются
друг от друга пробелами (при этом переход на следующую строку
считается эквивалентным пробелу).

Если рядом оказываются две скобки или скобка и символ, про-
белы между ними могут опускаться. Во всех остальных случаях
лексемы обязательно должны быть разделены хотя бы одним про-
белом.

В текстах Рефал-программ могут появляться в виде констант
следующие символы: символы-литеры, символы-слова и символы-
числа.

Символ-литера — это печатный знак. Последовательность
символов-литер изображается в виде последовательности соответ-
ствующих литер, заключенной в апострофы.

Примеры литер:

’A’ ’a’ ’*’ ’2’ ’ ’

Цепочка из трех литер ’A’, ’B’ и ’C’ может быть записана,
например, следующими способами (а также многими другими):

’A’ ’B’ ’C’
’ABC’
’AB’’C’

Символ-слово представляет собой цепочку литер и изобража-
ется в виде этой цепочки литер, заключенной в двойные кавычки.

Если слово начинается с прописной латинской буквы или под-
черкивания, и содержит только латинские буквы, цифры и под-
черкивания, то окружающие его двойные кавычки можно опу-
стить.

Примеры слов:

"Вася"
"Вот-Слово"
"вот-очень-очень-длинное-Слово"
X_25m3s__
"равно?"

Символ-число является целым числом. Изображается в виде
непустой последовательности десятичных цифр, перед которой
может стоять знак “+” или “-”. Например:

237
-99999999999999999999999999999999999999999999999
+13

Разрядность целых чисел произвольна.

1.2.2 Представление данных в виде
объектных выражений

Объектные выражения особенно удобны для представления
символьных (т. е. не чисто числовых) данных.

17



Пусть, например, мы хотим выбрать представление в виде
объектных выражений для алгебраических формул. Нам нужно
уметь представлять константы, переменные и формулы, которые
получаются в результате применения бинарной операции к двум
более простым формулам. Обозначим через dpe объектное выра-
жение, которое является представлением формулы p. Тогда мы
можем считать, что представлением целых чисел являются соот-
ветствующие символы-числа, представлением переменных явля-
ются соответствующие символы-слова, а для бинарных операций
имеют место соотношения

dp+ qe = ("плюс" dpe dqe )
dp− qe = ("минус" dpe dqe )
dp ∗ qe = ("умн" dpe dqe )
dp/qe = ("дел" dpe dqe )
dpqe = ("степ" dpe dqe )

Таким образом, формула

(X + Y 2)− 512

представляется в виде

("минус" ("плюс" X ("степ" Y 2)) 512)

В качестве другого примера рассмотрим представление шах-
матной позиции в виде объектного выражения.

Первым делом нужно обозначить название и цвет каждой
фигуры. Например ("бел" "Король"), ("черн" "пешка"). Затем
нужно указать поле каждой фигуры. Например ("e" 2), ("h" 7).
Теперь мы можем представить всю позицию в виде последова-
тельности объектных термов, каждый из которых содержит имя
и цвет фигуры, а также ее поле. Например

(("бел" "Король") ("g" 5))
(("черн" "Король") ("a" 7))
(("бел" "Пешка") ("c" 6))
(("бел" "Конь") ("g" 1))
(("черн" "Конь") ("a" 8))

18



1.2.3 Объекты и значения

В самом широком смысле под объектами обычно понимают
некоторые сущности, которые существуют во времени, развива-
ются, но при этом остаются сами собой.

Хорошим примером объекта может служить человек, который
рождается, растет, развивается и умирает, все же оставаясь, в
каком-то смысле, тем же самым человеком.

Другой знаменитый пример принадлежит Гераклиту (расцвет
творческих сил которого приходится приблизительно на 504–
501 гг. до н. э.). Гераклит учил, что нельзя дважды войти в одну
и ту же реку, ибо “на входящих в ту же самую реку набегают все
новые и новые воды” [Гер 500]. Таким образом, река тоже может
служить хорошим примером объекта.

Под значениями в широком смысле обычно понимают неко-
торые сущности, которые не меняются, не развиваются и, в этом
смысле, находятся вне времени.

Неизвестно, существуют ли значения в реальной жизни, но
они являются излюбленным предметом изучения в математике.
Примером значения, например, может служить число 25.

Конечно, значение можно считать частным, вырожденным
случаем объекта (а именно, застывшим объектом, не способным
к развитию), но мы все же обычно будем называть “объектами”
только такие объекты, которые не являются значениями.

Иметь дело с объектами труднее, чем со значениями, ибо они
могут изменяться. Поэтому объекты очень часто снабжают име-
нами. Основным свойством имени является то, что оно однознач-
но связано с объектом (однозначно идентифицирует этот объект).
В отличие от самих объектов, имена являются типичными зна-
чениями, поскольку они не меняются из-за того, что изменяется
сам объект. Например, несмотря на то, что состояние реки Волга
непрерывно изменяется, это никак не сказывается на слове “Вол-
га”. Другим примером (полного) имени могут служить паспорт-
ные данные человека: фамилия, имя, отчество, дата и место рож-
дения и т. д.

В рамках языка Рефал термины “объект” и “значение” имеют

более специальный смысл.
Любое значение в языке Рефал представляет собой некоторое

объектное выражение.
Любой объект в языке Рефал представляет собой “контейнер”,

который может содержать объектные выражения и другую ин-
формацию.

Объекты создаются во время компиляции или в процессе ра-
боты Рефал-программы. При создании объекта одновременно с
ним создается символ-ссылка, который мы будем называть име-
нем объекта или ссылкой на этот объект. Основное свойство имени
объекта заключается в том, что оно должно отличаться от всех
других символов-ссылок, существующих в момент создания объ-
екта. Благодаря этому между символами-ссылками и объектами
существует однозначное соответствие: каждому символу-ссылке
соответствует ровно один объект, а равным символам-ссылкам со-
ответствует один и тот же объект.

Наглядно взаимосвязь между именем объекта R, объектом и
содержимым объекта C можно изобразить следующим образом:

R→ C

Рефал Плюс имеет дело с объектами следующих типов.
Объекты-функции содержат скомпилированные определения

функций. Они создаются во время компиляции программы.
Все остальные объекты могут создаваться как статически (т. е.

во время компиляции программы), так и динамически (т. е. в про-
цессе исполнения программы).

Объекты-ящики предназначены для хранения объектных вы-
ражений. Каждый ящик содержит ровно одно объектное выраже-
ние.

Объекты-таблицы предназначены для хранения конечных
множеств упорядоченных пар. Каждая пара содержит два объ-
ектных выражения, первое из которых является ключом, а вто-
рое — значением, связанным с этим ключом. Все ключи в таблице
должны быть попарно различны. Таким образом, каждому ключу
в таблице однозначно соответствует связанное с ним значение.

Объекты-каналы дают возможность выполнять операции вво-
да-вывода.

Объекты-векторы предназначены для хранения конечных по-
следовательностей объектных выражений.

Объекты-строки предназначены для хранения конечных по-
следовательностей литер.

1.2.4 Сборка мусора

В языке Рефал Плюс имеется возможность порождать объ-
екты во время работы программы, но не предусмотрено никаких
специальных средств для явного уничтожения объектов. Таким
образом, может случиться так, что в процессе выполнения Рефал-
программы память будет заполняться все новыми и новыми объ-
ектами, хотя многие из них будут уже и не нужны. Конечно,
теоретически это не создает никаких проблем, однако, поскольку
Рефал-программы исполняются с помощью реальных компьюте-
ров, память которых конечна, во всех реализациях языка Рефал
предусмотрен механизм сборки мусора.

Сборка мусора автоматически запускается каждый раз, когда
исчерпается свободная память. При этом обнаруживаются и уда-
ляются все объекты, которые заведомо не могут повлиять на даль-
нейшие вычисления, а именно те объекты, до которых невозможно
добраться прямо или косвенно, исходя из значений переменных,
имеющихся в этот момент.

На рисунке 1.1 схематически изображены значения перемен-
ных, а также объекты и их содержимое. Белые кружочки изоб-
ражают некоторые элементы выражений, которые не являются
символами-ссылками, черные — символы-ссылки. Для удобства
изложения все объекты на рисунке пронумерованы. Символы-
ссылки обозначены цифрами.

Видно, что по ссылке из значения одной из переменных можно
добраться до объекта 1, и из него — до объекта 4. По ссылке из
значения другой переменной можно непосредственно добраться
до объекта 2 и косвенно (через объект 2) до объектов 4, 5, 6, 3.
Таким образом, нет способа извлечь информацию из объектов 7

и 8. Если в этот момент запустить сборку мусора, то объекты 7
и 8 будут уничтожены. Если теперь убрать ссылку на объект 1
из значения переменной, то станет недоступным и объект 1. Если
же ссылку на объект 1 оставить, но убрать ссылку на объект 2
из значений переменных, то окажутся ненужными все объекты,
кроме 1 и 4.

значения переменныхr b b b b b r b b b b b
b b r 1 r b b r 2 b b r 3

b b b 4 b r r 5 r b r 7 b r 8

b b r b 6

? ?

????
66

?

6
?

6

Рис. 1.1: Объекты и ссылки

1.3 Вычисление и анализ объектных выра-
жений

1.3.1 Результатные выражения

Результатные выражения в языке Рефал являются анало-
гом общеизвестных арифметических выражений. Так, например,
арифметическому выражению X ∗ Y + 3 в Рефале Плюс соответ-
ствует следующее результатное выражение:

<Add <Mult sX sY> 3>

Угловые скобки обозначают вызов функции. Причем первый
символ-слово после левой угловой скобки обозначает имя функ-
ции, а все остальное соответствует аргументам функции. Благода-
ря тому, что аргумент функции всегда явно заключается в угло-
вые “функциональные” скобки, отпадает необходимость использо-
вать круглые скобки для группировки операций. Например, вы-
ражение X ∗ (A+B) на Рефале записывается как

<Mult sX <Add sA sB>>

в то время как выражение X ∗ A + B на Рефале записывается в
виде

<Add <Mult sX sA> sB>

Результатные выражения, как и арифметические выражения
в других языках, используются для получения одних значений из
других. А именно, при вычислении результатного выражения все
входящие в него переменные заменяются на их значения, после
чего вычисляются вызовы всех функций. При этом, если вызовы
функций вложены друг в друга, то сначала вычисляются более
внутренние вызовы.

Ясно, что результатное выражение можно вычислить только в
том случае, если заданы значения для входящих в него перемен-
ных. Информацию о значениях переменных мы будем называть
средой. Запись

{V1 = E1, . . . , Vn = En}

будет изображать среду, в которой переменные V1, . . . , Vn имеют
значения E1, . . . , En соответственно.

Ясно, что запись арифметических выражений в виде резуль-
татных выражений довольно громоздка, однако, как мы увидим
в дальнейшем, у нее есть и определенные преимущества.

Дело в том, что выбор той или другой системы обозначений
определяется природой тех объектов, с которыми мы собираемся
иметь дело, а также набором операций, которые должны приме-
няться к этим объектам.

Желательно выбирать такие обозначения, чтобы те операции,
которые встречаются чаще всего, записывались бы как можно ко-
роче. Ясно, что самым кратким обозначением операции является
отсутствие всякого обозначения, т. е. пустое место. В случае ариф-
метических выражений мы имеем две основных операции: сложе-
ние и умножение, и одну из них можно обозначить пустотой. Так
обычно и поступают, опуская знак умножения.

В случае языка Рефал, основным объектом работы являют-
ся не числа, а объектные выражения, для которых имеется две

основных операции: конкатенация (сцепление) двух выражений
и заключение выражения в скобки. Синтаксис Рефала построен
таким образом, чтобы эти операции имели кратчайшие обозначе-
ния.

А именно, если у нас имеются два результатных выражения
Re′ и Re′′, то запись

Re′ Re′′

тоже является результатным выражением, которое означает, что
надо вычислить Re′ и Re′′, а полученные результаты сцепить вме-
сте. Т. е. если результатом вычисления Re′ и Re′′ являются объ-
ектные выражения E ′ и E ′′ соответственно, то результатом вычис-
ления Re′ Re′′ является объектное выражение E ′ E ′′.

Если у нас имеется результатное выражение Re, то запись

(Re)

тоже является результатным выражением, которое означает, что
надо вычислить Re, а полученный результат заключить в круглые
скобки. Т. е. если результатом вычисления Re является объектное
выражение E , то результатом вычисления (Re) является (E).

Например, результатом вычисления результатного выражения

sX ’+’ sY (eZ)

в среде {sX = 25, sY = 36, eZ = A (B C) D} является объект-
ное выражение

25 ’+’ 36 (A (B C) D)

1.3.2 Переменные

Каждая переменная в Рефале Плюс должна начинаться с при-
знака типа переменной. Признак типа указывает на множество
допустимых значений переменной и должен быть одной из че-
тырех букв: s, t, v или e. В соответствии с этим все перемен-
ные делятся на четыре категории: s-переменные, t-переменные,
v-переменные и e-переменные.

Значениями s-переменных могут быть только символы, зна-
чениями t-переменных — только объектные термы, значениями
v-переменных — только непустые объектные выражения. Что ка-
сается e-переменных, то их значениями могут быть произвольные
объектные выражения.

В дальнейшем мы будем говорить, что некоторая переменная
является ve-переменной, если она является либо v-переменной, ли-
бо e-переменной.

1.3.3 Образцы

В языке Рефал Плюс основным средством анализа объектных
выражений являются образцы.

Образцы могут содержать символы, круглые скобки и пере-
менные. Например:

A B C
tX (eY B)

Каждый образец изображает множество объектных выраже-
ний, которые могут быть получены из него путем замены пере-
менных на произвольные значения, удовлетворяющие их типам.
Например, образец A eX изображает множество объектных вы-
ражений, которые начинаются с символа A, а образец sX sY —
множество объектных выражений, состоящих ровно из двух сим-
волов.

Если одна и та же переменная входит в образец несколько раз,
то все ее вхождения должны иметь одно и то же значение. Напри-
мер, образец tX tX изображает множество выражений, состоящих
из двух одинаковых термов.

Если нам задано объектное выражение E и образец P , то все-
гда можно сопоставить E с P и установить, имеет ли E струк-
туру, предписанную образцом P . Если да, то мы говорим, что E
успешно сопоставляется с P , в противном случае мы говорим,
что результатом сопоставления является неуспех.

В результате успешного сопоставления E с P переменные,
входящие в P , связываются с соответствующими частями вы-
ражения E . Таким образом, результатом сопоставления E с P

является некоторая среда ρ. Например, если мы сопоставим
AAA BBB CCC с образцом eX sY, результатом сопоставления будет
среда {eX = AAA BBB, sY = CCC}.

Попробуем теперь сопоставить выражение A B C с образцом
e1 sX e2. Мы обнаружим, что это можно сделать тремя различ-
ными способами, в результате чего получаются три разных среды:

{e1 = , sX = A, e2 = B C}
{e1 = A, sX = B, e2 = C}
{e1 = A B, sX = C, e2 = }

Что считать результатом сопоставления в подобных случаях?
В Рефале Плюс эта проблема решается следующим образом. Счи-
тается, что правильными являются все варианты сопоставления,
однако одни варианты “предшествуют” другим, т. е. имеют прио-
ритет.

А именно, пусть ρ1 и ρ2 — два варианта сопоставления E с
P . Рассмотрим все вхождения переменных в P . Если ρ1 и ρ2 не
совпадают, они приписывают некоторым переменным различные
значения. Найдем в P самое первое слева вхождение переменной,
которому ρ1 и ρ2 приписывают разные значения и сравним длину
этих значений. Если значение, приписываемое средой ρ1, коро-
че, чем значение, приписываемое средой ρ2, то считается, что ρ1

“предшествует” ρ2, т. е. имеет приоритет перед ρ2, в противном
случае считается, что ρ2 “предшествует” ρ1.

Например, сопоставим объектное выражение (A1 A2 A3)
(B1 B2) с образцом e1 (eX sA eY) e2. В результате получится
следующее множество вариантов сопоставления

{e1 = , eX = , sA = A1, eY = A2 A3, e2 = (B1 B2)}
{e1 = , eX = A1, sA = A2, eY = A3, e2 = (B1 B2)}
{e1 = , eX = A1 A2, sA = A3, eY = , e2 = (B1 B2)}
{e1 = (A1 A2 A3), eX = , sA = B1, eY = B2, e2 = }
{e1 = (A1 A2 A3), eX = B1, sA = B2, eY = , e2 = }

где варианты сопоставления перечислены в соответствии с их при-
оритетами, т. е. самый первый вариант находится на первом месте
и т. д.

Описанный выше способ упорядочения вариантов сопостав-
ления называется сопоставлением слева направо. Однако в Ре-
фале Плюс имеется возможность упорядочивать варианты со-
поставления и справа налево, при этом упорядочение происхо-
дит не по самому левому вхождению переменной, а по самому
правому. Чтобы изменить порядок сопоставления, следует при-
писать к образцу слева ключевое слово $r. Например, если мы
сопоставим объектное выражение (A1 A2 A3)(B1 B2) с образцом
$r e1 (eX sA eY) e2, множество вариантов сопоставления будет
упорядочено следующим образом:

{e1 = (A1 A2 A3), eX = B1, sA = B2, eY = , e2 = }
{e1 = (A1 A2 A3), eX = , sA = B1, eY = B2, e2 = }
{e1 = , eX = A1 A2, sA = A3, eY = , e2 = (B1 B2)}
{e1 = , eX = A1, sA = A2, eY = A3, e2 = (B1 B2)}
{e1 = , eX = , sA = A1, eY = A2 A3, e2 = (B1 B2)}

1.4 Функции, определяемые в программе

1.4.1 Форматы функций

С чисто формальной точки зрения все функции в языке Рефал
Плюс имеют ровно один аргумент и всегда выдают ровно один
результат. Однако, во многих случаях, мы заранее знаем, какую
структуру должны иметь аргумент и результат. Например, функ-
ция Add всегда должна получать на входе объектное выражение,
состоящее из двух символов, и всегда выдает на выходе объектное
выражение, состоящее из одного символа.

Ограничения, накладываемые на структуру аргументов и ре-
зультатов функций, описываются с помощью объявлений функ-
ций. Так, например, объявление функции Add имеет вид:

$func Add sX sY = sZ;

В общем случае объявление функции F имеет вид

$func F Fin = Fout;

где Fin — входной формат функции, а Fout — ее выходной формат.
Форматы функции могут содержать символы, круглые скобки и
переменные. Индексы переменных никакой информации не несут,
используются в качестве комментариев и могут опускаться.

Входной и выходной форматы должны быть “жесткими”, т. е.
на одном уровне скобок может находиться не более чем одна ve-
переменная. Например, формат (e)(e) является жестким, а фор-
мат e A e — не является.

Все аргументы и результаты функции должны иметь структу-
ру, предписанную ее объявлением. Объявление функции должно
предшествовать ее первому использованию в каком-либо резуль-
татном выражении в программе. При этом, если функция опреде-
лена в самой программе, ее объявление появляется в тексте про-
граммы явно. Если же функция определена в другом модуле, ее
объявление должно быть импортировано в программу с помощью
директивы $use.

Во время компиляции программы производится проверка то-
го, что во всех вызовах функции структура ее аргумента соответ-
ствует ее входному формату. Например, результатное выражение

<Add 2 <Add sX sY>>

построено правильно. По отношению ко внутреннему вызову
функции это очевидно, но для того чтобы проверить корректность
наружного вызова, уже требуется привлечь информацию о струк-
туре результата функции Add. А именно, заменяем <Add sX sY>
на выходной формат функции Add, в результате чего получается
<Add 2 s>.

Отсюда видно, что аргумент наружного вызова соответству-
ет входному формату функции Add. С другой стороны, если мы
напишем в программе результатное выражение

<Add 2 <Add sX sY> 3>

оно будет расцениваться компилятором как ошибочное, поскольку
аргумент наружного вызова функции Add состоит из трех симво-
лов, а не из двух, как предписано ее входным форматом.

Таким образом, информация о входных и выходных форматах
функций позволяет находить многие ошибки в программе еще на
стадии компиляции.

1.4.2 Определения функций

Рефал-программа состоит из определений функций, каждое из
которых имеет следующий вид:

F \{ Snt1; Snt2; . . . Sntn; };

или

F { Snt1; Snt2; . . . Sntn; };

где F — имя функции, а Snt1, Snt2, . . . , Sntn — предложения.
(Тонкое различие между “\{” и “{” для нас пока не существенно
и будет рассмотрено позднее.)

Каждое предложение Sntj имеет вид Pj Rj , где Pj — входной
образец предложения, а Rj — хвост предложения.

Определение функции имеет следующий смысл. Предполо-
жим, что требуется вычислить вызов функции F вида

<F Re>

Тогда первым делом вычисляется результатное выражение Re.
Предположим, что в результате этого получилось объектное вы-
ражение E . Тогда делается попытка сопоставить E с входными
образцами P1, P2, . . . , Pn, пока не найдется такой образец Pj , для
которого сопоставление проходит успешно, т. е. имеется хотя бы
один вариант сопоставления E с P . После этого из всех вариантов
сопоставления E с P выбирается “самый первый” вариант ρ, по-
сле чего в среде ρ вычисляется хвост Rj . Если при этом получится
некоторое объектное выражение E ′, оно и считается результатом
вызова функции.

Пока что, простоты ради, будем считать, что каждый из хво-
стов Rj имеет вид

= Rej

где Rej — результатное выражение. Хвосты такого вида называ-
ются правыми частями. Вычисление правой части = Rej сводит-
ся к вычислению результатного выражения Rej , а то, что полу-
чится, считается результатом всего хвоста.

В качестве примера рассмотрим функцию SumSq, вычисляю-
щую сумму квадратов двух чисел. В общепринятых обозначениях
ее определение записывается в виде

SumSq(X,Y ) = X ∗X + Y ∗ Y

в то время как на Рефале ее определение принимает следующий
вид:

$func SumSq sX sY = sZ;

SumSq
{
sX sY = <Add <Mult sX sX> <Mult sY sY>>;
};

Обратите внимание, что объявление функции должно распо-
лагаться в программе раньше как ее определения, так и первого
вызова, так как информация о функции, которая содержится в ее
объявлении, необходима для правильной компиляции ее опреде-
ления и ее вызовов.

Если объявление функции имеет вид

$func F Fin = Fout;

то компилятор выполняет проверку, что все входные образцы P1,
P2, . . . , Pn являются “уточнениями” входного формата Fin, а все
хвосты R1, R2,. . . , Rn заведомо вырабатывают результаты, удо-
влетворяющие выходному формату Fout.

1.4.3 Определения функций из одного предложения

Если определение функции содержит только одно предложе-
ние Snt, т. е. имеет вид

F \{ Snt; };

его разрешается записывать в сокращенном виде следующим об-
разом:

F Snt;

Например, определение функции SumSq может быть записано
следующим образом:

SumSq sX sY = <Add <Mult sX sX> <Mult sY sY>>;

1.4.4 Локальные переменные

Определим функцию SqSub1, вычитающую из аргумента еди-
ницу и возводящую полученное число в квадрат:

SqSub1(X) = (X − 1) ∗ (X − 1)

Простейшее определение на Рефале имеет вид:

$func SqSub1 sX = sZ;

SqSub1 sX = <Mult <Sub sX 1> <Sub sX 1>>;

Очевидный недостаток такого определения в том, что дважды
производятся одни и те же вычисления: а именно, дважды вы-
числяется выражение <Sub sX 1>. Этого можно избежать посред-
ством введения вспомогательной функции Sq:

$func SqSub1 sX = sZ;
$func Sq sY = sZ;

SqSub1 sX = <Sq <Sub sX 1>>;
Sq sY = <Mult sY sY>;

Единственное назначение функции Sq — подождать, пока за-
кончится вычитание единицы, “поймать” полученный результат и
затем продолжить вычисления. Ясно, что введение вспомогатель-
ных функций, как правило, загромождает программу и делает ее
более трудной для восприятия. Поэтому в Рефале Плюс имеется

возможность вводить новые переменные для обозначения проме-
жуточных результатов вычислений.

А именно, мы можем записать определение функции Sq-Sub1
следующим образом:

$func SqSub1 sX = sZ;

SqSub1 sX =
<Sub sX 1> :: sY,
<Mult sY sY>;

где <Sub sX 1> :: sY означает, что нужно вычислить <Sub sX 1>
и запомнить результат в переменной sY. Затем нужно вычислить
<Mult sY sY>. При этом разрешается использовать значение пе-
ременной sY. А то, что получится, считается результатом всей
правой части предложения.

1.4.5 Лирико-синтаксическое отступление: тропы,
хвосты и источники

В этом месте нам придется на время отвлечься от темы “ло-
кальные переменные” и обсудить некоторые особенности синтак-
сиса Рефала Плюс. Эти особенности связаны не с сутью Рефала
Плюс, а с тем, что при его разработке была сделана попытка до-
биться максимальной краткости при записи программ. Однако, к
сожалению, это привело к усложнению системы понятий, связан-
ных с синтаксисом Рефала Плюс.

По сути, все конструкции Рефала Плюс, используемые в опре-
делениях функций, занимаются либо анализом структуры объект-
ных выражений, либо что-то вычисляют и выдают результат. Для
анализа данных служат образцы, а для выработки результатов —
конструкции, которые называются тропами.

Термин тропа был использован для того, чтобы подчеркнуть,
что выработка результата — это, в общем случае, процесс слож-
ный, который может потребовать выполнения некоторой последо-
вательности шагов: вычисление постепенно, как бы шаг за шагом,
продвигается вдоль тропы. А результатное выражение — это про-
стейший пример тропы, когда результат получается “за один шаг”.

Конструкция

= <Sub sX 1> :: sY,
<Mult sY sY>;

является более сложным примером тропы, когда вычисление ре-
зультата выполняется в два этапа. Сначала вычисляется проме-
жуточный результат, а потом он используется при вычислении
результатного выражения <Mult sY sY>.

При этом, запятая , не обозначает никаких действий. Ее роль
— чисто синтаксическая. Если ее убрать, то переменная sY “при-
липнет” к следующему за ней результатному выражению

= <Sub sX 1> :: sY
<Mult sY sY>;

и станет непонятно, где заканчивается часть тропы, ответствен-
ная за присваивание, и где начинается выработка результата.

В связи с этим, в описании Рефала Плюс приходится исполь-
зовать два дополнительных синтаксических понятия: хвост и ис-
точник. Хвосты и источники — это “хорошие” тропы, обладающие
некоторыми полезными синтаксическими свойствами.

Хвосты — это тропы, которые начинаются с какого-то ключе-
вого слова, благодаря чему они однозначно отделяются от того,
что стоит перед ними. Сразу же следует отметить, что “ключе-
вые слова” в Рефале Плюс могут состоять не только из букв, они
могут содержать и другие литеры. Например, тропы

= A B C
, A B C

являются хвостами. (Тонкая разница между = и , для нас пока
несущественна, поскольку она проявляется только при обработке
“неуспехов”.)

Источники — это тропы, которые не содержат запятых на
верхнем уровне фигурных скобок. Например:

A B C
\{ <Sub sX 1> :: sY, <Mult sY sY>; }

являются источниками. Слово источник было использовано по-
тому, что в конструкции присваивания “источник” является ис-
точником значений для переменных.

В дальнейшем, при описании Рефала Плюс, мы будем обозна-
чать тропы через Q, хвосты — через R, а источники — через S.

Если некоторая тропа Q не является хвостом, её (без измене-
ния смысла) всегда можно превратить в хвост, приписав спереди
запятую: , Q.

Если некоторая тропа Q не является источником, её (без из-
менения смысла) всегда можно превратить в источник, заключив
в фигурные скобки: \{ Q;}.

1.4.6 Локальные переменные (продолжение)

Теперь мы можем вернуться к вопросу о локальных перемен-
ных.

А именно, присваивание локальным переменным делается с
помощью тропы вида

S :: He R

где S — источник, R — хвост, а He — так называемое “жесткое
выражение”. Жесткое выражение состоит из символов, скобок и
переменных и должно удовлетворять следующим ограничениям.
Во-первых, He не может содержать два вхождения одной и той
же переменной, во-вторых, на каждом уровне скобок может нахо-
диться не более одной ve-переменной. Легко видеть, что жесткое
выражение He является в то же время и форматным выраже-
нием, и во время компиляции программы выполняется проверка,
что S заведомо вырабатывает результаты, удовлетворяющие фор-
мату He.

Тропа S :: He R вычисляется следующим образом. Первым
делом вычисляется источник S. Если в результате получается
объектное выражение E , то переменные из He связываются с
соответствующими частями E , после чего вычисляется хвост R, и
полученный результат считается результатом всей конструкции.

34



Обратите внимание, что при вычислении тропы S :: He R ис-
точник S вычисляется в той среде, в которой находится вся кон-
струкция. После этого среда пополняется значениями переменных
изHe, после чего хвост R вычисляется уже в новой, исправленной
среде. Таким образом, результатом вычисления тропы

100 :: sX, <Add sX 1> :: sX = sX

является 101.
В тех случаях, когда жесткое выражение He — пустое, тропа

вида S :: R может быть записана в сокращенном виде как S R.
Такая конструкция называется условием и обычно применя-

ется в тех случаях, когда требуется вычислить S не для получе-
ния какого-то объектного выражения, а ради побочных эффектов,
возникающих при вычислении S. Например, в результате вычис-
ления тропы

<PrintLn "A">, <PrintLn "B">, <PrintLn "C"> =

будет напечатано три строчки, первая из которых будет содер-
жать литеру A, вторая — литеру B, а третья — литеру C.

Если же в присваивании S :: He R хвост R состоит из од-
ной запятой, его можно опустить, в результате чего получается
конструкция S :: He.

1.4.7 Рекурсия

Определение функции может содержать как вызовы библио-
течных функций, так и вызовы функций, определяемых в про-
грамме. В частности, функция может вызывать и саму себя (либо
непосредственно, либо через другие определяемые функции). В
этом случае будем говорить, что определение функции является
рекурсивным.

Необходимость в рекурсивных определениях обычно возника-
ет в тех случаях, когда функция должна быть определена для
бесконечного множества аргументов и при этом размер аргумен-
та может быть произвольно большим.

Рассмотрим, например, следующую задачу. Пусть требуется
определить функцию Reverse, которая “переворачивает” выраже-
ние, т. е. переставляет термы аргумента в обратном порядке. А
именно, если на вход поступает объектное выражение вида

T1 T2 . . . Tn

где T1, T2, . . . , Tn — некоторые объектные термы, то на выходе
должно получиться объектное выражение

Tn . . . T2 T1

Если бы длина входного выражения была ограничена, напри-
мер, если бы мы знали, что n<=3, можно было бы разобрать че-
тыре разных случая и написать следующее определение функции:

$func Reverse e.Exp = e.Exp;

Reverse
{
= ;
t1 = t1;
t1 t2 = t2 t1;
t1 t2 t3 = t3 t2 t1;
};

Однако, если длина входного выражения не ограничена свер-
ху, получается, что нужно разобрать в программе бесконечное
число случаев, что приводит к программе бесконечно больших
размеров.

Можно, однако, преодолеть это затруднение с помощью ре-
курсии. В данном случае можно рассуждать следующим образом.
Рассмотрим входное выражение

T1 T2 . . . Tn

Если n = 0, то результатом должно быть пустое выражение. Если
же n >= 1, то можно свести исходную задачу к более простой:

отбросить первый терм входного выражения, получив выражение
длины n− 1,

T2 . . . Tn

и перевернуть это более короткое выражение. Получится

Tn . . . T2

Теперь ясно, что если к этому выражению приписать справа T1,
мы как раз и получим желаемый результат

Tn . . . T2 T1

Эти рассуждения приводят нас к следующему рекурсивному
определению функции Reverse:

Reverse
{
= ;
t.X e.Rest = <Reverse e.Rest> t.X;
};

Интересно отметить, что для задачи переворачивания выра-
жения существует и другое решение, которое ничем не хуже пер-
вого: чтобы свести задачу к более простой, можно было бы отбро-
сить не первый терм, а последний. Тогда получилось бы следую-
щее решение:

Reverse
{
= ;
e.Rest t.X = t.X <Reverse e.Rest>;
};

Нетрудно заметить также, что в действительности сущность
решения состоит в том, что мы разбиваем исходное выражение E
на два непустых выражения E1 и E2 меньшего размера, таких, что

E = E1 E2

37



Теперь мы можем перевернуть каждое из выражений E1 и E2
по отдельности. Пусть при этом получатся выражения E ′1 и E ′2
соответственно. Тогда ясно, что

E ′2 E ′1

равно результату обращения исходного выражения E .
Если Рефал реализован на многопроцессорной машине таким

образом, что обращение выражений E1 и E2 можно выполнять од-
новременно, оказывается, что выгодно, чтобы E1 и E2 были при-
близительно одинаковой длины. Тогда получаем следующий ва-
риант описания функции, в котором используются библиотечные
функции из модулей Access и Arithm:

$func Reverse e.Exp = e.Exp;

Reverse
{
= ;
t1 = t1;
eX,
<Length e.X> :: sLen,
<Div sLen 2> :: sDiv,
= <Reverse <Middle sDiv 0 eX>>
<Reverse <Left 0 sDiv eX>>;

};

1.5 Неуспехи и аварии

1.5.1 Неуспех при вычислении результатных
выражений и троп

До сих пор мы считали, что результатом вычисления тропы
Q является некоторое объектное выражение E . В действительно-
сти, однако, это вычисление может закончиться неуспехом или
аварией.

В случае неуспеха результатом вычисления является особое
значение, “неуспех”, которое не является объектным выражением.
Простейший способ породить неуспех — это вычислить хвост вида

$fail

В случае аварии результатом вычисления является особое зна-
чение, которое не является объектным выражением и имеет вид
$error(E), где E — некоторое объектное выражение, которое яв-
ляется сообщением об ошибке. Обычно E начинается с символа-
слова, являющегося именем функции, во время работы которой
возникла авария. Например, если мы попытаемся вычислить вы-
зов функции

<Div 10 0>

возникнет авария “деление на ноль”, а результатом вычисления
этого вызова будет

$error(Div "Divide by zero")

Значения вида $error(E) обладают следующим свойством.
Пусть нам требуется вычислить некоторую конструкцию. То-
гда, если при вычислении некоторой составной части этой кон-
струкции получается результат $error(E), дальнейшее вычисле-
ние конструкции прекращается и результатом вычисления всей
конструкции считается $error(E). Единственным исключением
из этого правила является конструкция $trap, специально пред-
назначенная для “перехвата” аварий.

Все тонкости, связанные с обработкой аварий, подробно опи-
саны в главах 2 и 3, поэтому, чтобы не загромождать изложение,
в данной главе мы будем их по возможности опускать.

1.5.2 Перестройки

Рассмотрим следующую задачу: пусть нам задано некоторое
объектное выражение E , которое заведомо содержит на нуле-
вом уровне не менее двух символов-литер ’+’. Требуется раз-
бить это выражение E на три части Ex, Ea и Ey, такие, что

E = Ex ’+’ Ea ’+’ Ey, и при этом Ex и Ey не содержат ’+’ на
нулевом уровне скобок. Функцию, решающую данную задачу, на-
зовем PlusPlus. Тогда, например

<PlusPlus ’AAA+BBB+CCC+DDD+EEE’> =>
(’AAA’)(’BBB+CCC+DDD’)(’EEE’)

Таким образом, требуется найти в E самый левый ’+’ и самый
правый ’+’. Самый левый ’+’ легко найти с помощью сопостав-
ления E с образцом $l eX ’+’ eP, а самый правый — с помощью
сопоставления с образцом $r eQ ’+’ eY. Внутри одного образца
сопоставление выполняется либо слева направо, либо справа на-
лево, поэтому мы не можем объединить эти два образца в один, и
вынуждены выполнять сопоставление в два приема. Таким обра-
зом, мы можем решить задачу следующим образом:

$func PlusPlus eZ = (eX)(eA)(eY);
$func PlusPlusAux (eX)(eP) = (eX)(eA)(eY);

PlusPlus $l eX ’+’ eP = <PlusPlusAux (eX)(eP)>;
PlusPlusAux $r (eX)(eA ’+’ eY) = (eX)(eA)(eY);

Мы видим, что из-за необходимости выполнять анализ выра-
жения в два этапа, пришлось ввести вспомогательную функцию.
Однако, без этого можно было бы обойтись.

Перестройками называются тропы вида

S : Snt

где S – источник, а Snt – предложение вида P R, состоящее из
образца P и хвоста R.

Если хвост R состоит из одной запятой, его разрешается опус-
кать, в результате чего перестройка принимает вид S : P .

Вычисление перестроек производится следующим образом.
Сначала вычисляется S. Если получится неуспех, то результа-
том всей перестройки считается неуспех. Если же в результате
вычисления S получится объектное выражение E , то E сопостав-
ляется с образцом P . При этом, на рассматриваемые варианты

сопоставления накладывается дополнительное ограничение: если
в той среде, в которой вычисляется перестройка, значения некото-
рых переменных уже определены, то рассматриваются только те
варианты сопоставления, для которых эти переменные сохраняют
прежние значения.

Например, если переменная sX уже имеет значение 1, то в ре-
зультате сопоставления объектного выражения 1 2 1 2 с образ-
цом eA sX eB получатся только два варианта сопоставления

{eA = , sX = 1, eB = 2 1 2}
{eA = 1 2, sX = 1, eB = 2}

а не четыре.
Пусть теперь ρ1, ρ2, . . . , ρn — все получившиеся варианты со-

поставления, перечисленные в соответствии с порядком, введен-
ным ранее для вариантов сопоставления. Далее делается попытка
вычислить R, используя значения переменных из первого вариан-
та сопоставления. Если результатом вычисления R является объ-
ектное выражение E , оно и считается результатом перестройки.
Если же вычисление хвоста R завершилось неуспехом, то этот
неуспех “перехватывается”, т. е. самый первый вариант сопостав-
ления отбрасывается и делается попытка точно так же вычислить
R для оставшихся вариантов сопоставления.

Если для всех вариантов сопоставления результатом вычисле-
ния хвоста R является неуспех, то и результатом всей перестройки
является неуспех.

Например, вычисление тропы

’ABC’ : $r e1 sX e2, <Print sX> $fail

приведет к тому, что будет напечатанa последовательность литер
’CBA’, после чего вычисление тропы завершится с результатом
“неуспех”.

Теперь мы можем дать новое определение функции PlusPlus,
не использующее вспомогательную функцию PlusPlusAux:

$func PlusPlus eZ = (eX)(eA)(eY);

PlusPlus
$l eX ’+’ eP,
eP : $r eA ’+’ eY
= (eX)(eA)(eY);

1.5.3 Перехват неуспехов

В языке Рефал Плюс имеется возможность проверить, закон-
чилось ли вычисление некоторой тропы неуспехом и, в зависимо-
сти от этого, предпринять те или иные действия.

Отрицаниями условий называются тропы вида

# S R

где S — источник, а R — хвост. Если хвост R состоит из одной
запятой, его разрешается опускать, в результате чего отрицание
условия принимает вид # S.

С синтаксической точки зрения любое отрицание условия яв-
ляется хвостом.

Вычисление таких хвостов производится следующим образом.
Первым делом делается попытка вычислить источник S. Если в
результате этого получается пустое объектное выражение, резуль-
татом всего хвоста является неуспех. Если же результатом вычис-
ления источника S является неуспех, то вычисляется хвост R, и
то, что получится, считается результатом всей конструкции.

Распутьями называются тропы вида

\{ Q1; Q2; . . . Qn; }

где Q1, Q2, . . . , Qn — некоторые тропы. С синтаксической точки
зрения любое распутье является источником.

Вычисление распутья происходит следующим образом. Пер-
вым делом делается попытка вычислить тропу Q1. Если в ре-
зультате этого получается объектное выражение E , оно счита-
ется результатом всего распутья. Если же результатом вычисле-
ния тропы Q1 является неуспех, то делается попытка вычислить
\{ Q2; . . . , Qn; }, и то, что получится, является результатом всего
распутья. Таким образом, распутье “перехватывает” неуспехи.

Если результатом вычисления всех троп является неуспех, то
и результатом всего распутья является неуспех.

Рассмотрим, например, тропу

1 :: sX,
\{ sX : 0 = 1; sX : 1 = 0; }

Она вычисляется следующим образом. Первым делом выпол-
няется определение локальной переменной 1 :: sX, в результате
чего sX получает значение 1. После этого делается попытка вы-
числить первую тропу распутья, т. е. sX : 0 = 1. При попытке
выполнить сопоставление sX : 0 выясняется, что сопоставление
проходит неуспешно, поэтому результатом этой тропы является
неуспех. Вследствие этого делается попытка вычислить следую-
щую тропу sX : 1 = 0, в результате чего получается результат 0,
который и является результатом вычисления всей тропы.

В некоторых случаях, однако, полезным оказывается другой
вариант распутья, имеющий следующий вид:

{ Q1; Q2; . . . Qn; }

Его отличие от предыдущего варианта проявляется в том,
что если результатом вычисления всех троп является неуспех,
то результатом вычисления распутья является ошибка $error(F
"Unexpected fail"), где F — имя функции, в которой находится
распутье.

Таким образом, мы можем считать, что пара скобок \{ ... }
“прозрачна” для неуспехов, в то время как пара скобок { ... }
для них “непрозрачна”, ибо неуспех не может “выскочить” из рас-
путья { Q1; Q2; . . . Qn; }.

Довольно часто встречаются тропы следующего вида:

S:V e, \{V e : Snt1; V e : Snt2; . . . V e : Sntn;}
S:V e, {V e : Snt1; V e : Snt2; . . . V e : Sntn;}

где V e — некоторая e-переменная, которая больше нигде в опре-
делении функции не используется, а каждое из предложений Sntj

имеет вид Pj Rj . Для таких троп предусмотрена, соответственно,
следующая сокращенная форма записи:

S : \{Snt1; Snt2; . . . Sntn;}
S : {Snt1; Snt2; . . . Sntn;}

называемая выбором, при этом конструкции \{Snt1; Snt2;
. . . Sntn;} и {Snt1; Snt2; . . . Sntn;} называются образцовыми
распутьями.

Хвосты Rj , состоящие из одной запятой, разрешается опус-
кать, в результате чего соответствующие предложения Pj Rj при-
нимают вид Pj .

Особенно следует отметить, что если образцовое распутье ис-
пользуется в определении функции на самом верхнем уровне, то
хвосты, состоящие из одной запятой, тоже могут быть опущены.
Например, функция

A_B \{ A = ; B = ; };

возвращает пустое выражение, если на вход подать A или B, а
в противном случае вырабатывает неуспех. Она эквивалентна
функции

A_B \{ A , ; B , ; };

которая эквивалентна функции

A_B \{ A; B; };

Таким образом, распутье { sX : 0 = 1; sX : 1 = 0; } бо-
лее кратко записывается в виде выбора sX : { 0 = 1; 1 = 0; },
а распутье \{ sX : A,; sX : B,; } может быть сокращено до
sX : \{ A; B; }.

С синтаксической точки зрения, все выборы являются источ-
никами, что позволяет записывать, например, присваивания сле-
дующего вида:

sX : { 0 = 1; 1 = 0; } :: sY = <Add sX sY>

При этом сначала вычисляется источник sX : {0 = 1;
1 = 0;}, затем полученный результат присваивается переменной
sY, после чего вычисляется хвост = <Add sX sY>.

1.5.4 Управление перехватом неуспехов

Как мы видели выше, Рефал Плюс предоставляет довольно бо-
гатый набор конструкций для перехвата неуспехов. Однако, ино-
гда возникает желание породить неуспех такой силы, чтобы он
смог преодолеть все ловушки, расставленные для его перехвата
(или хотя бы часть из них). Для этого служат заборы и отсече-
ния.

Заборами называются хвосты вида \? Q, а отсечениями —
хвосты вида \! Q.

Заборы и отсечения являются пометками в программе, поз-
воляющими управлять распространением неуспехов. А именно,
всякое отсечение \! Q должно находиться внутри некоторого за-
бора \? . . . \! Q . . . . Вычисление отсечения \! Q производится
следующим образом. Первым делом вычисляется тропа Q. Если
это вычисление завершается с некоторым результатом X, этот ре-
зультат считается результатом всей конструкции \?. . . \! Q . . . .
В частности, если X — неуспех, то неуспехом заканчивается и
вычисление всей конструкции \? . . . \! Q . . . .

Рассмотрим пример, иллюстрирующий использование заборов
и отсечений. Предположим, что мы пытаемся вычислить следую-
щую тропу:

eA : e1 ’+’ e2 ’-’ e3
= (e1)(e2)(e3)

Если значение переменной eA содержит ’+’, а вслед за ним содер-
жит ’-’ на нулевом уровне скобок, то в результате сопоставления
с образцом будут найдены самый левый ’+’ и ближайший за ним
’-’, после чего вычисление тропы успешно завершается. Теперь
рассмотрим случай, когда eA содержит ’+’ на нулевом уровне,
но не содержит на нулевом уровне ни одного ’-’. Тогда первым
делом будет найден первый ’+’, а затем будет просмотрена вся
оставшаяся часть выражения, чтобы найти ’-’. Поскольку это
сделать не удастся, будет выполнено удлинение значения перемен-
ной e1, после чего снова будет выполнен поиск ’-’ в оставшейся
части выражения. Между тем, этого можно было бы и не делать,

45



ибо если ’-’ не нашелся первый раз, он заведомо не найдется и
во второй раз.

Мы можем избежать такого перебора с помощью \? и \!. Для
этого первым делом разобьем исходную перестройку на две части:

eA : e1 ’+’ еX,
eX : e2 ’-’ e3
= (e1)(e2)(e3)

Теперь, если сопоставление значения eX с образцом e2 ’-’ e3 за-
вершается неуспешно, делается попытка испробовать следующий
вариант сопоставления для eA : e1 ’+’ еX. Этого, однако, мож-
но избежать,добавив \? и \! следующим образом:

\? eA : e1 ’+’ еX
\! eX : e2 ’-’ e3
= (e1)(e2)(e3)

Теперь, если внутренняя перестройка выдает неуспех, этот
неуспех сразу же становится результатом всей тропы.

1.5.5 Смысл правых частей

Правые части, имеющие вид = Q , где Q — некоторая тропа,
являются еще более мощным средством ограничения перебора,
чем заборы и отсечения.

В общих чертах различие между ключевыми словами , и =
заключается в следующем. Ключевое слово = предписывает, что
надо окончательно зафиксировать значения переменных, имею-
щиеся к данному моменту, и приступать к вычислению тропы Q.
Если результатом вычисления Q будет неуспех, то надо считать,
что неуспехом закончилось и вычисление всей конструкции, объ-
емлющей = Q. При этом не следует пытаться подбирать для пе-
ременных какие-то другие значения.

А в случае хвоста , Q , если результатом вычисления Q яв-
ляется неуспех, это необязательно приводит к неуспеху всей кон-
струкции, поскольку делается попытка подобрать для перемен-
ных другие значения.

Например, тропа

A B C : e sX e, sX : C , sX

вырабатывает значение C, поскольку для переменной sX в конце-
концов подбирается “правильное” значение C. В то же время, тро-
па

A B C : e sX e = sX : C , sX

вырабатывает неуспех, поскольку переменная sX принимает зна-
чение A и после прохождения через = это значение фиксируется.
Затем терпит неудачу перестройка sX : C, но попытки подбора
нового значения для sX уже не происходит.

Такого интуитивного понимания вполне достаточно для прак-
тического написания программ. Поэтому большинство читателей
может смело пропустить остаток данного раздела. Ибо сомнения
возникают только при рассмотрении достаточно сложных троп,
наподобие следующей:

A B : e sX e,
sX : { A = B; B = A; } :: sY,
sY : A, Ok

Здесь в начале тропы стоит перестройка A B : e sX e, кото-
рая сначала присваивает переменной sX значение A. Затем делает-
ся попытка вычислить то, что находится справа от перестройки.
А именно, значение переменной sX подается на вход образцового
распутья { A = B; B = A; }, вырабатывающего значение B, если
на него подано A, и значение A, если на него подано B. Затем значе-
ние, выданное образцовым распутьем, присваивается переменной
sY. Дальше идет перестройка, которая сопоставляет значение sY
с символом A. Поскольку первоначально переменной sX присва-
ивается значение A, переменной sY присваивается значение B, и
перестройка sY : A терпит неудачу.

Возникает интересный вопрос: что происходит даль-
ше? Ведь при вычислении значения образцового распутья
{ A = B; B = A; } вычисление проходит через ключевое слово =,
которое “фиксирует значения переменных, имеющиеся к данному
моменту”. Означает ли это, что когда перестройка sY : A терпит

неудачу, перестройка A B : e sX e уже не должна пытаться по-
добрать новое значение? Согласно подходу, принятому в Рефале
Плюс, в данном случае = не отменяет перебор для перестройки
A B : e sX e. Поэтому, для sX подбирается новое значение
B, которое подается на образцовое распутье, и переменная sY
принимает новое значение B. После этого перестройка sY : A
срабатывает успешно, и, в качестве результата всей тропы,
выдается значение Ok.

В пользу такого подхода имеется следующее соображение. Об-
разцовое { A = B; B = A; } фактически представляет собой за-
маскированный вызов безымянной функции. Если определить эту
функцию в явном виде (назвав, например, AB)

$func AB s = s;
AB { A = B; B = A; };

всю тропу можно переписать в виде

A B : e sX e,
<AB sX> :: sY,
sY : A, Ok

А если бы считалось, что область действия = не имеет ограни-
чений, то не было бы возможности легко выносить части троп в
отдельные функции, или, наоборот, заменять вызовы функций на
образцовые распутья из определений этих функций.

Для того чтобы можно было описать, какие ограничения на-
кладываются на области действия ключевого слова =, требуется
ввести несколько дополнительных понятий.

Пусть какая-то конструкция входит в качестве составной ча-
сти в некоторую другую конструкцию. Мы будем говорить, что
эта внутренняя конструкция является поставщиком, если резуль-
тат ее работы, согласно семантике языка, считается результатом
работы объемлющей конструкции. Например, в тропе S :: He R
хвост R является поставщиком, ибо результат вычисления R счи-
тается результатом вычисления всей конструкции.

Если некоторая конструкция не является поставщиком для
объемлющей конструкции, мы будем говорить, что она является

ассистентом. Например, в тропе S :: He R источник S являет-
ся ассистентом, ибо его результат не считается результатом всей
конструкции (хотя и может косвенно использоваться в R через
значения переменных).

Теперь рассмотрим различные конструкции, и для каждой из
них укажем в явном виде, какие их части являются поставщика-
ми, а какие — ассистентами.

Если определение функции имеет вид F Palt , то образцовое
распутье Palt является, по определению, ассистентом. Это вы-
глядит странно, поскольку Palt вырабатывает значение функции.
Но здесь — случай особый, поскольку речь идет о самом верхнем
уровне в определении функции.

Если тропа имеет один из следующих видов: S R , S :: He R
, S : P R или #S R, то источник S является ассистентом.

Если источник имеет вид S : Palt, то источник S является
ассистентом.

Рассмотрим теперь некоторую конструкцию вместе со всеми
объемлющими ее ассистентами. Среди этих ассистентов всегда
существует наименьший, и мы будем называть его хозяином рас-
сматриваемой конструкции. В частности, всякий ассистент явля-
ется своим собственным хозяином.

Теперь мы можем описать семантику правых частей.
Пусть хозяином правой части = Q является некоторая объем-

лющая конструкция . . . = Q . . . . Тогда первым делом вычисляется
тропа Q. Если это вычисление завершилось с некоторым резуль-
татом X, то X считается результатом всего хозяина . . . = Q . . . .

В частности, если X — неуспех, то и значением хозяина
. . . = Q . . . будет неуспех, даже если внутри него расставлены ло-
вушки, предназначенные для перехвата неуспехов.

Например, результатом вычисления тропы

\{ A B C : $l e sX e, sX : B, sX} :: sY, sY

будет символ B, ибо sX принимает значение A, сопоставление кото-
рого с символом B терпит неуспех, в результате чего sX принимает
новое значение B, и вычисление успешно завершается. Но если мы
заменим запятую на равенство, мы получим тропу

49



\{ A B C : $l e sX e = sX : B, sX} :: sY, sY

вычисление которой терпит неудачу.
На использование заборов, отсечений и правых частей накла-

дываются некоторые ограничения.
И отсечение \!Q , и соответствующий ему забор \? . . . \!Q . . .

должны иметь общего хозяина.
На пути от \! к соответствующему ему \? не должно быть ни

одного = .

1.5.6 Откатные и безоткатные функции

Все функции, определяемые и используемые в программе на
Рефале Плюс делятся на две категории: откатные и безоткат-
ные.

Если функция F — безоткатная, то вычисление вызова <F Re>
не может закончиться получением неуспеха. Если же функция F
— откатная, то вычисление вызова <F Re>, вообще говоря, может
закончиться получением неуспеха в качестве результата.

До сих пор мы считали, что объявления функций имеют вид

$func F Fin = Fout;

однако это верно только в том случае, если функция F — без-
откатная. Если же F — откатная, ее объявление должно иметь
вид

$func? F Fin = Fout;

Теперь мы можем более точно описать семантику определений
функций. Пусть определение функции F имеет вид

F Palt

где Palt — образцовое распутье вида \{P1 R1; P2 R2; . . . Pn Rn;}
или {P1 R1; P2 R2; . . . Pn Rn;}, и пусть требуется вычислить ее
вызов <F Re>. Тогда прежде всего вычисляется результатное вы-
ражение Re. Если при этом получается неуспех, то вызов функ-
ции F не выполняется и считается, что результатом вычисления

<F Re> является неуспех. Предположим теперь, что в результате
вычисления Re получилось объектное выражение E . Тогда выпол-
няется вызов функции, а именно, вычисляется источник

E : Palt

в пустой среде, т. е. в среде, в которой ни одна переменная не
имеет значения. Пусть в результате этого получилось значение
X. Если X — объектное выражение, то X считается результатом
вызова <F Re>. Если же X является неуспехом, то дальнейшие
действия зависят от того, является ли функция F откатной.

Если F — откатная, и X является неуспехом, то результатом
вызова <F Re> считается неуспех.

Если F — безоткатная, и X является неуспехом, то этот
неуспех “перехватывается” и преобразуется в ошибку

$error(F "Unexpected fail").

1.6 Логические условия

1.6.1 Проверки и предикаты

В некоторых случаях в процессе работы программы требуется
проверить справедливость некоторого условия и, в зависимости от
результатов проверки, направить вычисления либо по одному, ли-
бо по другому пути. При программировании на Рефале Плюс мы
будем придерживаться следующего соглашения. Тропа Q будет
называться условием, если результатом ее вычисления является
либо пустое объектное выражение, либо неуспех. Если результат
— пустое выражение, мы считаем, что условие выполнено, а если
результат — неуспех, мы считаем, что условие не выполнено. Та-
ким образом, пустой результат означает, что результат проверки
— “истина”, а неуспех означает, что результат проверки — “ложь”.

Конечно, не следует забывать, что вычисление тропы Q может
вообще не закончиться или же привести к возникновению ситуа-
ции ошибки, но мы будем считать, что это свидетельствует либо
о том, что программа неправильна, либо о том, что программе
были заданы недопустимые исходные данные.

Некоторые функции, входящие в библиотеку функций, ис-
пользуются только для проверки условий. Такие функции называ-
ются предикатами. В Рефале Плюс функции-предикаты выдают
пустое выражение, если их аргументы удовлетворяют условию и
выдают неуспех в случае, если их аргументы не удовлетворяют
условию. Например, функция Lt проверяет, что ее первый аргу-
мент меньше второго. А именно, результатом вычисления вызова
вида <Lt (E1)(E2)> является пустое выражение, если объектное
выражение E1 “меньше”, чем объектное выражение E2. Если же
это не так — результатом является неуспех.

Если в программе определяется некоторая функция-предикат,
ее объявление должно иметь следующий вид:

$func? F Fin= ;

Дальше мы рассмотрим различные способы использования и
комбинирования условий.

1.6.2 Ветвления

Пусть у нас имеется условие, заданное источником S и две
тропы Q′ и Q′′. Рассмотрим тропу

\? {S \! Q′; \! Q′′;}

Тогда, если результат вычисления S — пустое выражение, то вы-
числяется тропа Q′, и то, что получится, является результатом
всей конструкции. Если же результат вычисления источника S —
неуспех, то вычисляется тропа Q′′ и то, что получится, является
результатом всей конструкции.

Следует обратить внимание на использование отсечений \!.
Они существенны в том случае, если вычисление Q′ или Q′′ закан-
чивается неуспехом. Предположим, что они убраны, в результате
чего тропа приобретает следующий вид:

{S, Q′; Q′′;}

Тогда, если условие S выполнено, производится вычисление тро-
пы Q′. Предположим, однако, что вычисление тропы Q′ заканчи-
вается неуспехом. Тогда этот неуспех не будет выдан в качестве

результата всей конструкции, а будет перехвачен, что приведет к
вычислению тропы Q′′. В данном случае это совсем не то, что нам
было нужно. Таким образом, первое отсечение необходимо, чтобы
предотвратить “перескок” на следующую тропу в распутье.

Теперь рассмотрим случай, когда условие S не выполнено,
т. е. результатом вычисления S является неуспех. В этом слу-
чае неуспех перехватывается и начинается вычисление тропы Q′′.
Предположим, что вычисление тропы Q′′ завершается неуспехом.
В этом случае неуспех ерехватывается и делается попытка вы-
числить следующую тропу в распутье. Но, поскольку следующей
тропы нет, возникает ситуация ошибки, а это опять-таки не то,
что мы хотели!

Если ветвление имеет вид

\? {S \! = Q′; \! =Q′′;}

его можно записать более кратко следующим образом:

{ S = Q′; = Q′′;}

(что мы и будем часто делать в дальнейшем).
Рассмотрим следующий пример. Определим функцию MinE,

которая сравнивает два объектных выражения E1 и E2 и выдает
E1, если E1 предшествует E2. Если же E1 равно E2 или E2 предше-
ствует E1, то MinE выдает E2.

$func MinE (eX)(eY) = e.MinXY;

MinE (eX)(eY) =
{
<Lt (eX)(eY)>

= eX;
= eY;

};

Теперь рассмотрим случай, когда условие задано тропой Q и
нужно вычислить тропу Q′, если условие выполнено, либо тропу
Q′′, если условие не выполнено. Этого можно достичь, заключив

Q в фигурные скобки и превратив его в источник \{ Q; }. После
этого ветвление может быть записано следующим образом:

\? { \{Q;} \! Q′; \! Q′′;}

1.6.3 Логические связки

В некоторых случаях приходится вычислять сложные логиче-
ские условия. При этом удобно выражать сложные условия через
более простые с помощью логических связок “и”, “или” и “не”. Хотя
Рефал Плюс и не содержит логических связок в явном виде, они
легко изображаются с помощью имеющихся в нем конструкций.

• Логическое “И”

Пусть у нас имеются два условия и требуется проверить, что
они оба выполнены.

Если оба условия заданы тропами Q′ и Q′′, достаточно вычис-
лить тропу

\{ Q′;}, Q′′

Если первое условие задано источником S, второе — тропой
Q, достаточно вычислить тропу S, Q.

И, наконец, если оба условия заданы результатными выраже-
ниями Re′ и Re′′, достаточно вычислить результатное выражение
Re′ Re′′.

• Логическое “ИЛИ”

Пусть у нас имеются два условия и требуется проверить, что
хотя бы одно из них выполнено.

Если оба условия заданы тропами Q′ и Q′′, достаточно вычис-
лить тропу

\{ Q′; Q′′; }

• Логическое “НЕ”

Пусть у нас имеется условие, заданное тропой Q, и требуется
проверить, что оно не выполнено. Этого можно достичь, вычислив
тропу

# \{Q;}

которая является сокращенной записью для тропы # \{Q;}, . Ес-
ли же условие задано источником S, достаточно вычислить тропу

# S

которая является сокращенной записью для тропы # S , .
И в том, и в другом случае мы пользуемся возможностью опус-

кать хвосты, состоящие из одной запятой.

1.6.4 Пример: программа дифференцирования

Опишем функцию, которая будет выполнять символьное диф-
ференцирование формул [Хен 1983]. Чтобы не загромождать из-
ложение, мы будем рассматривать только простые формулы, в
которые входят целые числа, переменные, а также операции + и
∗, хотя обобщить рассмотрение на случай более сложных формул
не составило бы большого труда.

Будем обозначать через x и y произвольные переменные, через
i — произвольное целое число, через e — произвольную формулу,
а через Dx(e) — результат дифференцирования e по x. Тогда пра-
вила дифференцирования можно записать в виде

Dx(x) = 1
Dx(y) = 0 (где y не равно x)
Dx(i) = 0
Dx(e1 + e2) = Dx(e1) +Dx(e2)
Dx(e1 ∗ e2) = e1 ∗Dx(e2) + e2 ∗Dx(e1)

Прежде чем писать программу дифференцирования, нужно
выбрать для формул представление в виде объектных выражений.

Обозначим через dee представление формулы e. Выберем следу-
ющее представление

dxe = x
die = i
de1 + e2e = (Sum de1e de2e)
de1 ∗ e2e = (Prod de1e de2e)

Теперь не составляет труда определить функцию Diff, пер-
вым аргументом которой является переменная, по которой надо
дифференцировать, а вторым аргументом — формула, подлежа-
щая дифференцированию. Результатом работы является продиф-
ференцированная формула.

$func Diff sX tE = tE;

Diff sX tE =
tE :
{
sX = 1;
sY = 0;
(s.Oper t.E1 t.E2) =

<Diff sX tE1> :: t.DxE1,
<Diff sX tE2> :: t.DxE2,
s.Oper :
{
Sum = (Sum t.DxE1 t.DxE2);
Prod = (Sum (Prod t.E1 t.DxE2)

(Prod t.E2 t.DxE1));
};

};

Это определение функции Diff страдает очевидным недостат-
ком: формулы, которые получаются в результате дифференциро-
вания, содержат слишком много частей, которые можно сокра-
тить. Например

DX(3 ∗ (X ∗X) + 5) =
(3 ∗ ((X ∗ 1) + (X ∗ 1)) + (X ∗X) ∗ 0) + 0

в соответствии с приведенными выше правилами дифференци-
рования. Между тем, после очевидных упрощений эту формулу
можно привести к виду

3 ∗ (X +X)

Для этого достаточно применить к формуле следующие пра-
вила преобразования:

0 + e2 ==> e2
e1 + 0 ==> e1
0 ∗ e2 ==> 0
e1 ∗ 0 ==> 0
1 ∗ e2 ==> e2
e1 ∗ 1 ==> e1

(Более сложные преобразования мы не рассматриваем, чтобы не
загромождать изложения.)

Существует два способа реализовать эти упрощения. Можно
сначала построить результат дифференцирования полностью, а
затем анализировать его, пытаясь сделать упрощения. Другой
способ — делать упрощения сразу же, в процессе дифференци-
рования. Мы поступим именно так.

Определим функции Sum и Prod. Эти функции будут получать
на входе две формулы, и выдавать формулу, которая является их
суммой или произведением соответственно. При этом будут сразу
же делаться упрощающие преобразования.

$func Sum t1 t2 = t;
$func Prod t1 t2 = t;

Sum
{
0 t2 = t2;
t1 0 = t1;
t1 t2 = (Sum t1 t2);
};

Prod
{
0 t2 = 0;
1 t2 = t2;
t1 0 = 0;
t1 1 = t1;
t1 t2 = (Prod t1 t2);
};

Теперь переделаем определение функции Diff, вставив в со-
ответствующие места вызовы функций Sum и Prod.

Diff sX tE =
tE :
{
sX = 1;
sY = 0;
(s.Oper t.E1 t.E2) =

<Diff sX tE1> :: t.DxE1,
<Diff sX tE2> :: t.DxE2,
s.Oper :
{
Sum = <Sum t.DxE1 t.DxE2>;
Prod = <Sum <Prod t.E1 t.DxE2>

<Prod t.E2 t.DxE1>>;
};

};

1.6.5 Пример: сравнение множеств

В качестве примера использования логических связок и рекур-
сии рассмотрим следующую задачу.

Как известно из теории множеств, два множества считаются
равными, если они состоят из одних и тех же элементов. Пред-
положим, что мы хотим запрограммировать на Рефале операцию
сравнения конечных множеств на равенство. Для этого прежде
всего следует придумать какой-то способ изображения множеств

в виде объектных выражений. Сначала рассмотрим множества,
элементами которых могут быть только символы (в смысле Рефа-
ла). Тогда множество символов S1,S2, . . . ,Sn можно изобразить в
виде объектного выражения

S1 S2 . . .Sn

При этом оказывается, что одному и тому же множеству мо-
гут соответствовать несколько его изображений. Например, мно-
жество {John, Mary} можно изобразить как в виде выражения
John Mary, так и в виде выражения Mary John, или даже в виде
Mary John John Mary. Таким образом, неравные выражения мо-
гут изображать равные множества.

Как известно, элементы множеств сами могут быть множе-
ствами. Поэтому мы теперь несколько усложним нашу задачу.
Предположим, что рассматриваемые нами множества могут со-
держать в качестве элементов как символы, так и множества (ко-
торые, в свою очередь снова могут содержать множества и т. д.).
Теперь мы должны как-то изобразить те элементы множества,
которые сами являются множествами.

Мы поступим следующим образом. Если элемент множества
— символ S, мы будем его изображать в виде символа S. Если же
элемент множества — множество X, мы будем изображать его в
виде терма вида ( dXe ), где dXe — изображение множества X.
Таким образом, множество {A, {A,B}, {A}} можно изобразить,
например, в виде выражения A (AB) (A) .

Теперь постараемся определить функцию-предикат IsEqSet ,
которая будет определять, являются ли два ее аргумента пред-
ставлениями одного и того же множества. Для этого мы поста-
раемся разложить понятие равенства множеств на более простые
понятия.

А именно, два множества A и B равны в том и только том
случае, если A является подмножеством B и B является подмно-
жеством A. Далее, множество A является подмножеством множе-
ства B в том и только том случае, если для любого элемента X
множества A верно, что X принадлежит B. В формализованном

виде это можно записать следующим образом:

A = B ⇔ A ⊆ B ∧B ⊆ A
A ⊆ B ⇔ (∀x ∈ A)(x ∈ B)

Поэтому мы определим не одну, а четыре функции-предиката:
IsEqSet будет проверять, что ее аргументы представляют одно и
то же множество, IsSubset будет проверять, что множество, пред-
ставляемое первым аргументом, является подмножеством множе-
ства, представляемого вторым аргументом, IsEl будет проверять,
что ее первый аргумент представляет элемент, входящий в мно-
жество, представляемое вторым аргументом, а IsEqEl будет про-
верять, что ее аргументы представляют один и тот же элемент
множества.

Интересно отметить, что если два элемента множества сами
являются множествами, то их сравнение приводит к необходимо-
сти сравнивать два множества, поэтому функция IsEqEl вынуж-
дена вызвать функцию IsEqSet . Таким образом, оказывается,
что функция IsEqSet в конечном итоге рекурсивно определяется
через себя.

$func? IsEqSet (eA)(eB) = ;
$func? IsSubset (eA)(eB) = ;
$func? IsEl tX (eA) = ;
$func? IsEqEl tX tY = ;

IsEqSet (eA)(eB) =
<IsSubset (eA)(eB)><IsSubset (eB)(eA)>;

IsSubset (eA)(eB) =
eA :
{
= ;
tX eR = <IsEl tX (eB)><IsSubset (eR)(eB)>;
};

IsEl tX (eA) =
eA : tY eR,

\{ <IsEqEl tX tY>; <IsEl tX (eR)>; };

IsEqEl tX tY =
\{
tX tY : s s

= tX : tY;
tX tY : (eA)(eB)

= <IsEqSet (eA)(eB)>;
};

1.7 Использование селекторов прямого
доступа

Одним из типичных случаев, когда требуется прямой доступ к
частям выражения, являются алгоритмы, основанные на методе
“разделяй и властвуй”. Этот метод заключается в том, что исход-
ная задача разбивается на подзадачи меньшего размера и решает-
ся для этих подзадач. Затем решение исходной задачи получает-
ся исходя из решений подзадач. Как правило, принцип “разделяй
и властвуй” применяется совместно с принципом балансировки,
гласящим, что исходную задачу следует разбивать на подзадачи
примерно равного размера [АХУ 1979].

В качестве классического примера рассмотрим задачу сорти-
ровки (т. е. упорядочения в порядке неубывания) множества це-
лых чисел.

Одним из способов решения этой задачи является сортировка
слиянием [АХУ 1979], когда исходное множество чисел S разбива-
ется на два непересекающихся множества S1 и S2 примерно рав-
ного размера. Затем S1 и S2 упорядочиваются независимо друг
от друга. В результате получаются две упорядоченные последо-
вательности чисел Q1 и Q2, которые соединяются (сливаются) в
одну упорядоченную последовательность Q, которая и является
решением исходной задачи.

Теперь определим функцию MSort, которая получает на входе
последовательность целых чисел, разбивает ее на две примерно

равные части, сортирует их (обратившись к себе рекурсивно), а
затем сливает получившиеся последовательности, обратившись к
функции Merge.

$func MSort eS = eS;
$func Merge (eX)(eY) = eZ;

MSort eS =
<Length eS> :: sLen,
{
<Le (sLen) (1)>

= eS;
= <Div sLen 2> :: sK,

<Left 0 sK eS> :: eS1,
<Middle sK 0 eS> :: eS2,

<Merge ( <MSort eS1> )( <MSort eS2> )>;
};

Теперь определим функцию Merge, которая получает на входе
две упорядоченные последовательности и объединяет их в одну
упорядоченную последовательность.

Merge (eX)(eY) =
{
eX :

= eY;
eY :

= eX;
(eX)(eY) : (sA eXRest)(sB eYRest)

= {
<Le (sA)(sB)>

= sA <Merge (eXRest)(eY)>;
= sB <Merge (eX)(eYRest)>;

};
};

1.8 Функции, выдающие несколько резуль-
татов

1.8.1 Сквозной просмотр выражений

Рассмотрим несколько примеров, демонстрирующих полез-
ность функций, выдающих несколько результатов.

Предположим, что требуется определить функцию Nmb, заме-
няющую все символы выражения на их порядковые номера. На-
пример

<Nmb A (B A) C A> => 1 (2 3) 4 5

Основная трудность здесь заключается в том, что встретив па-
ру скобок, функция не знает заранее, сколько символов находит-
ся внутри скобок, а эта информация нужна, чтобы продолжить
обработку “хвоста” выражения, стоящего после скобок. Поэто-
му функция, нумерующая символы, должна иметь два аргумен-
та: выражение, подлежащее обработке, и номер, который следует
приписать первому символу, который встретится. Эта же функ-
ция должна выдавать два результата: обработанное выражение и
первый “неиспользованный” номер символа. Таким образом, мы
приходим к следующему определению функции Nmb (использую-
щему две вспомогательные функции NmbExp и NmbTerm):

$func Nmb e.Exp = e.Exp;
$func NmbExp e.Exp sN = e.Exp sN;
$func NmbTerm t.Exp sN = t.Exp sN;

Nmb e.Exp =
<NmbExp e.Exp 1> :: e.Exp s,

e.Exp;

NmbExp e.Exp sN =
e.Exp :
{
= sN;
tX e.Rest =

63



<NmbTerm tX sN> :: tX sN,
<NmbExp e.Rest sN> :: e.Rest sN,

tX e.Rest sN;
};

NmbTerm tX sN =
tX :
{
s =

sN <Add sN 1>;
(eE) =

<NmbExp eE sN> :: eE sN,
(eE) sN;

};

1.8.2 Быстрая сортировка

Следующий пример — это так называемая быстрая сортиров-
ка [АХУ 1979], сущность которой заключается в следующем.

Пусть требуется отсортировать множество целых чисел S. Вы-
берем из S произвольное число X и разобьем S на три подмноже-
ства S1, S2 и S3, такие, что S1 содержит числа, которые меньше
X, S2 содержит числа, равные X, а S3 содержит числа, которые
больше X. Отсортируем S1, S2 и S3, пусть при этом получились
упорядоченные последовательности Q1, Q2 и Q3 (сортировка мно-
жества S2 — тривиальна, ибо все его элементы равны X). Тогда
мы можем соединить Q1, Q2 и Q3 в новую последовательность
Q1Q2Q3, которая и будет решением исходной задачи.

Определим функцию QSort, которая будет сортировать подан-
ную на вход последовательность описанным выше методом. Для
разбиения исходного множества чисел на три части используется
функция Split.

$func QSort eS = eQ;
$func Split sX eS = (eS1)(eS2)(eS3);
$func SplitAux sX (eS1)(eS2)(eS3) eS =

(eS1)(eS2)(eS3);

QSort eS =
{
eS :

= ;
eS : t

= eS;
eS : sX e

= <Split sX eS> :: (eS1)(eS2)(eS3),
<QSort eS1> eS2 <QSort eS3>;

};

Split sX eS =
<SplitAux sX ()()() eS>;

SplitAux sX (eS1)(eS2)(eS3) eS =
eS :
{
=

(eS1)(eS2)(eS3);
sY eRest =

{
<Lt (sY)(sX)>

= <SplitAux sX (eS1 sY)(eS2)(eS3) eRest>;
<Gt (sY)(sX)>

= <SplitAux sX (eS1)(eS2)(eS3 sY) eRest>;
= <SplitAux sX (eS1)(eS2 sY)(eS3) eRest>;

};
};

1.9 Итеративные циклы

Основным средством изображения циклов в Рефале Плюс яв-
ляются рекурсивные определения функций. Во многих случаях,
однако, это средство оказывается слишком универсальным. По-
этому в Рефале Плюс предусмотрена конструкция поиска, кото-

рая имеет следующий вид:

S′′ $iter S′ :: He R

и синтаксически является тропой. При этом источники S′′ и S′ яв-
ляются ассистентами, а хвост R — поставщиком (что существенно
при наличии правых частей вида = Q в S′′, S′ или R).

Если хвост R состоит из одной запятой, его разрешается опус-
кать. Также, если жесткое выражение He — пустое, его разреша-
ется опускать вместе с ключевым словом :: .

Тропа-поиск вводит новые локальные переменные (так же,
как и тропа-присваивание S :: He R ). Начальные значения
этих переменных получаются путем вычисления источника S′′.
Затем делается попытка вычислить хвост R. Если она завершает-
ся успешно, то полученный результат считается результатом всей
конструкции. Если же вычисление хвоста R кончается неуспехом,
то вычисляются новые значения для переменных, входящих в He
(для этого вычисляется источник S′, причем при этом использу-
ются старые значения переменных). Далее делается новая попыт-
ка вычислить хвост R и т. д.

Таким образом, тропа-поиск как бы пытается подобрать для
переменных из He такие значения, при которых вычисление хво-
ста R успешно завершится.

Точный смысл конструкции поиска проще всего можно опре-
делить через смысл более простых конструкций: присваивания и
распутья. А именно, поиск S′′ $iter S′ :: He R эквивалентен
тропе

S′′ :: He, \{ R; S′ $iter S′ :: He R; }

Эта тропа опять содержит конструкцию поиска, которую можно
“развернуть” с помощью точно такого же преобразования. Полу-
чаем

S′′ :: He, \{ R;
S′ :: He, \{ R;

S′ $iter S′ :: He R;
};}

Повторив этот процесс бесконечное число раз, получаем, что ис-
ходная конструкция поиска эквивалентна бесконечной конструк-
ции

S′′ :: He, \{ R;
S′ :: He, \{ R;

S′ :: He, \{ R;
. . .
. . . };};}

Рассмотрим пример использования конструкции поиска —
определение функции факториал:

$func Fact sN = sFact;

Fact
{
0 = 1;
sN = <Mult sN <Fact <Sub sN 1>>>;
};

Недостаток этого рекурсивного определения в том, что проис-
ходит накопление вызовов функции Mult, ожидающих, пока ре-
курсия развернется до конца. Можно дать “более итеративное”
определение функции Fact (при этом потребуется вспомогатель-
ная функция FactAux).

$func Fact sN = sFact;
$func FactAux sR sK = sFact;

Fact sN =
<FactAux 1 sN>;

FactAux sR sK =
{
sK : 0

= sR;
= <FactAux <Mult sR sK> <Sub sK 1>>;

};

То же самое изображается с помощью конструкции поиска сле-
дующим образом:

$func Fact sN = sFact;

Fact sN =
1 sN

$iter <Mult sR sK> <Sub sK 1>
:: sR sK,

sK : 0,
= sR;

1.10 Перебор с возвратом

1.10.1 Задача о ферзях

Рассмотрим классическую задачу о восьми ферзях [Хен 1983].
Как известно, в шахматах ферзи атакуют друг друга по горизон-
тали, по вертикали или по диагоналям шахматной доски. Задача
состоит в том, чтобы найти такое размещение 8 ферзей на доске
8× 8, чтобы никакие два ферзя не атаковали друг друга.

Мы будем решать несколько более общую задачу, считая, что
требуется разместить n ферзей на доске размера n× n.

Занумеруем вертикали и горизонтали доски числами от 1 до
n. Будем говорить, что поле имеет координаты (i, j) или, другими
словами, является полем (i, j), если оно находится на пересечении
i-й вертикали и j-й горизонтали.

Заметим, что для полей любой диагонали, идущей слева на-
право и снизу вверх, сумма номеров столбцов и строк — посто-
янна, и что для полей любой диагонали, идущей слева направо
и сверху вниз, разность номеров столбцов и строк — постоянна
(рис. 1.2).

Таким образом, поля (I, J) и (I ′, J ′) лежат на одной диагонали,
если I + J = I ′ + J ′ или I − J = I ′ − J ′.

Это условие легко проверить. А именно, если вычисление тро-
пы

1 2 3 4 5 1 2 3 4 5
1 2 3 4 5 6
2 3 4 5 6 7
3 4 5 6 7 8
4 5 6 7 8 9
5 6 7 8 9 10

1 0 -1 -2 -3 -4
2 1 0 -1 -2 -3
3 2 1 0 -1 -2
4 3 2 1 0 -1
5 4 3 2 1 0

строка + столбец строка − столбец

Рис. 1.2: Номера диагоналей

\{
<Add sI sJ> :: sN1, <Add sI1 sJ1> :: sN2, sN1 : sN2;
<Sub sI sJ> :: sN1, <Sub sI1 sJ1> :: sN2, sN1 : sN2;
}

завершается успешно, поля (I, J) и (I ′, J ′) лежат на одной диаго-
нали.

Теперь надо выбрать способ, которым мы будем кодировать
расположение ферзей на доске.

1 2 3 4

4
3
2
1 r

r r
r

Рис. 1.3: Пример решения

Достаточно рассматривать только такие позиции, для которых
на одной вертикали находится не более одного ферзя, ибо любые
два ферзя, стоящие на одной вертикали, атакуют друг друга, и
такая позиция не может быть решением. С другой стороны, ко-
личество ферзей, которые следует расставить, равно количеству
вертикалей, из чего следует, что на каждой вертикали должен сто-
ять ровно один ферзь. Поэтому мы будем изображать позицию в

виде последовательности чисел

I1 I2 . . . In

где число Ik означает, что на k-й вертикали ферзь находится на
горизонтали с номером Ik. Например, позиция, изображенная на
рис. 1.3, кодируется выражением

3 1 4 2

Будем строить позицию постепенно, заполняя вертикали одну
за другой. При этом всякий раз, пытаясь поставить на доску оче-
редного ферзя, надо проверять, не атакован ли он одним из уже
стоящих на доске ферзей. Предположим, что на доске уже сто-
ит k ферзей на вертикалях 1, 2, . . . , k. Будем кодировать частично
построенную позицию как последовательность чисел

I1 I2 . . . Ik

где Im — номер горизонтали для ферзя на m-й вертикали.
Теперь мы можем определить предикат UnderAttack, который

выдает пустое выражение, если поле (I, J) атаковано ферзями,
уже стоящими на доске, и выдает неуспех, если это поле не ата-
ковано.

$func? UnderAttack sI sJ ePos = ;

UnderAttack sI sJ ePos =
ePos : $r eRest e, eRest : e sJ1,
<Length eRest> :: sI1,
\{
sI1 : sI;
sJ1 : sJ;
<Add sI sJ> :: sN1, <Add sI1 sJ1> :: sN2,

sN1 : sN2;
<Sub sI sJ> :: sN1, <Sub sI1 sJ1> :: sN2,

sN1 : sN2;
};

Отметим, что проверку I ′ = I можно и убрать, поскольку в
нашей программе функция UnderAttack будет вызываться толь-
ко таким образом, что параметр I будет заведомо больше, чем
номера столбцов у тех ферзей, которые уже поставлены на доску.

Теперь опишем функцию NextQueen, которая пытается доба-
вить к частично построенной позиции нового ферзя, перебирая
различные горизонтали. Если ферзя удается поставить, но этот
ферзь не последний, делается попытка поставить следующего
ферзя и т. д. Если очередного ферзя поставить не удается, про-
исходит возврат назад и делается попытка переставить на новое
место предыдущего ферзя.

$func? NextQueen sI sN ePos = ePos;

NextQueen sI sN ePos =
1 $iter \{ <Lt (sJ) (sN)> = <Add sJ 1>; }

:: sJ,
# <UnderAttack sI sJ ePos>,
ePos sJ :: ePos,
\? {
sI : sN

\! ePos;
\! <NextQueen <Add sI 1> sN ePos>;

};

Некоторые тонкости в определении функции NextQueen заслу-
живают пояснения.

Во-первых, конструкция поиска делает попытку вычислить
свое тело, последовательно пробуя для переменной J значения
1, 2, . . . , n и увеличивая значение переменной J на 1 после каж-
дой неудачной попытки вычислить свое тело.

Во-вторых, неудача при вычислении тела конструкции поиска
может возникнуть по двум причинам: либо поле (I, J) атаковано
ферзями, уже стоящими на доске, и тогда успешно завершится
вызов функции UnderAttack (а, значит, потерпит неудачу отри-
цание этого условия), либо очередного ферзя можно поставить на
поле (I, J), но не удается разместить следующих ферзей (и тогда

терпит неудачу рекурсивный вызов функции NextQueen).
Теперь осталось определить последнюю функцию Solution,

которая имеет один аргумент — размер доски, и выдает решение
задачи, если оно существует, либо терпит неудачу, если решение
не существует:

$func? Solution sN = ePos;

Solution sN =
<NextQueen 1 sN >;

1.10.2 Задача о цепочках:

Рассмотрим следующую задачу [Вир 1977]. Требуется найти
объектное выражение E , обладающее следующими свойствами:

• E не содержит скобок, а всякий входящий в него символ
является одним из символов 1, 2 или 3;

• E имеет заданную длину len;

• не существует таких объектных выражений Ea, Eb и Ec, что
Ec — не пусто и при этом

E = Ea Ec Ec Eb

т. е. E не содержит двух смежных непустых совпадающих
подвыражения.

Для построения требуемого выражения можно поступить сле-
дующим образом: начать с пустого выражения, а затем постепен-
но добавлять к нему числа, следя за тем, чтобы не получалось
запрещенной комбинации. При этом, чтобы гарантировать, что
цепочка не имеет вида Ea Ec Ec Eb, где Ec — непустое, достаточно
после добавления каждого числа проверять, что не образовалось
выражение вида

Ea Ec Ec
Для распознавания выражений такого вида определим преди-

кат IsUnacceptable.

$func? IsUnacceptable e.String = ;

IsUnacceptable e.String =
<Div <Length e.String> 2> :: s.Max,
{
s.Max : 0

= $fail;
= 1

$iter \{ <Lt (sK) (s.Max)> = <Add sK 1>; }
:: sK,
<Right 0 sK <Middle 0 sK e.String>> :: eU,
<Right 0 sK e.String> :: eV,
eU : eV;

};

Теперь определим функцию Extend, которая пытается доба-
вить к выражению очередной символ, пока цепочка не достигнет
заданной длины. Если цепочку продолжить невозможно, проис-
ходит возврат назад, и делается попытка изменить предыдущие
символы.

$func? Extend s.Len e.String = e.String;

Extend s.Len e.String =
{
<Length e.String> : s.Len

= e.String;
= 1 $iter

\{ <Lt (s.Digit) (3)> = <Add s.Digit 1>; }
:: s.Digit,

e.String s.Digit :: e.String,
# <IsUnacceptable e.String>,

<Extend s.Len e.String>;
};

Теперь осталось определить начальную функцию FindString,
которая получает в качестве параметра желаемую длину цепочки,

и выдает искомую цепочку (если она существует), либо терпит
неудачу (если она не существует).

$func? FindString s.Len = e.String;

FindString s.Len =
<Extend s.Len >;

1.11 Пример: компилятор для простого
императивного языка

В этом разделе мы рассматриваем методы написания компи-
ляторов на Рефале Плюс. Эти методы демонстрируются на при-
мере небольшого компилятора с простого императивного языка,
позаимствованного из [Уор 1980].

Хотя этот компилятор носит чисто учебный характер, он все
же значительно больше всех остальных программ, рассмотренных
в этой книге, и состоит из нескольких модулей.

1.11.1 Входной язык компилятора

Программа на входном языке представляет собой конечную
последовательность лексем. Каждая из лексем изображается с по-
мощью цепочки литер, синтаксис которой может быть описан по-
средством следующей грамматики (см. главу II, раздел 1):

$ Лексема =
$ КлючевоеСлово | Идентификатор | Число.
$ КлючевоеСлово =
$ ";" | "(" | ")" | "+" | "-" | "*" | "-" |
$ ":=" | "<=" | ’<>’ | "<" | ">=" | ">" | "=".
$ "DO" | "ELSE" | "IF" | "READ" | "THEN" |
$ "WHILE" | "WRITE".
$ Идентификатор = Буква { Буква | Цифра }.
$ Число = Цифра { Цифра }.

Ключевые слова являются зарезервированными, т. е. не разре-
шается использовать идентификаторы, совпадающие по начерта-
нию с ключевыми словами.

Прописные и строчные буквы в ключевых словах не различа-
ются, т. е. считаются полностью эквивалентными.

Соседние лексемы разделяются между собой “межевыми” ли-
терами: пробелами, табуляциями и литерами конца строки. В тех
случаях, когда нет опасности “склеивания” соседних лексем в од-
ну, разделять их не обязательно.

Не всякая последовательность лексем является синтаксически
правильной программой. Поэтому, после того как входная после-
довательность литер разбита на лексемы, следует проверить, что
полученная последовательность лексем имеет следующий синтак-
сис:

$ Программа = ПослОператоров.
$ ПослОператоров = Оператор { ";" Оператор }.
$ Оператор =
$ "IF" Условие "THEN" Оператор "ELSE" Оператор |
$ "WHILE" Условие "DO" Оператор |
$ "READ" ИмяПеременной |
$ "WRITE" Выражение |
$ "(" ПослОператоров ")".
$ ИмяПеременной ":=" Выражение |
$ Пусто.
$ Пусто = .
$ Условие = Выражение ОперОтношения Выражение.
$ ОперОтношения =
$ "=" | "<=" | "<>" | "<" | ">=" | ">".
$ Выражение = Слагаемое { АддитивнОпер Слагаемое}.
$ Слагаемое = Множитель { МультиплОпер Множитель}.
$ Множитель =
$ ИмяПеременной | Значение | "(" Выражение ")".
$ АддитивнОпер = "+" | "-".
$ МультиплОпер = "*" | "/".
$ ИмяПеременной = Идентификатор.
$ Значение = Целое.

Программа представляет собой последовательность операто-
ров, которые выполняются слева направо. Каждый из операторов
может использовать значения переменных и изменять значения
переменных.

Условный оператор

IF Cond THEN St1 ELSE St2

проверяет условие Cond. Затем, если оно выполнено, то выполня-
ется оператор St1, иначе — оператор St2.

Оператор цикла

WHILE Cond DO St

проверяет условие Cond. Если условие выполнено, то выполняется
оператор St, а затем все повторяется сначала. Если же условие не
выполнено, то выполнение оператора цикла завершается.

Оператор ввода

READ V ar

читает из устройства ввода одно число и присваивает его пере-
менной V ar.

Оператор вывода

WRITE Expr

вычисляет арифметическое выражение Expr и полученное число
записывает на устройство вывода.

Составной оператор

( St1; St2; . . . Stn )

означает, что следует последовательно выполнить операторы St1,
St2, . . . , Stn.

Оператор присваивания

V ar := Expr

вычисляет выражение Expr и полученное число присваивает пе-
ременной V ar.

Пустой оператор не выполняет никаких действий.
Условия и арифметические выражения имеют общепринятый

смысл. Операции умножения и деления имеют приоритет перед
операциями сложения и вычитания.

Переменные, которые используются в программе, описывать
не требуется. Перед началом работы программы их значения не
определены.

Ниже приведен пример программы, которая вводит одно чис-
ло, а затем вычисляет и выводит его факториал.

read value;
count:=1;
result:=1;
while count<value do

(
count:=count+1;
result:=result*count
);

write result

1.11.2 Выходной язык

Результатом работы компилятора является целевая програм-
ма в “машинном коде”, имеющая следующий синтаксис:

$ Программа = { Директива }.
$ Директива =
$ Инструкция | "BLOCK" "," Значение ";".
$ Инструкция =
$ КодОперации "," Значение ";" |
$ КодОперации =
$ ADD | SUB | MULT | DIV | LOAD | STORE |
$ ADDC | SUBC | MULTC | DIVC | LOADC |
$ JUMPEQ | JUMPNE |
$ JUMPLT | JUMPGT | JUMPLE | JUMPGE
$ JUMP | READ | WRITE | HALT.
$ Значение = Число.

Программа представляет собой последовательность директив,
а каждая директива является либо “инструкцией”, т. е. машинной
командой, либо директивой резервирования памяти.

Мы считаем, что оперативная память машины состоит из
ячеек, каждая из которых имеет адрес — целое положительное
число (таким образом, ячейки пронумерованы с 1). Каждая ячей-
ка может содержать либо одну инструкцию, либо одно целое чис-
ло.

Выполнение программы всегда начинается с первой ячейки.
Помимо оперативной памяти в машине имеется сумматор, ко-

торый может содержать одно число.
Директива

BLOCK,Int;

означает, что в этом месте программы следует оставить незаня-
тыми Int ячеек памяти. Обычно эта директива находится в конце
программы и используется для выделения ячеек памяти, предна-
значенных для хранения значений переменных.

Каждая инструкция имеет вид:

Op,V alue;

гдеOp— название выполняемой операции, а V alue— операнд этой
операции. Смысл операнда V alue определяется названием опера-
ции. Для некоторых операций V alue является адресом некото-
рой ячейки, которая используется при выполнении операции. Для
других операций V alue представляет собой целое число, которое
также используется при выполнении операции. Для некоторых
операций, однако, операнд не требуется, и в этом случае V alue
должно быть равно нулю.

Инструкция LOAD,Addr; загружает в сумматор содержимое
ячейки с адресом Addr. Инструкция STORE,Addr; записывает со-
держимое сумматора в ячейку с адресом Addr.

Инструкция LOADC,Int; загружает число Int в сумматор.
Инструкции ADD, SUB, MULT и DIV имеют вид Op,Addr; и вы-

числяют соответственно сумму, разность, произведение и целую

часть частного двух чисел. При этом первое число берется из сум-
матора, второе — из ячейки с адресом Addr, а результат записы-
вается в сумматор.

Инструкции ADDC, SUBC, MULTC и DIVC имеют вид Op,Int; и
вычисляют соответственно сумму, разность, произведение и це-
лую часть частного двух чисел. При этом первое число берется
из сумматора, второе число равно Int, а результат записывается
в сумматор.

Инструкция READ,Addr; читает с устройства ввода очеред-
ное число и помещает его в ячейку с адресом Addr. Инструк-
ция WRITE,0; записывает число, находящееся на сумматоре, на
устройство вывода.

Инструкция HALT,0; прекращает работу программы. Ин-
струкция JUMP,Addr; передает управление на ячейку с адресом
Addr.

И наконец, последняя группа инструкций — это условные пе-
реходы JUMPEQ, JUMPNE, JUMPLT, JUMPGT, JUMPLE и JUMPGE, кото-
рые имеют вид Op,Addr;. Выполняются они следующим образом.
Сначала содержимое сумматора сравнивается с нулем. Если про-
веряемое условие выполнено, то управление передается инструк-
ции с адресом Addr, если не выполнено — следующей инструкции.
Какую проверку выполнять, определяют две последние буквы в
имени инструкции. EQ означает проверку того, что содержимое
сумматора равно 0, NE — что оно не равно 0, LT — что оно меньше
0, GT — что оно больше 0, LE — что оно меньше или равно 0, GE —
что оно больше или равно 0.

Программа вычисления факториала, приведенная выше, будет
переведена компилятором в следующую программу в машинном
коде:

001 READ,21; 008 JUMPGE,16; 015 JUMP,6;
002 LOADC,1; 009 LOAD,19; 016 LOAD,20;
003 STORE,19; 010 ADDC,1; 017 WRITE,0;
004 LOADC,1; 011 STORE,19; 018 HALT,0;
005 STORE,20; 012 LOAD,20; 019 BLOCK,3;
006 LOAD,19; 013 MULT,19;
007 SUB,21; 014 STORE,20;

Слева от директив выписаны соответствующие им адреса.

1.11.3 Общая структура компилятора

Наш компилятор имеет “классическую” структуру и состоит
из следующих частей.

Входной поток литер, который часто называют также кон-
кретной программой, поступает в сканер, который разбивает его
на лексемы.

Затем поток лексем попадает в синтаксический анализатор,
который превращает этот поток в дерево синтаксического раз-
бора, которое часто называют также программой, записанной в
абстрактном синтаксисе, или абстрактной программой.

Далее абстрактная программа попадает в генератор кода, ре-
зультатом работы которого является программа на языке ассем-
блера. Ассемблерная программа, по сути, является уже почти го-
товой программой в машинном коде, однако, вместо конкретных
адресов ячеек она содержит метки, каждая из которых изобра-
жает некоторый еще неизвестный адрес.

Полученная ассемблерная программа затем поступает в ассем-
блер, который заменяет все метки на конкретные адреса, в резуль-
тате чего получается окончательная программа в машинном коде.

Информация о соответствии между именами переменных и
метка- ми хранится в словаре переменных. Таким образом, в ком-
пиляторе имеется модуль, который обеспечивает работу со слова-
рем. Этот модуль используется при генерации кода и при ассем-
блировании программы.

Может сложиться впечатление, что данный компилятор име-
ет неоправданно сложную структуру, не соответствующую про-
стоте входного языка. И действительно, его вполне можно было
бы упростить за счет объединения многих компонент. Например,
можно было бы делать сканирование программы, синтаксический
анализ и генерацию кода одновременно. Однако, не следует забы-
вать, что в случае более сложного входного языка такой подход
мог бы привести к запутанному и ненадежному компилятору, в
котором было бы очень трудно разобраться. Наш же компилятор

служит только в качестве учебного примера по программирова-
нию на Рефале Плюс и его назначение — продемонстрировать,
как Рефал Плюс позволяет применить классические методы по-
строения компиляторов. Взяв этот пример за основу, читатель
может попробовать развить его в двух направлениях. Во-первых,
в сторону более реалистического и сложного входного языка, во-
вторых — в сторону упрощения компилятора за счет устранения
из него чрезмерной “научности” и общности подхода.

1.11.4 Модули компилятора и их интерфейсы

Компилятор состоит из следующих модулей:

Cmp − головная программа
CmpScn − сканер
CmpPrs − синтаксический анализатор
CmpGen − генератор кода и ассемблер
CmpDic − работа со словарем

Головной модуль не имеет интерфейсной части и содержит опре-
деление начальной функции Main. Прочие модули состоят из двух
частей: интерфейса модуля и реализации модуля.

Модуль CmpScn имеет следующую интерфейсную часть:

//
// File CmpScn.rfi
//

$func InitScanner s.Channel = ;
$func ReadToken = s.TokenClass s.TokenInfo;
$func TermScanner = ;

Из него экспортируются три функции.
Функция InitScanner служит для инициализации сканера.

Параметр s.Channel является ссылкой на канал, из которо-
го сканер начинает читать литеры. Прежде чем обращаться к
InitScanner, следует открыть этот канал на чтение.

Функция TermScanner должна быть вызвана после того как
вся исходная программа будет прочитана. Это дает возможность

сканеру закончить свою деятельность и приготовиться к чтению
другой программы.

Функция ReadToken выдает очередную лексему исходной про-
граммы в виде двух символов: первый символ указывает, к какому
классу принадлежит лексема, а второй символ содержит допол-
нительную информацию о лексеме.

Модуль CmpPrs имеет следующую интерфейсную часть:

//
// File: CmpPrs.rfi
//

$func Parse s.Channel = t.Program;

Из него экспортируется одна функция Parse, которая читает
(обращаясь к сканеру) исходную программу из канала s.Channel
и выдает абстрактную программу t.Program. Прежде чем обра-
щаться к Parse, следует открыть канал s.Channel на чтение.

Если исходная программа содержит синтаксические ошибки,
функция Parse выдает $error(E), где E — сообщение об ошибке.

Модуль CmpGen имеет следующую интерфейсную часть:

//
// File: CmpGen.rfi
//

$func GenCode t.Program = t.Code;
$func WriteCode t.Code = ;

Из него экспортируются две функции.
Функция GenCode получает абстрактную программу

t.Program и выдает скомпилированную программу в ма-
шинном коде t.Code, во внутреннем представлении в виде
синтаксического дерева.

Функция WriteCode получает программу в машинном коде в
виде синтаксического дерева и превращает ее в поток литер, ко-
торый выводит на стандартное устройство вывода.

Модуль CmpDic имеет следующую интерфейсную часть:

//
// File: CmpDic.rfi
//

$func MakeDic = s.Dic;
$func LookupDic s.Key s.Dic = s.Ref;
$func AllocateDic s.Dic s.StartAddr = s.FreeAddr;

Из него экспортируются четыре функции.
Функция MakeDic создает новый пустой словарь и выдает

ссылку на него.
Функция LookupDic находит в словаре, на который указывает

s.Dic, метку, соответствующую ключу s.Key и выдает ее. Если
ключ s.Key ранее не был зарегистрирован в словаре, то создается
новая уникальная метка, которая связывается с ключом s.Key и
выдается в качестве значения.

Функция AllocateDic просматривает словарь, на который
указывает s.Dic и связывает все зарегистрированные в нем мет-
ки с различными адресами. Если в словаре зарегистрировано N
ключей, то соответствующим меткам присваиваются последова-
тельные адреса, начиная с s.StartAddr. В качестве результата
функция выдает первый свободный адрес.

1.11.5 Головной модуль компилятора

Головной модуль компилятора связывает все части компиля-
тора между собой. Предполагается, что имя файла, в котором на-
ходится исходная программа, передается компилятору через пер-
вый параметр в командной строке. Таким образом, компилятор
вызывается следующим образом:

Cmp FileName

где FileName — имя файла. Это имя запрашивается с помощью
библиотечной функции Arg.

//
// File Cmp.rf

//

$use Dos StdIO;
$use CmpPrs CmpGen;

$func Main = e;
$func Compile e.FileName = ;

Main =
<Arg 1> :: e.FileName,
<Compile e.FileName>;

Compile e.FileName =
<Channel> :: s.Chl,
<OpenFile s.Chl e.FileName "r">,
<Parse s.Chl> :: t.AProgram,
<CloseChannel s.Chl>,
<GenCode t.AProgram> :: t.Code,
<WriteCode t.Code>;

1.11.6 Сканер

Результатом работы сканера является поток лексем, каждая
из которых представлена двумя символами. Первый символ ука-
зывает на класс лексемы. В дальнейшем мы будем описывать
синтаксис объектных выражений с помощью расширенной БНФ,
нетерминалы которой записаны в виде переменных Рефала Плюс.
При этом мы предполагаем, что значения нетерминалов должны
соответствовать их типам. Таким образом, лексемы, поступающие
из сканера, имеют следующий синтаксис:

$ e.Tokens = { e.Token }.
$ e.Token =
$ Key s.Key | Name s.Name | Value s.Value |
$ Char s.Char.
$ s.Key = s.Word.
$ s.Name = s.Word.

$ s.Value = s.Int.

Лексема вида Key s.Key изображает ключевое слово, где
s.Key — символ-слово, печатное изображение которого соответ-
ствует ключевому слову. Лексема вида Name s.Name изображает
имя переменной, где s.Name — символ-слово, печатное изображе-
ние которого соответствует идентификатору — имени перемен-
ной. Лексема вида Value s.Value изображает числовую констан-
ту, где s.Value — соответствующий символ-число. Лексема вида
Char s.Char изображает неопознанную литеру s.Char.

Когда исходная программа прочитана до конца, сканер выдает
в конце лексему Key Eof.

Модуль CmpScn имеет следующую реализацию:

//
// File: CmpScn.rf
//

$use StdIO Class Convert Box;

$func ScanToken s.Chl e.Line =
s.TokenKey s.TokenInfo (e.Line1);

$func ScanIdRest (e.IdChars) e.Chars =
s.TokenKey s.Word (e.Rest);

$func ScanIntRest (e.IntChars) e.Chars =
s.TokenKey s.Int (e.Rest);

$func? IsBlank s.Char = ;
$func? IsOneCharToken s.Char = ;
$func? CompoundToken s.Char e.Line = s.Word e.Rest;
$func? IsKeyWord s.Word = ;

// Ящики для хранения канала, из которого читаются
// лексемы, и остатка очередной строки.

$box ScanChl ScanLine;

InitScanner s.Chl = // Инициализация сканера.

<Store &ScanChl s.Chl>, // Прячем канал в ящик.
<Store &ScanLine >; // Текущая строка - пустая.

TermScanner = // Завершение работы.
<Store &ScanChl >, // Забываем канал
<Store &ScanLine >; // и текущую строку.

ReadToken = // Чтение лексемы.
<Get &ScanChl> : s.Chl,
<Get &ScanLine> :: e.Line,
<ScanToken s.Chl e.Line>

:: s.TokenKey s.TokenInfo (e.Line),
<Store &ScanLine e.Line>,

= s.TokenKey s.TokenInfo;

ScanToken s.Chl e.Line =
e.Line :
{
= // Остаток строки -

{ // пустой. Читаем
<ReadLineCh s.Chl> :: e.Line // следующую.

= <ScanToken s.Chl e.Line>;
= Key Eof (); // Конец файла.

};
s.Char e.Rest = // Изучаем очередную

{ // литеру.
<IsBlank s.Char>

= <ScanToken s.Chl e.Rest>;
<IsLetter s.Char>

= <ScanIdRest (s.Char) e.Rest>;
<IsDigit s.Char>

= <ScanIntRest (s.Char) e.Rest>;
<IsOneCharToken s.Char>

= Key <ToWord s.Char> (e.Rest);
<CompoundToken s.Char e.Rest> :: s.Word e.Rest

= Key s.Word (e.Rest);

= Char s.Char (e.Rest); // Неопознанная литера.
};

};

// Отщепляем остаток идентификатора.

ScanIdRest (e.IdChars) e.Rest =
{
e.Rest : s.Char e.Rest1,

\{<IsLetter s.Char>; <IsDigit s.Char>;}
= <ScanIdRest (e.IdChars s.Char) e.Rest1>;
= <ToWord <ToUpper e.IdChars>> : s.Word,

{<IsKeyWord s.Word> = Key; = Name;} :: s.TokenKey,
= s.TokenKey s.Word (e.Rest);

};

// Отщепляем остаток целого числа.

ScanIntRest (e.IntChars) e.Rest =
{
e.Rest : s.Char e.Rest1, <IsDigit s.Char>

= <ScanIntRest (e.IntChars s.Char) e.Rest1>;
= Value <ToInt e.IntChars> (e.Rest);

};

IsBlank s.Char = // "Межевая" литера?
’ \n\t’ : e s.Char e;

IsOneCharToken s.Char = // Однолитерная лексема?
’;()+-*/’ : e s.Char e;

CompoundToken // Пытаемся отщепить
\{ // многолитерную лексему.
’:=’ e.Rest = ":=" e.Rest;
’<=’ e.Rest = "<=" e.Rest;
’<>’ e.Rest = "<>" e.Rest;

’<’ e.Rest = "<" e.Rest;
’>=’ e.Rest = ">=" e.Rest;
’>’ e.Rest = ">" e.Rest;
’=’ e.Rest = "=" e.Rest;
};

IsKeyWord // Идентификатор - ключевое слово?
\{
DO ; ELSE ; IF ; READ ; THEN ; WHILE ; WRITE ;
};

1.11.7 Синтаксический анализатор

Задача синтаксического анализатора, находящегося в модуле
CmpPrs, — преобразовать поток лексем в абстрактную программу,
т. е. в дерево грамматического разбора.

Для выполнения синтаксического анализа мы используем ме-
тод нисходящего грамматического разбора.

Рассмотрим, для примера, следующую грамматику:

$ Sentence = Subject Predicate.
$ Subject = "cats" | "dogs".
$ Predicate = "sleep" | "eat".

Пусть нам дана последовательность лексем

"dogs" "eat"

и мы хотим узнать, является ли она правильным предложением.
Для этого мы должны узнать, может ли эта цепочка быть порож-
дена нетерминалом Sentence (“предложение”). Но из граммати-
ки следует, что множество цепочек, порождаемых нетерминалом
Sentence, совпадает с множеством цепочек, порождаемых цепоч-
кой двух нетерминалов Subject Predicate. Таким образом, зада-
ча сводится к тому, чтобы проверить, нельзя ли от входной цепоч-
ки отщепить начало, которое порождается нетерминалом Subject
(“подлежащее”), а затем проверить, что остаток цепочки может
быть порожден нетерминалом Predicate (“сказуемое”).

Как отщепить от цепочки начало, порождаемое нетерминалом
Subject? Очевидно, что для этого достаточно проверить, что це-
почка лексем начинается с лексемы "cats" или лексемы "dogs".

Таким образом, мы приходим к следующему методу анализа.
Каждому нетерминалу A в грамматике мы ставим в соответ-

ствие функцию A, имеющую следующее объявление:

$func? A e.Token = e.Rest;

Эта функция проверяет, что входная цепочка символов e.Token
начинается с цепочки символов, выводимой из нетерминала A. Ес-
ли это действительно так, то функция A отбрасывает от e.Token
ее начало, выводимое из нетерминала A и выдает то, что останет-
ся, в качестве результата. Если же от e.Token невозможно отще-
пить начальную цепочку, выводимую из нетерминала A, то функ-
ция A выдает в качестве результата неуспех.

Конечно, описанный метод синтаксического анализа подходит
только в том случае, если для любого нетерминала A и любой
входной цепочки Z существует не более чем один способ отщепить
от X цепочку, выводимую из A. Во многих случаях, тем не ме-
нее, грамматика может быть преобразована к такому виду, чтобы
это ограничение выполнялось. За более подробным обсуждением
этого вопроса мы отсылаем читателя к [Вир 1985].

Руководствуясь вышеизложенными соображениями, мы мо-
жем определить функцию Sentence, отщепляющую от цепоч-
ки символов начальную цепочку, выводимую из нетерминала
Sentence, либо терпящую неуспех, если это невозможно.

$func? Sentence e.Token = e.Rest;
$func? Subject e.Token = e.Rest;
$func? Predicate e.Token = e.Rest;
$func? Token s e.Token = e.Rest;

Sentence eZ =
<Subject eZ> :: eZ,
<Predicate eZ> :: eZ,

= eZ;

Subject eZ =
\{
<Token "dogs" eZ> :: eZ = eZ;
<Token "cats" eZ> :: eZ = eZ;
};

Predicate eZ =
\{
<Token "sleep" eZ> :: eZ = eZ;
<Token "eat" eZ> :: eZ = eZ;
};

Token s eZ =
eZ : s eZ0

= eZ0;

Функция Token введена для отщепления произвольных терми-
нальных символов.

Теперь мы можем вернуться к рассмотрению модуля CmpPrs.
Здесь мы сталкиваемся с двумя дополнительными проблемами.

Во-первых, исходная программа поступает из сканера не вся
целиком, а постепенно, лексема за лексемой. Поэтому аргумен-
том каждой из функций грамматического анализа является не
вся цепочка лексем, а только одна лексема, которая была прочи-
тана последней. Эта лексема является первой лексемой из еще не
проанализированного остатка программы. Точно так же, функ-
ции анализа выдают не весь остаток программы, а только первую
лексему из остатка программы. При этом каждая лексема пред-
ставлена не одним символом, а двумя.

Во-вторых, синтаксический анализатор должен не только про-
верять правильность исходной программы, но еще и преобразовы-
вать поток лексем в абстрактную программу, т. е. в дерево грам-
матического разбора. Поэтому функция анализа, соответствую-
щая нетерминалу A, как правило, имеет следующий формат:

$func A sC sI = sC sI tX;

где sC sI соответствует очередной лексеме, а tX — результат пе-
ревода отщепленной части программы в абстрактный синтаксис.

В третьих, если обнаруживается синтаксическая ошибка, син-
таксический анализатор должен выдавать не просто неуспех, а
ошибку вида $error(E), где E — сообщение, как-то описывающее
ошибку. Поэтому функции анализа объявлены как безоткатные.

Абстрактная программа имеет следующий синтаксис:

$ t.Program = (Program t.Statement).
$ t.Statement =
$ (Assign s.Name t.Expr) |
$ (If t.Test t.Statement t.Statement) |
$ (While t.Test t.Statement) |
$ (Read s.Name) |
$ (Write t.Expr) |
$ (Seq t.Statement t.Statement)
$ (Skip).
$ t.Test = (Test s.CompOper t.Expr t.Expr).
$ t.Expr =
$ (Const s.Value) |
$ (Name s.Name) |
$ (Op t.Oper t.Expr t.Expr).
$ s.CompOper = Eq | Ne | Gt | Ge | Lt | Le.
$ s.Oper = Add | Sub | Div | Mult.
$ s.Name = s.Word.
$ s.Value = s.Int.

Таким образом, любая конструкция в абстрактном синтаксисе,
как правило, принимает следующий вид:

(SkeyWord T1 T2 . . . Tn)

где ключевое слово SkeyWord — это символ-слово, являющееся на-
званием конструкции, а объектные термы T1, T2, . . . , Tn — вло-
женные конструкции, тоже представленные в абстрактном син-
таксисе. Соответствие между конструкциями в конкретном и аб-
страктном синтаксисе очевидно, и мы не будем останавливаться
на этом вопросе более подробно.

Ниже приведена реализация модуля CmpPrs:

//
// File: CmpPrs.rf
//

$use CmpScn;

$func Program sC sI = sC sI tX;
$func StatementSeq sC sI = sC sI tX;
$func RestStSeq sC sI tX0 = sC sI tX;
$func Statement sC sI = sC sI tX;
$func Test sC sI = sC sI tX;
$func Expr sC sI = sC sI tX;
$func RestExpr sC sI tX1 = sC sI tX;
$func Term sC sI = sC sI tX;
$func RestTerm sC sI tX1 = sC sI tX;
$func Factor sC sI = sC sI tX;
$func CompOp sC sI = sC sI s.CompOper;
$func? AddOp sC sI = sC sI s.Oper;
$func? MultOp sC sI = sC sI s.Oper;
$func? Token sX sC sI = sC sI ;
$func Accept sX sC sI = sC sI ;
$func? Name sC sI = sC sI s.Name;
$func? Value sC sI = sC sI s.Value;

Parse s.Chl =
<InitScanner s.Chl>,
<Program <ReadToken>> :: sC sI t.Program,
<TermScanner>,
{ // Проверяем, что остаток программы
sC sI : Key Eof // пуст.

= t.Program;
= $error sC sI " instead of Eof after the program";

};

Program sC sI = // Программа.
<StatementSeq sC sI> :: sC sI tX,

= sC sI (Program tX);

StatementSeq sC sI = // Последовательность
<Statement sC sI> :: sC sI tX0, // операторов.

= <RestStSeq sC sI tX0>;

RestStSeq sC sI tX0 =
\? {
<Token ";" sC sI> :: sC sI \!

<StatementSeq sC sI> :: sC sI tX,
= sC sI (Seq tX0 tX);

\!
= sC sI tX0;

};

Statement sC sI = // Оператор.
\? {
<Name sC sI> :: sC sI s.Name \!

<Accept ":=" sC sI> :: sC sI,
<Expr sC sI> :: sC sI t.Expr,

= sC sI (Assign s.Name t.Expr);
<Token "IF" sC sI> :: sC sI \!

<Test sC sI> :: sC sI t.Test,
<Accept "THEN" sC sI> :: sC sI,
<Statement sC sI> :: sC sI t.Then,
<Accept "ELSE" sC sI> :: sC sI,
<Statement sC sI> :: sC sI t.Else,

= sC sI (If t.Test t.Then t.Else);
<Token "WHILE" sC sI> :: sC sI \!

<Test sC sI> :: sC sI t.Test,
<Accept "DO" sC sI> :: sC sI,
<Statement sC sI> :: sC sI t.Do,

= sC sI (While t.Test t.Do);
<Token "READ" sC sI> :: sC sI \!

<Name sC sI> :: sC sI s.Name,
= sC sI (Read s.Name);

<Token "WRITE" sC sI> :: sC sI \!
<Expr sC sI> :: sC sI t.Expr,

= sC sI (Write t.Expr);
<Token "(" sC sI> :: sC sI \!

<StatementSeq sC sI> :: sC sI t.Stmt,
<Accept ")" sC sI> :: sC sI,

= sC sI t.Stmt;
\!

= sC sI (Skip);
};

Test sC sI = // Условие.
<Expr sC sI> :: sC sI t.Expr1,
<CompOp sC sI> :: sC sI t.Op,
<Expr sC sI> :: sC sI t.Expr2,

= sC sI (Test t.Op t.Expr1 t.Expr2);

Expr sC sI = // Выражение.
<Term sC sI> :: sC sI t.X0,

= <RestExpr sC sI t.X0>;

RestExpr sC sI t.X1 =
\? {
<AddOp sC sI> :: sC sI s.Op \!

<Term sC sI> :: sC sI t.X2,
= <RestExpr sC sI (Op s.Op t.X1 t.X2)>;

\!
= sC sI t.X1;

};

Term sC sI = // Слагаемое.
<Factor sC sI> :: sC sI t.X0,

= <RestTerm sC sI t.X0>;

RestTerm sC sI t.X1 =
\? {
<MultOp sC sI> :: sC sI s.Op \!

<Factor sC sI> :: sC sI t.X2,
= <RestTerm sC sI (Op s.Op t.X1 t.X2)>;

\!
= sC sI t.X1;

};

Factor sC sI = // Множитель.
\? {
<Name sC sI> :: sC sI s.Name \!

= sC sI (Name s.Name);
<Value sC sI> :: sC sI s.Value \!

= sC sI (Const s.Value);
<Token "(" sC sI> :: sC sI \!

<Expr sC sI> :: sC sI t.Expr,
<Accept ")" sC sI> :: sC sI,

= sC sI t.Expr;
\!

$error "Invalid factor start: " sC sI;
};

CompOp sC sI = // Операция отношения.
{
sC : Key,
("=" Eq) ("<>" Ne)
("<=" Le) ("<" Lt) (">=" Ge) (">" Gt)
: e (sI s.Op) e

= <ReadToken> s.Op;
= $error "Invalid comparison operation: " sC sI;

};

AddOp Key sI = // Аддитивная операция.
("+" Add) ("-" Sub) : e (sI s.Op) e

= <ReadToken> s.Op;

MultOp Key sI = // Мультипликативная операция.
("*" Mult) ("/" Div) : e (sI s.Op) e

= <ReadToken> s.Op;

// Эта функция пытается отщепить ключевое слово sI.
// Если это не удается, то результатом является неуспех.

Token sI Key sI = <ReadToken>;

// Эта функция пытается отщепить ключевое слово sI.
// Если это не удается, то результатом является ошибка.

Accept
{
sI Key sI = <ReadToken>;
sX sC sI = $error sC sI " instead of " Key sX;
};

// Имя переменной.

Name Name sI = <ReadToken> sI;

// Значение.

Value Value sI = <ReadToken> sI;

1.11.8 Генератор кода

При генерации машинного кода порождается его промежуточ-
ное представление в виде объектного терма, имеющее следующий
синтаксис:

$ t.Code =
$ (Seq { t.Code } ) |
$ (Instr s.Instr s.Operand) |
$ (Label s.Label) |

$ (Block s.Value).
$ s.Operand = s.Label | s.Value.
$ s.Label = s.Box.
$ s.Value = s.Int.
$
$ s.Instr =
$ ADD | SUB | DIV | MULT | LOAD | STORE |
$ ADDC | SUBC | DIVC | MULTC | LOADC |
$ JUMPEQ | JUMPNE |
$ JUMPLT | JUMPGT | JUMPLE | JUMPGE
$ JUMP | READ | WRITE | HALT |

Промежуточное представление может содержать метки, кото-
рые впоследствии, после ассемблирования, заменяются на адреса.
Ассемблирование выполняется в два прохода. На первом прохо-
де выясняются адреса всех инструкций и переменных, при этом
адреса, соответствующие меткам, предварительно помещаются в
ящики, на которые указывают метки. На втором проходе метки
заменяются на их значения, т. е. ссылки на ящики заменяются не
содержимое ящиков.

Модуль CmpGen имеет следующую реализацию:

//
// File: CmpGen.rf
//

$use StdIO Class Arithm Box;
$use CmpDic;

$func EncProgram t.Program s.Dic = t.Code;
$func EncSt t.St s.Dic = t.Code;
$func EncTest t.Test s.Label s.Dic = t.TestC;
$func UnlessOp s.Op = s.JumpIf;
$func EncExpr t.Expr s.Dic = t.ExprC;
$func EncSubExpr t.Expr sN s.Dic = t.ExprC;
$func LiteralOp s.Op = s.OpCode;
$func MemoryOp s.Op = s.OpCode;

$func Assemble t.Code s.StartAddr = s.FreeAddr;
$func AssembleSeq e.CodeSeq s.Addr = s.FreeAddr;
$func Dereference t.Code = t.Target;
$func DereferenceSeq e.CodeSeq = e.CodeSeqD;
$func WriteCodeSeq e.CodeSeq = ;

// Генерация кода из абстрактной программы.

GenCode t.Program =
// Создается пустой словарь.

<MakeDic> :: s.Dic,
// Компилируется абстрактная программа.

<EncProgram t.Program s.Dic> :: t.Code,
// Распределяется память под инструкции.

<Assemble t.Code 1> :: s.FreeAddr,
// Распределяется память под переменные.

<AllocateDic s.Dic s.FreeAddr> :: s.EndAddr,
// Метки заменяются на адреса.

<Dereference t.Code> :: t.CodeD,
// Генерируется директива BLOCK.

<Sub s.EndAddr s.FreeAddr> :: s.BlockLength,
(Seq t.CodeD (Block s.BlockLength)) :: t.Target,

= t.Target;

// Компиляция программы.

EncProgram (Program t.St) s.Dic =
<EncSt t.St s.Dic> :: t.StC,
<Box> :: s.L,

= (Seq t.StC (Instr HALT 0) (Label s.L));

// Компиляция оператора.

EncSt (s.KeyWord e.Info) s.Dic =
(s.KeyWord e.Info) :
{

(Assign sX t.Expr) =
<LookupDic sX s.Dic> :: s.Addr,
<EncExpr t.Expr s.Dic> :: t.ExprC,

= (Seq t.ExprC (Instr STORE s.Addr));
(If t.Test t.Then t.Else) =

<Box> :: s.L1, <Box> :: s.L2,
<EncTest t.Test s.L1 s.Dic> :: t.TestC,
<EncSt t.Then s.Dic> :: t.ThenC,
<EncSt t.Else s.Dic> :: t.ElseC,

= (Seq
t.TestC
t.ThenC
(Instr JUMP s.L2)

(Label s.L1)
t.ElseC

(Label s.L2)
);

(While t.Test t.Do) =
<Box> :: s.L1, <Box> :: s.L2,
<EncTest t.Test s.L2 s.Dic> :: t.TestC,
<EncSt t.Do s.Dic> :: t.DoC,

= (Seq
(Label s.L1)

t.TestC
t.DoC
(Instr JUMP s.L1)

(Label s.L2)
);

(Read s.X) =
<LookupDic s.X s.Dic> :: s.Addr,

= (Instr READ s.Addr);
(Write t.Expr) =

<EncExpr t.Expr s.Dic> :: t.ExprC,
= (Seq t.ExprC (Instr WRITE 0));

(Seq t.St1 t.St2) =
<EncSt t.St1 s.Dic> :: t.StC1,

<EncSt t.St2 s.Dic> :: t.StC2,
= (Seq t.StC1 t.StC2);

(Skip) =
= (Seq );

};

// Компиляция условия.

EncTest (Test s.Op t.Arg1 t.Arg2) s.Label s.Dic =
<EncExpr (Op Sub t.Arg1 t.Arg2) s.Dic> :: t.ExprC,
<UnlessOp s.Op> :: s.JumpIf,

= (Seq t.ExprC (Instr s.JumpIf s.Label));

UnlessOp // Генерация операции перехода.
{
Eq = JUMPNE; Ne = JUMPEQ;
Lt = JUMPGE; Gt = JUMPLE;
Le = JUMPGT; Ge = JUMPLT;
};

// Компиляция арифметического выражения.
// При этом происходит выделение рабочих переменных
// для хранения результатов вычисления подвыражений.
// При компиляции бинарных операций выбирается такой
// порядок вычислений, который позволяет уменьшить
// количество используемых рабочих ячеек.

EncExpr t.Expr s.Dic
= <EncSubExpr t.Expr 0 s.Dic>;

EncSubExpr (s.KeyWord e.Info) sN s.Dic =
(s.KeyWord e.Info) :
{
(Const sC) =

= (Instr LOADC sC);
(Name sX) =

<LookupDic sX s.Dic> :: s.Addr,
= (Instr LOAD s.Addr);

(Op s.Op t.Expr1 t.Expr2) =
t.Expr2 :
{
(Const sC2) =

<EncSubExpr t.Expr1 sN s.Dic> :: t.Expr1C,
<LiteralOp s.Op> :: s.OpCode,

= (Seq t.Expr1C (Instr s.OpCode sC2));
(Name sX2) =

<EncSubExpr t.Expr1 sN s.Dic> :: t.Expr1C,
<MemoryOp s.Op> :: s.OpCode,
<LookupDic sX2 s.Dic> :: s.Addr,

= (Seq t.Expr1C (Instr s.OpCode s.Addr));
(Op e) =

<LookupDic sN s.Dic> :: s.Addr,
<EncSubExpr t.Expr2 sN s.Dic> :: t.Expr2C,
<Add sN 1> :: sN1,
<EncSubExpr t.Expr1 sN1 s.Dic> :: t.Expr1C,
<MemoryOp s.Op> :: s.OpCode,

= (Seq
t.Expr2C
(Instr STORE s.Addr)
t.Expr1C
(Instr s.OpCode s.Addr)

);
};

};

LiteralOp // Генерация имен инструкций
{ // с непосредственным операндом.
Add = ADDC; Sub = SUBC;
Mult = MULTC; Div = DIVC;
};

MemoryOp // Генерация имен инструкций

{ // с операндом-адресом.
Add = ADD; Sub = SUB;
Mult = MULT; Div = DIV;
};

// Распределение памяти под инструкции.

Assemble t.Code s.A0 =
t.Code :
{
(Seq e.CodeSeq) =

= <AssembleSeq e.CodeSeq s.A0>;
(Instr s s) =

= <Add s.A0 1>;
(Label s.Label) =

<Store s.Label s.A0>
= s.A0;

};

AssembleSeq e.CodeSeq s.A0 =
e.CodeSeq :
{
t.Code e.Rest =

<Assemble t.Code s.A0> :: s.A1,
= <AssembleSeq e.Rest s.A1>;

=
= s.A0;

};

// Замена меток на их значения.

Dereference t.Code =
t.Code :
{
(Seq e.CodeSeq) =

(Seq <DereferenceSeq e.CodeSeq>);

(Instr s.Instr s.Value) =
{
<IsInt s.Value>

= t.Code;
<IsBox s.Value>

= (Instr s.Instr <Get s.Value>);
};

(Label s.Label) =
(Label <Get s.Label>);

};

DereferenceSeq
{
t.Code e.CodeSeq =

<Dereference t.Code><DereferenceSeq e.CodeSeq>;
= ;
};

// Преобразование машинного кода в поток литер
// и вывод на стандартное устройство вывода.

WriteCode
{
(Seq e.CodeSeq) =

<WriteCodeSeq e.CodeSeq>;
(Instr s.Instr s.Value) =

<Print " "><Print s.Instr><Print ",">
<Print s.Value><Print ";\n">;

(Label s.Label) =
<Print s.Label><Print ":\n">;

(Block s.Value) =
<Print " BLOCK,"><Print s.Value><Print ";\n">;

};

WriteCodeSeq
{

103



t.Code e.CodeSeq =
<WriteCode t.Code><WriteCodeSeq e.CodeSeq>;

= ;
};

1.11.9 Модуль работы со словарем

Словари реализованы в виде бинарных деревьев [АХУ 1979].
Каждый узел дерева представлен ящиком, который содержит три
символа: ключ, значение ключа, ссылки на левое поддерево и
ссылки на правое поддерево. Пустое дерево представляется ссыл-
кой на пустой ящик.

Модуль CmpDic имеет следующую реализацию:

//
// File: CmpDic.rf
//

$use Box Compare Arithm;

// Создание нового пустого словаря.

MakeDic = <Box>;

// Поиск в словаре s.Dic метки, связанной с ключом s.Key.
// Если ключа s.Key еще нет в словаре, он заносится
// в словарь и связывается с новой уникальной меткой.

LookupDic s.Key s.Dic =
<Get s.Dic> :
{
=

<Box> :: s.Ref,
<Store s.Dic s.Key s.Ref <Box> <Box>>,

= s.Ref;
s.Key1 s.Ref1 s.DicL s.DicR =

<Compare (s.Key)(s.Key1)> :

{
’<’ = <LookupDic s.Key s.DicL>;
’>’ = <LookupDic s.Key s.DicR>;
’=’ = s.Ref1;
};

};

// Распределение памяти под метки, находящиеся в словаре.
// s.A - начальный адрес. Адреса, соответствующие меткам,
// заносятся в ящики, на которые указывают метки.

AllocateDic s.Dic s.A =
<Get s.Dic> :
{
= s.A;
s.Key s.Ref s.DicL s.DicR =

<AllocateDic s.DicL s.A> :: s.A,
<Store s.Ref s.A>,
<Add s.A 1> :: s.A,

= <AllocateDic s.DicR s.A>;
};

$func WriteDic s = ;

WriteDic s.Dic =
<Get s.Dic> :
{
= <Print "_">;
s.Key s.Ref s.DicL s.DicR =

<Print "(">, <WriteDic s.DicL>, <Print " ">,
<Write s.Key>,
<Get s.Ref> :

{
= ;
e.Value = <Print "->"> <Write e.Value>;
},

105



<Print " ">,
<WriteDic s.DicR>, <Print ")">;

};

Глава 2

Синтаксис и семантика
Рефала Плюс

2.1 Нотация для записи синтаксиса

Для описания синтаксиса используется расширенная форма
Бекуса-Наура (РБНФ).

Синтаксические понятия (нетерминалы) обозначаются слова-
ми, выражающими их интуитивный смысл. Цепочки терминаль-
ных символов заключаются в двойные кавычки либо апострофы,
с тем, чтобы их можно было отличить от нетерминалов.

Если конструкция A представляет собой конкатенацию кон-
струкций B и C, т. е. состоит из конструкции B и следующей
вслед за ней конструкции C, мы будем называть B и C синтак-
сическими множителями и описывать A следующей синтаксиче-
ской формулой:

A = B C.

Если конструкция A является либо конструкцией B, либо кон-
струкцией C, мы будем называть B и C синтаксическими слага-
емыми и описывать A следующей синтаксической формулой:

A = B | C.

Для группировки синтаксических слагаемых и множителей
можно использовать круглые скобки.

Если конструкция A представляет собой либо конструкцию B,
либо является пустой цепочкой, это выражается в виде

A = [B].

Если же A представляет собой произвольную (может быть пу-
стую) последовательность из конструкций B, это выражается в
виде

A = {B}.

Приведем несколько примеров того, как множества предложе-
ний описываются формулами в РБНФ.

(A|B)(C|D) A C, A D, B C, B D
A[B]C A B C, A C
A {B A} A, A B A, A B A B A, A B A B A B A, ...
{A|B} C C, A C, B C, A A C, A B C, B B C, B A C, ...

Чтобы отличать описания синтаксиса на РБНФ от окружаю-
щего их текста на русском языке, мы будем помечать все строчки,
в которых находится текст на РБНФ, литерой $ в первой позиции.

Поскольку сама РБНФ представляет собой некоторый язык,
ее синтаксис также может быть описан с помощью РБНФ следу-
ющим образом:

$ Синтаксис = { СинтФормула }.
$ СинтФормула =
$ Идентификатор "=" СинтВыражение ".".
$ СинтВыражение =
$ СинтСлагаемое { "|" СинтСлагаемое }.
$ СинтСлагаемое =
$ СинтМножитель { СинтМножитель }.
$ СинтМножитель = Идентификатор | Цепочка |
$ "(" СинтВыражение ")" |
$ "[" СинтВыражение "]" |
$ "{" СинтВыражение "}".

2.2 Естественный метод описания
семантики

Для описания процесса выполнения программ на Рефале
Плюс будет использован метод, известный под названием “есте-
ственная семантика” или “структурная операционная семантика”
[Плоткин 1983], [Апт 1983].

Название “естественная семантика” объясняется сходством
этого метода описания с генценовским естественным выводом в
математической логике. При таком способе описания семантикой
языка является неупорядоченное множество утверждений о про-
граммах и фрагментах программ.

Например, предположим, что в описываемом языке имеются
выражения, которые могут содержать переменные и вычисление
которых может приводить к побочным эффектам (например, из-
за выполнения операций ввода-вывода). Тогда определение семан-
тики языка может включать утверждения вида

ρ, σ′ ` E ⇒ X,σ′′

где E — выражение языка, ρ — среда, которая связывает пере-
менные с их значениями в том контексте, в котором находится E,
σ′ и σ′′ — глобальное состояние до начала вычисления E и после
вычисления E, а X — результат вычисления E. Глобальное со-
стояние может включать состояние памяти, состояние файлов и
т. п.

Смысл таких утверждений неформально может быть истолко-
ван следующим образом: если вычисление выражения E начина-
ется в среде ρ и глобальном состоянии σ′, оно может завершиться
получением значения X в глобальном состоянии σ′′.

Знак " ` " (который можно читать как “влечет” или “вызыва-
ет”) служит напоминанием о том, что результат вычисления выра-
жения E зависит от среды ρ, в которой вычисляется выражение,
ибо это выражение может содержать переменные.

Таким образом, задача описания семантики языка при таком
подходе сводится к задаче описания множества истинных утвер-
ждений о программах и их фрагментах.

“Естественное” описание семантики представляет собой неупо-
рядоченное множество правил вывода, позволяющих из одних ис-
тинных утверждений выводить другие.

Каждое правило вывода выглядит как дробь, в числителе ко-
торой находятся посылки, а в знаменателе — заключение. Если у
правила нет ни одной посылки, оно называется аксиомой, и в этом
случае горизонтальная черта может опускаться.

Кроме того, для каждого правила могут иметься дополнитель-
ные условия, которые накладывают какие-то ограничения на вид
посылок и заключения. Эти ограничения мы будем записывать
справа от правила или непосредственно под ним.

Например, предположим, что в описываемом языке имеется
конструкция if E then E′ else E′′, смысл которой неформаль-
но может быть описан следующим образом.

Вычислить E. Если получится true, то вычислить E′ и по-
лученный результат считать результатом вычисления всей кон-
струкции. Если же в результате вычисления E получилось false,
то вычислить E′′ и полученный результат считать результатом
вычисления всей конструкции.

Недостатком такого описания является то, что мы забыли ука-
зать, в какой среде происходит вычисление каждого из выраже-
ний E, E′ и E′′. Поэтому переформулируем описание более точно.

Если результатом вычисления выражения E в среде ρ явля-
ется true, а результатом вычисления выражения E′ в среде ρ яв-
ляется X, то X является результатом вычисления конструкции
if E then E′ else E′′ в среде ρ.

Если результатом вычисления выражения E в среде ρ явля-
ется false, а результатом вычисления выражения E′′ в среде ρ
является X, то X является результатом вычисления конструкции
if E then E′ else E′′ в среде ρ.

Это многословное определение можно записать в более крат-
кой и наглядной форме в виде двух правил вывода:

ρ, σ ` E ⇒ true , σ′

ρ, σ′ ` E′ ⇒ X,σ′′

ρ, σ ` if E then E′ else E′′ ⇒ X,σ′′

ρ, σ ` E ⇒ false , σ′

ρ, σ′ ` E′′ ⇒ X,σ′′

ρ, σ ` if E then E′ else E′′ ⇒ X,σ′′

Обратите внимание, что формализованное описание семанти-
ки, в отличие от словесного, точно определяет, как изменяется
глобальное состояние в процессе вычисления.

2.3 Лексическая структура программы

Программа на языке Рефал Плюс представляет собой конеч-
ную последовательность литер. Анализ такой программы прово-
дится в два этапа. Сначала производится сканирование програм-
мы, в результате чего поток литер разбивается на лексемы. Затем
результат сканирования подвергается синтаксическому анализу,
результатом которого является дерево синтаксического анализа.
В связи с этим описание синтаксиса состоит из двух частей: сна-
чала описывается лексическая структура программы, т.е. правила
изображения лексем с помощью цепочек литер, а затем — синтак-
сическая структура, т. е. правила, по которым программы состав-
ляются из лексем.

$ Программа = { Межа Лексема } Межа.
$ Межа = РазделительЛексем | Комментарий.
$ РазделительЛексем =
$ Пробел | Табуляция | КонецСтроки.

Программа представляет собой конечную последовательность
лексем. В качестве разделителей лексем используются следующие
литеры: пробел, табуляция и конец строки. Избыточные раздели-
тели лексем могут появляться как до, так и после лексемы, но не
внутри лексем.

Если две соседние лексемы могут быть однозначно отделены
одна от другой, разделители лексем между ними могут отсутство-
вать.

2.3.1 Комментарии

$ Комментарий = "//" ОкончаниеКомментария КонецСтроки
$ | "/*" ТелоКомментария "*/".
$ ОкончаниеКомментария =
$ любая цепочка литер, не содержащая КонецСтроки.
$ ТелоКомментария =
$ любая цепочка литер, не содержащая "*/".

Существует два способа вставлять комментарии между лек-
семами. Первый способ — поставить литеры //. При этом весь
остаток строки игнорируется. Второй способ — заключить ком-
ментарий в пару “комментарных скобок” /* и */. Например:

// Это - комментарий.
// И это - комментарий.

/* И это - тоже! */

2.3.2 Лексемы

$ Лексема =
$ Скобка | КлючевоеСлово |
$ ИзображениеЦепочкиЛитер |
$ ИзображениеСлова | ИзображениеЧисла |
$ Переменная.

$ Скобка = "(" | ")" | "{" | "\{" | "}" | "<" | ">".

Лексемы представляют собой скобки, ключевые слова, изоб-
ражения цепочек символов-литер, изображения символов-слов,
изображения символов-чисел, и переменные.

2.3.3 Ключевые слова

$ КлючевоеСлово =
$ "$box" | "$channel | "$const" | "$error" |
$ "$fail" | "$func" | "$func?" | "$iter" |
$ "$l" | "$r" | "$string" | "$table" |

$ "$trace | "$traceall" |
$ "$trap" | "$use" | "$vector" | "$with" |
$ "#" | "&" | "," | ":" | "::" | ";" | "=" |
$ "\?" | "\!".

2.3.4 Символы-литеры

$ ИзображениеЦепочкиЛитер =
$ "’" { ИзображениеЛитеры } "’".

$ ИзображениеЛитеры =
$ ИзображениеОбычнойЛитеры |
$ ИзображениеОсобойЛитеры |
$ ШестнадцатиричноеИзображениеЛитеры.
$ ИзображениеОбычнойЛитеры =
$ любая литера, кроме апострофа (’), кавычки ("),
$ обратной косой (\) и конца строки.
$ ИзображениеОсобойЛитеры =
$ "\n" | "\t" | "\b" | "\r" | "\f" |
$ "\\" | "\’" | ’\"’ .
$ ШестнадцатиричноеИзображениеЛитеры =
$ "\x" ШестнЦифра ШестнЦифра.

$ Цифра =
$ "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" |
$ "8" | "9".

$ ШестнЦифра =
$ Цифра |
$ "A" | "B" | "C" | "D" | "E" | "F" |
$ "a" | "b" | "c" | "d" | "e" | "f".

Символ-литера соответствует одной литере и записывается в
виде изображения этой литеры, заключенного в апострофы. На-
пример:

’A’ ’a’ ’7’ ’$’

Как правило, изображением литеры является сама эта литера,
за исключением следующих литер, для которых предусмотрены
особые обозначения:

Новая строка (перевод строки) HL (LF) ’\n’
Горизонтальная табуляция HT ’\t’
Возврат на шаг BS ’\b’
Возврат каретки CR ’\r’
Перевод формата FF ’\f’
Обратная косая \ ’\\’
Апостроф ’ ’\’’
Двойная кавычка " ’\"’

Изображения литер, входящие в изображения цепочек
символов-литер и в изображения символов-слов, могут записы-
ваться в виде

\xZZ

где ZZ— две шестнадцатиричные цифры, задающие однобайтовый
код литеры. При этом символы-слова, содержащие такие изобра-
жения литер, должны заключаться в двойные кавычки.

Например, \x2A и \x2a эквивалентны *.
Последовательность из нескольких символов-литер может

быть записана в виде одной последовательности изображений ли-
тер, заключенной в апострофы. Например:

’ABC’
’123’
’\"I don\’t like swimming!\" - said a little girl.’

Таким образом, последовательность из трех символов-литер
’A’, ’B’ и ’C’ может быть записана любым из перечисленных
ниже способов:

’A’ ’B’ ’C’
’A’’B’’C’
’ABC’

2.3.5 Символы-слова

$ ИзображениеСлова =
$ Идентификатор |
$ ’"’ { ИзображениеЛитеры } ’"’.

$ Идентификатор =
$ НачалоИдентификатора ХвостИдентификатора.
$ НачалоИдентификатора = ПрописнаяБуква | "_".
$ ХвостИдентификатора = { Буква | Цифра | "_" }.

$ Буква = ПрописнаяБуква | СтрочнаяБуква.

$ ПрописнаяБуква =
$ "A" | "B" | "C" | "D" | "E" | "F" | "G" | "H" |
$ "I" | "J" | "K" | "L" | "M" | "N" | "O" | "P" |
$ "Q" | "R" | "S" | "T" | "U" | "V" | "W" | "X" |
$ "Y" | "Z".

$ СтрочнаяБуква =
$ "a" | "b" | "c" | "d" | "e" | "f" | "g" | "h" |
$ "i" | "j" | "k" | "l" | "m" | "n" | "o" | "p" |
$ "q" | "r" | "s" | "t" | "u" | "v" | "w" | "x" |
$ "y" | "z".

Символ-слово соответствует цепочке литер и записывается
в виде цепочки изображений этих литер, заключенной в двой-
ные кавычки. Изображения литер такие же, как и при записи
символов-литер. Например:

"ABC"
"123"
"\"I don\’t like swimming!\" - said a little girl."

Следует обратить внимание, что "ABC" является изображени-
ем одного символа-слова, в то время как ’ABC’ — это изображение
цепочки из трех символов-литер. Кроме того, считается, что все

символы-слова, состоящие из одной литеры, не совпадают с со-
ответствующими символами-литерами. Например, символ-литера
’A’ и символ-слово "A" — разные символы.

В некоторых случаях символы-слова разрешается записывать
без окружающих двойных кавычек. Для этого должны быть вы-
полнены следующие условия.

Во-первых, этот символ-слово может содержать только следу-
ющие литеры: прописные буквы, строчные буквы, цифры, под-
черкивание.

Во-вторых, этот символ-слово должен начинаться с прописной
буквы или подчеркивания.

Например, ниже приведено два изображения одного и того же
символа-слова:

I_do_not_like_swimming
"I_do_not_like_swimming"

2.3.6 Символы-числа

$ ИзображениеЧисла =
$ [ "+" | "-" ] Цифра { Цифра } |
$ "0x" ШестнЦифра { ШестнЦифра }.

Символы-числа соответствуют целым числам со знаком. Они
могут быть неограниченного размера. Например:

123 +121 -123 -123456789012345678901234567890

Неотрицательные целые числа можно записывать в виде

0xZZZ...ZZ

где ZZZ...ZZ — непустая последовательность шестнадцатиричных
цифр.

Например, 0xFF и 0xff эквивалентны 255.

116



2.3.7 Переменные

$ Переменная =
$ s-переменная | t-переменная | e-переменная |
$ v-переменная.
$ s-переменная = "s" [ "." ] ИндексПеременной.
$ t-переменная = "t" [ "." ] ИндексПеременной.
$ v-переменная = "v" [ "." ] ИндексПеременной.
$ e-переменная = "e" [ "." ] ИндексПеременной.

$ УказательТипаПеременной = "s" | "t" | "v" | "e".
$ ИндексПеременной = ХвостИдентификатора.

Переменная состоит из указателя типа переменной, необяза-
тельной точки и индекса переменой. Например:

tHead eTail e.1 e1 tX s t e

При этом, eI и e.I — различные изображения одной и той же
переменной.

Если две переменные стоят рядом, они должны быть разделе-
ны межой. Например, sAeB является одной переменной, в то время
как sA eB — две переменные.

Если индекс переменной отсутствует, это означает, что подра-
зумевается, что эта переменная имеет уникальный индекс и отли-
чается от всех прочих переменных в программе. Таким образом,
например, если в программе не используются переменные e1000
и e2000, то запись e e может быть заменена на e1000 e2000.

Переменные делятся на четыре класса: s-переменные, t-
переменные, v-переменные и e-переменные, в соответствии с их
указателями типа.

2.3.8 Нормализация потока лексем

В процессе ввода программы сканер разбивает входной поток
литер на лексемы. Многие лексемы имеют один и тот же смысл.
Так, например, все три лексемы

125 000125 +125

обозначают одно и то же число 125.
Именно поэтому при описании лексической структуры про-

граммы мы употребляли такие термины как “изображение числа”
и “изображение слова” вместо “число” и “слово”.

Кроме того, такая лексема как “изображение цепочки литер”
является изображением не одного синтаксического объекта, а ко-
нечной последовательности литер.

В дальнейшем, при описании синтаксиса, мы будем считать,
что сканер произвел “нормализацию” лексем, т. е. преобразовал
лексемы к стандартному виду таким образом, чтобы одним и тем
же синтаксическим объектам всегда соответствовали одни и те
же лексемы. При этом изображение цепочки из N литер разби-
вается на N отдельных лексем, каждая из которых изображает
ровно одну литеру. Таким образом, появляется возможность при
описании синтаксиса говорить не об “изображениях объектов”, а
об “объектах”.

Соответствие между исходными лексемами и нормализован-
ными лексемами следующее:

ИзображениеЦепочкиЛитер ⇒ Литера1 Литера2 . . . Литераn

ИзображениеСлова ⇒ Слово
ИзображениеЧисла ⇒ Число

Не следует путать литеры, из которых составлены исходные
лексемы с символами-литерами, которые получаются на выходе
сканера.

2.4 Объекты и значения

Под объектами обычно понимают некоторые сущности, кото-
рые существуют во времени, могут изменяться, но при этом оста-
ются сами собой.

Под значениями обычно понимают некоторые сущности, кото-
рые не могут изменяться и, в этом смысле, находятся вне времени.

Конечно, значение можно считать частным, вырожденным
случаем объекта (а именно, застывшим объектом, не способным

к развитию), но мы все же обычно будем называть “объектами”
только такие объекты, которые не являются значениями.

Иметь дело с объектами труднее, чем со значениями, ибо они
могут изменяться. Поэтому объекты часто снабжают именами.
Основным свойством имени является то, что оно однозначно свя-
зано с объектом (однозначно идентифицирует этот объект). В от-
личие от самих объектов имена являются типичными значениями,
поскольку они не меняются из-за того, что изменяется сам объект.

Программы на Рефале Плюс имеют дело как с объектами, так
и со значениями.

Все значения, с которыми работают Рефал-программы, яв-
ляются объектными выражениями, которые представляют собой
конечные последовательности символов и скобок, правильно по-
строенные относительно скобок. Скобки служат для придания
объектным выражениям древовидной структуры, в то время как
символы представляют собой элементарные данные (литеры, чис-
ла, слова и ссылки на объекты).

Объекты, с которыми работают Рефал-программы, могут со-
держать внутри себя объектные выражения (в частности — ссыл-
ки на объекты). Содержимое объектов может изменяться в про-
цессе работы Рефал-программы. Доступ к объектам осуществля-
ется через их имена, в качестве которых служат символы-ссылки.

Объекты создаются как во время компиляции Рефал-
программы, так и во время ее исполнения. С теоретической точки
зрения, любой объект, будучи создан, существует бесконечно. На
практике, однако, Рефал-программы исполняются на компьюте-
рах, имеющих конечный объем памяти, поэтому все реализации
Рефала Плюс должны предусматривать сборку мусора. Сборка
мусора обеспечивает автоматическое уничтожение тех объектов,
которые недоступны программе и которые, в силу этого, заведомо
не могут повлиять на дальнейшую работу программы.

2.5 Объектные выражения

2.5.1 Синтаксис объектных выражений

$ ОбъектноеВыражение = { ОбъектныйТерм }.
$ ОбъектныйТерм = Символ | "(" ОбъектноеВыражение ")".

В дальнейшем мы будем обозначать объектные выражения че-
рез E , объектные термы — через T , а символы — через S.

2.5.2 Статические и динамические символы

$ Символ = СтатическийСимвол | ДинамическийСимвол.
$ СтатическийСимвол = Литера | Слово | Число.
$ ДинамическийСимвол =
$ СсылкаНаФункцию | СсылкаНаТаблицу |
$ СсылкаНаЯщик | СсылкаНаВектор |
$ СсылкаНаСтроку | СсылкаНаКанал.

Символы делятся на две категории: статические и динамиче-
ские.

К статическим символам относятся символы-литеры,
символы-слова и символы-числа.

Статические символы существуют “объективно”: статический
символ может быть выведен в канал ввода-вывода, а затем вве-
ден обратно. При этом мы получим опять тот же символ. В этом
смысле статические символы существуют еще до начала испол-
нения программы и продолжают существовать после завершения
исполнения программы.

Динамические символы являются ссылками на объекты. А
именно, каждый динамический символ представляет собой ука-
затель на то место в памяти, где находится объект во время ис-
полнения программы. При этом объект может быть определением
функции, ящиком, вектором, строкой, таблицей или каналом.

Динамические символы, в противоположность статическим,
“субъективны”. Они создаются в момент загрузки программы и
в процессе ее исполнения. Хотя их можно вывести в канал ввода-
вывода, их нельзя ввести обратно. Когда выполнение программы

завершается, все динамические символы, созданные в процессе ее
выполнения, теряют всякий смысл.

К динамическим символам относятся ссылки на функции,
ссылки на ящики, ссылки на векторы, ссылки на строки, ссыл-
ки на таблицы и ссылки на каналы.

2.5.3 Символические имена

Различным сущностям в программах на Рефале Плюс могут
присваиваться символические имена, которые, с синтаксической
точки зрения, являются идентификаторами.

А именно, в программах могут использоваться:

• Имена констант (раздел 2.12.1).

• Имена функций (раздел 2.12.3).

• Имена ссылок на объекты (раздел 2.12.2).

• Имена модулей (раздел 2.14).

Подробнее об этом будет сказано в соответствующих разделах.

2.5.4 Символические имена выражений

$ ИменованноеВыражение = "&" ИмяВыражения.
$ ИмяВыражения = Идентификатор.

Объектным выражениям в программе на Рефале Плюс могут
присваиваться символические имена. Каждое из этих имен обо-
значает либо статическое объектное выражение (раздел 2.12.1),
либо ссылку на объект (раздел 2.12.2), либо ссылку на функцию
(раздел 2.12.3).

Если имя N обозначает объектное выражение E , то конструк-
ция вида &N означает, что в этом месте должно быть подставлено
объектное выражение E .

Поскольку все объекты, как и ссылки на них, создаются при
компиляции программы или в процессе ее загрузки и исполне-
ния, символы-ссылки не могут появляться в тексте программы

121



непосредственно. Тем не менее, если объект объявлен в програм-
ме, то ссылке на него присваивается символическое имя, и, таким
образом, эта ссылка может быть подставлена в нужные места про-
граммы.

2.5.5 Устранение символических имен выражений

Если N является символическим именем объектного выра-
жения E , то все использования этого имени в программе можно
устранить, заменив каждую конструкцию &N на E .

В дальнейшем, при описании контекстных ограничений и се-
мантики языка, мы будем предполагать, что это преобразование
уже выполнено, и программа не содержит использований симво-
лических имен. При этом, с другой стороны, в тексте преобразо-
ванной программы могут появиться динамические символы.

2.6 Значения переменных и среды

Вычисление любой конструкции Рефала Плюс требует знания
значений переменных, входящих в эту конструкцию. Эта инфор-
мация о значениях переменных естественным образом представ-
ляется в виде функции с конечной областью определения, ставя-
щей в соответствие каждой переменной из ее области определения
значение этой переменной.

А именно, пусть ρ — среда с областью определения
{V1, . . . , Vn}. Тогда ρ(Vj) = Ej является значением переменной Vj .
Такую среду мы будем обозначать через {V1 = E1, . . . , Vn = En}.
В частности, пустая среда будет обозначаться через {}.

Область определения среды ρ мы будем обозначать че-
рез dom[ρ]. Таким образом, dom[{V1 = E1, . . . , Vn = En}] =
{V1, . . . , Vn}.

При этом, без дальнейших оговорок, мы всегда будем пред-
полагать, что переменные обязательно имеют значения, соответ-
ствующие их типам. Т. е. значениями s-переменных могут быть
только символы, значениями t-переменных — только объектные

термы, значениями e-переменных — произвольные объектные вы-
ражения, а значениями v-переменных — непустые объектные вы-
ражения.

В дальнейшем ρ + ρ′ будет означать пополнение среды ρ зна-
чениями переменных из среды ρ′, которое строится следующим
образом.

dom[ρ + ρ′] содержит все переменные из dom[ρ′], а также все
переменные из dom[ρ], индексы которых отличаются от индексов
переменных из dom[ρ′].

Для любой переменной V из dom[ρ+ρ′], если ρ′(V ) определено,
то (ρ+ρ′)(V ) = ρ′(V ), а если ρ′(V ) не определено, то (ρ+ρ′)(V ) =
ρ(V ).

Например:

{sX = 1, sY = 2} + {sY = 200, sZ = 300}
= {sX = 1, sY = 200, sZ = 300}

{sX = 1, sY = 2} + {eY = 200, sZ = 300}
= {sX = 1, eY = 200, sZ = 300}

2.7 Результатные выражения

2.7.1 Синтаксис

$ РезультатноеВыражение =
$ { РезультатныйТерм | ИменованноеВыражение }.
$ РезультатныйТерм =
$ СтатическийСимвол | Переменная |
$ "(" РезультатноеВыражение ")" |
$ ВызовФункции.
$ ВызовФункции =
$ "<" ИмяФункции АргументВызова ">".
$ ИмяФункции = Идентификатор.
$ АргументВызова =
$ РезультатноеВыражение.

В дальнейшем мы будем обозначать результатные выражения

через Re, результатные термы — через Rt, переменные — через
V , e-переменные — через V e, а имена функций — через F .

2.7.2 Вычисление результатных выражений

Результатные выражения служат для порождения объектных
выражений.

Результатное выражение Re может быть вычислено в любой
среде ρ, при условии, что ρ дает значения для всех переменных,
входящих в Re.

Если вычисление результатного выражения Re заканчивается,
то результатом этого вычисления является либо объектное выра-
жение E , либо неуспех $fail(0), либо ошибка $error(E), где E —
сообщение об ошибке.

Вычисление вызовов функций может приводить к изменению
глобального состояния программы (например, в случае выпол-
нения операций ввода-вывода или каких-то действий над объек-
тами), поэтому, если результатное выражение содержит вызовы
функций, его вычисление может также привести к изменению гло-
бального состояния.

Запись ρ, σ ` Re ⇒ X,σ′ означает, что результатом вычис-
ления результатного выражения Re в среде ρ является X. При
этом, если вычисление началось в глобальном состоянии σ, оно
закончится в глобальном состоянии σ′.

Результатное выражение Re вычисляется слева направо. При
этом вместо переменных подставляются их значения и выполня-
ются вызовы функций.

Вычисление вызовов функций выполняется следующим обра-
зом. Если вызов имеет вид <F Re>, то вычисляется результатное
выражение Re. Если в результате получится объектное выраже-
ние E , то функция F применяется к E .

Запись σ ` <F E> ⇒ X,σ′ означает, что результатом приме-
нения функции F к объектному выражению E является X, при-
чем, если вычисление началось в глобальном состоянии σ, оно

закончится в глобальном состоянии σ′.

ρ, σ ` ⇒, σ

ρ, σ ` Re⇒ E ′, σ′
ρ, σ′ ` Rt⇒ E ′′, σ′′
ρ, σ ` Re Rt⇒ E ′ E ′′, σ′′

ρ, σ ` Re⇒ E ′, σ′
ρ, σ′ ` Rt⇒ $fail(0), σ′′

ρ, σ ` Re Rt⇒ $fail(0), σ′′

ρ, σ ` Re⇒ E ′, σ′
ρ, σ′ ` Rt⇒ $error(E ′′), σ′′
ρ, σ ` Re Rt⇒ $error(E ′′), σ′′

ρ, σ ` Re⇒ $fail(0), σ′

ρ, σ ` Re Rt⇒ $fail(0), σ′

ρ, σ ` Re⇒ $error(E ′), σ′
ρ, σ ` Re Rt⇒ $error(E ′), σ′

ρ, σ ` S ⇒ S, σ
ρ, σ ` V ⇒ E , σ

где E = ρ(V ).

ρ, σ ` Re⇒ E , σ′
ρ, σ ` (Re)⇒ (E), σ′

ρ, σ ` Re⇒ $fail(0), σ′

ρ, σ ` (Re)⇒ $fail(0), σ′

ρ, σ ` Re⇒ $error(E), σ′
ρ, σ ` (Re)⇒ $error(E), σ′

ρ, σ ` Re⇒ E , σ′
σ′ ` <F E>⇒ X,σ′′

ρ, σ ` <F Re>⇒ X,σ′′

125



ρ, σ ` Re⇒ $fail(0), σ′

ρ, σ ` <F Re>⇒ $fail(0), σ′

ρ, σ ` Re⇒ $error(E), σ′
ρ, σ ` <F Re>⇒ $error(E), σ′

2.7.3 Примеры

Ниже приведены примеры результатных выражений:

(A B) C D
t.Head e.Tail
While t.Condition Do t.Statement
<Mult sN <Factorial <Sub sN 1>>

Следующие результатные выражения являются результатны-
ми термами:

(A B)
t.Head
<Mult sN <Factorial <Sub sN 1>>

Пусть ρ = {sM = 2, sN = 3, eA = A B C, tB = (D E F)},
и пусть Add— имя функции, выполняющей сложение целых чисел,
а Mult — имя функции, выполняющей умножение целых чисел,
т. е. в частности имеет место

σ ` <Add 3 100>⇒ 103, σ
σ ` <Mult 2 103>⇒ 206, σ

для любого глобального состояния σ, ибо функции Add и Mult не
изменяют глобальное состояние.

Тогда имеем

ρ, σ ` eA (eA tB) tB⇒
A B C (A B C (D E F)) (D E F), σ

ρ, σ ` <Mult sM <Add sN 100>>⇒ 206, σ

для любого глобального состояния σ.

2.8 Образцы

2.8.1 Синтаксис

$ Образец = УказательНаправления ОбразцовоеВыражение.
$ УказательНаправления = [ "$l" | "$r" ].

$ ОбразцовоеВыражение =
$ { ОбразцовыйТерм | ИменованноеВыражение }.
$ ОбразцовыйТерм =
$ СтатическийСимвол | Переменная |
$ "(" ОбразцовоеВыражение ")".

Образец представляет собой образцовое выражение, перед ко-
торым может стоять указатель направления $l или $r. Указатель
направления $l означает, что сопоставление с образцом должно
выполняться слева направо. Указатель $r — что справа налево.
Если указатель направления отсутствует, подразумевается указа-
тель $l.

В дальнейшем мы будем обозначать образцы через P , образ-
цовые выражения — через Pe, образцовые термы — через Pt, а
указатели направления сопоставления — через D.

2.8.2 Сопоставление с образцом

Мы говорим, что среда ρ′ является результатом сопоставления
объектного выражения E с образцом P в начальной среде ρ, если
выполнены следующие условия.

• Среда ρ′ является расширением среды ρ, т. е. область опреде-
ления среды ρ′ является расширением области определения
среды ρ, и для любой переменной V из области определения
среды ρ имеет место ρ′(V ) = ρ(V ).

• Если все переменные, входящие в P , заменить на их значе-
ния в соответствии с ρ′, а указатель направления отбросить,
получится объектное выражение E .

В этом случае среду ρ′ мы будем также называть вариантом
сопоставления E с P в среде ρ, a множество таких вариантов со-
поставления будем обозначать через M(ρ, E , P ).

На множестве M(ρ, E , P ) вводится отношение порядка следу-
ющим образом.

Пусть M(ρ, E , P ) содержит два различных варианта сопостав-
ления ρ′ и ρ′′. Рассматриваются все вхождения переменных в P .

Если P имеет указатель направления $l, то отыскивается са-
мое первое слева вхождение, которое принимает для двух вари-
антов сопоставления ρ′ и ρ′′ различные значения.

Если P имеет указатель направления $r, то отыскивается са-
мое первое справа вхождение, которое принимает для двух вари-
антов сопоставления ρ′ и ρ′′ различные значения.

Пусть найденное вхождение является вхождением переменной
V . Тогда по определению объектные выражения ρ′(V ) и ρ′′(V )
не равны. Более того, можно показать, что заведомо либо ρ′(V )
является частью ρ′′(V ), либо ρ′′(V ) является частью ρ′(V ).

Тогда если ρ′(V ) является частью ρ′′(V ), считается, что ρ′

предшествует ρ′′. В противном случае считается, что ρ′′ предше-
ствует ρ′.

Конечную последовательность сред ρ1, ρ2, . . . , ρn будем за-
писывать в виде [ρ1, ρ2, . . . , ρn], пустую последовательность — в
виде [ ].

Запись [ρ0] _ [ρ1, . . . , ρn] будет обозначать [ρ0, ρ1, . . . , ρn].
Запись ρ ` E : P ⇒ [ρ1, . . . ρn] будет означать, что

M(ρ, E , P ) = {ρ1, . . . , ρn} и при этом ρi предшествует ρj для
всех i < j.

2.8.3 Примеры

Ниже приведены примеры образцов:

t.Head e.Tail
eX (eY)
eA ’+’ eB
$l eA ’+’ eB
$r eA ’+’ eB

Далее приведены примеры сопоставления с образцом:

{} ` A () C D E : $l sX tY tZ e1⇒ [
{sX = A, tY = (), tZ = C, e1 = D E} ]

{} ` 1 2 3 : $l eA eB⇒ [
{eA = , eB = 1 2 3},
{eA = 1, eB = 2 3},
{eA = 1 2, eB = 3},
{eA = 1 2 3, eB = } ]

{} ` 1 2 3 : $r eA eB⇒ [
{eA = 1 2 3, eB = },
{eA = 1 2, eB = 3 },
{eA = 1, eB = 2 3 },
{eA = , eB = 1 2 3 } ]

{eA = 1 2} ` $l 1 2 3 4 5 : eA eB⇒ [
{eA = 1 2, eB = 3 4 5} ]

2.9 Жесткие выражения

2.9.1 Синтаксис

$ ЖесткоеВыражение =
$ { ЖесткийКрай } |
$ { ЖесткийКрай } e-переменная { ЖесткийКрай } |
$ { ЖесткийКрай } v-переменная { ЖесткийКрай }.
$ ЖесткийКрай =
$ { ЖесткийТерм | ИменованноеВыражение }.
$ ЖесткийТерм =
$ СтатическийСимвол |
$ s-переменная | t-переменная |
$ "(" ЖесткоеВыражение ")".

Таким образом жесткое выражение на каждом уровне скобок
содержит не более чем одну e-переменную или v-переменную.

Индексы всех переменных, входящих в одно и то же жесткое
выражение, должны быть попарно различны.

В дальнейшем мы будем обозначать жесткие выражения через
He, а жесткие термы — через Ht.

2.9.2 Сопоставление с жестким выражением

Любое жесткое выражение He представляет собой частный
случай образцового выражения. Его особенностью является то,
что для любого объектного выражения E либо сопоставление E с
He невозможно, либо может быть сделано только одним способом.
Таким образом, имеет место либо {} ` E : He ⇒ [ ], либо {} `
E : He⇒ [ρ].

Запись ρ ` E :: He⇒ ρ′ будет означать, что {} ` E : He⇒
ρ′′ и ρ′ = ρ+ ρ′′, т. е. что ρ′ получается из ρ следующим образом.
Сначала E сопоставляется с He в пустой среде. Т. е. прежние зна-
чения переменных игнорируются. В результате получается среда
ρ′′, которая содержит значения для переменных из He. После это-
го исходная среда ρ корректируется в соответствии с новыми зна-
чениями переменных из He, в результате чего получается новая
среда ρ′.

2.9.3 Примеры

Ниже приведены примеры жестких выражений:

t.Head e.Tail
sX (eY) eZ (A eA)

Далее приведены примеры сопоставления с жесткими выра-
жениями:

{sX = XXX, eA = A B C} ` X Y Z :: sY eA⇒
{sX = XXX, eA = Y Z, sY = X}

{sX = XXX, eA = A B C} ` X Y Z :: eA sY⇒
{sX = XXX, eA = X Y, sY = Z}

2.10 Тропы

2.10.1 Синтаксис

$ Тропа =
$ Условие | Присваивание | Поиск | Перестройка |
$ Хвост | Источник.

$ Условие =
$ Источник Хвост.
$ Присваивание =
$ Источник "::" ЖесткоеВыражение [ Хвост ].
$ Поиск =
$ Источник "$iter" Источник
$ [ "::" ЖесткоеВыражение ] [ Хвост ].
$ Перестройка =
$ Источник ":" Образец [ Хвост ].

$ Хвост =
$ ОгражденнаяТропа | ОтрицаниеУсловия |
$ Забор | Отсечение |
$ Тупик | ПраваяЧасть | Авария |
$ ПерехватАварий.

$ ОгражденнаяТропа =
$ "," Тропа.
$ ОтрицаниеУсловия =
$ "#" Источник [ Хвост ].
$ Забор =
$ "\?" Тропа.
$ Отсечение =
$ "\!" Тропа.
$ Тупик =
$ "$fail".
$ ПраваяЧасть =
$ "=" Тропа.
$ Авария =

$ "$error" Тропа.
$ ПерехватАварий =
$ "$trap" Тропа "$with" ОбразцовоеРаспутье.

$ Источник =
$ Распутье | Выбор | РезультатноеВыражение.

$ Распутье =
$ "\{" ПослТроп "}" |
$ "{" ПослТроп "}".

$ Выбор =
$ Источник ":" ОбразцовоеРаспутье.

$ ОбразцовоеРаспутье =
$ "\{" ПослПредложений "}" |
$ " {" ПослПредложений "}" |

$ ПослТроп = { Тропа ";" }.

$ ПослПредложений =
$ { Предложение ";" }.

$ Предложение = Образец [ Хвост ].

В дальнейшем мы будем обозначать тропы через Q, хвосты –
через R, источники – через S, образцовые распутья – через Palt,
а предложения – через Snt.

Синтаксис троп выглядит несколько запутанным. Это вызва-
но стремлением избавить пользователя от необходимости писать
лишние разделители в тех случаях, когда в этом нет реальной
необходимости.

Для этого особо выделяются два частных случая троп: “хво-
сты” и “источники”, обладающие полезными синтаксическими
свойствами. Хвосты начинаются с легко распознаваемых ключе-
вых слов и не могут “склеиться” с предшествующими результат-

ными выражениями и образцами, а источники не содержат за-
пятую на нулевом уровне фигурных скобок, благодаря чему они
однозначно отделяются от последующих конструкций.

2.10.2 Вычисление троп

Вычисление тропы Q производится при условии, что задана
среда ρ и целое неотрицательное число m. Среда ρ содержит зна-
чения переменных, которые могут потребоваться в процессе вы-
числения тропы. Число m, именуемое уровнем тропы, характери-
зует степень вложенности тропы в заборы, т. е. количество забо-
ров \?, окружающих Q, еще не закрытых отсечениями \!.

Если вычисление тропы заканчивается, то результатом вы-
числения является либо объектное выражение E , либо неуспех
$fail(k), где целое неотрицательное k характеризует “степень се-
рьезности” неуспеха, либо ошибка $error(E), где E — сообщение
об ошибке.

Если тропа находится на уровне m, а результатом ее вычис-
ления является неуспех $fail(k), “серьезность неуспеха” всегда
удовлетворяет условию 0 ≤ k ≤ m + 1. В частности, если тропа
находится на уровне 0, всегда либо k = 0, либо k = 1.

В дальнейшем запись ρ,m, σ ` Q ⇒ X,σ′ означает, что если
тропа Q находится на уровне m, то результатом ее вычисления
в среде ρ является X, причем если вычисление тропы Q нача-
лось в глобальном состоянии σ, то оно завершится в глобальном
состоянии σ′.

Хвосты и источники являются частными случаями троп, по-
этому это же обозначение используется и применительно к вычис-
лению хвостов и троп.

Во многих случаях смысл тропы Q может быть объяснен че-
рез смысл другой тропы Q′. А именно, чтобы вычислить тропу Q,
следует вычислить тропу Q′, и то, что получится, следует считать
результатом вычисления тропы Q. Формально это можно выра-
зить с помощью следующего правила вывода:

ρ,m, σ ` Q′ ⇒ X,σ′

ρ,m, σ ` Q⇒ X,σ′

Поскольку такие правила встречаются довольно часто, мы бу-
дем в дальнейшем записывать их в сокращенном виде следующим
образом:

Q⇒⇒ Q′

2.10.3 Условия

Тропа S R означает, что следует вычислить источник S, а
затем, в случае успеха, вычислять хвост R.

ρ, 0, σ ` S ⇒, σ′
ρ,m, σ′ ` R⇒ X,σ′′

ρ,m, σ ` S R⇒ X,σ′′

ρ, 0, σ ` S ⇒ $fail(k), σ′

ρ,m, σ ` S R⇒ $fail(0), σ′

ρ, 0, σ ` S ⇒ $error(E), σ′
ρ,m, σ ` S R⇒ $error(E), σ′

2.10.4 Присваивания

Тропа S :: He R позволяет вычислить источник S и связать
полученные результаты с переменными в He. В результате полу-
чается новая среда, в которой и вычисляется хвост R.

Считается, что источник S находится на нулевом уровне.
Если хвост R является огражденной пустой тропой (всегда

вырабатывающей пустое выражение), он может быть опущен.

S :: He⇒⇒ S :: He,

ρ, 0, σ ` S ⇒ E , σ′
ρ ` E :: He⇒ ρ′

ρ′,m, σ′ ` R⇒ X,σ′′

ρ,m, σ ` S :: He R⇒ X,σ′′

ρ, 0, σ ` S ⇒ $fail(k), σ′

ρ,m, σ ` S :: He R⇒ $fail(0), σ′

ρ, 0, σ ` S ⇒ $error(E), σ′
ρ,m, σ ` S :: He R⇒ $error(E), σ′

Например, результатом выполнения тропы

100 :: sN, <"+" sN 1> :: sN = sN

является число 101.

2.10.5 Поиски

Тропа S′′ $iter S′ :: He означает, что для переменных из
He следует подобрать такие значения, что попытка вычислить
R успешно завершится. При этом полученный результат и будет
результатом всей конструкции.

Если жесткое выражениеHe — пустое, оно может быть опуще-
но вместе с ключевым словом :: . Если хвост R является ограж-
денной пустой тропой (всегда вырабатывающей пустое выраже-
ние), он может быть опущен.

S′′ $iter S′ ⇒⇒ S′′ $iter S′ :: ,
S′′ $iter S′R ⇒⇒ S′′ $iter S′ :: R
S′′ $iter S′ :: He ⇒⇒ S′′ $iter S′ :: He ,

Источник S′′ служит для вычисления начальных значений пе-
ременных из He, а источник S′ служит для вычисления новых
значений переменных из He на основе их предыдущих значений.

Считается, что источники S′′ и S′ находятся на нулевом
уровне.

Подбор значений переменных делается следующим образом.
Сначала для переменных вычисляются начальные значения. Для
этого вычисляется значение источника S′′ и полученное объект-
ное выражение E сопоставляется с образцом He. Затем делается
попытка вычислить R в новой среде. Если эта попытка заканчи-
вается неуспехом $fail(0), то вычисляется S′ и полученное объ-
ектное выражение вновь сопоставляется с He, после чего опять
делается попытка вычислить R и т. д.

S′′ $iter S′ :: He R⇒⇒
S′′ :: He, \{ R; S′ $iter S′ :: He R; }

Например, если значения переменных eA и eB еще не опреде-
лены, перестройка

eX : $l eA eB,
<Writeln eA>, <Writeln eB> $fail

эквивалентна поиску

()(eX)
$iter \{ eB : t1 e2 = (eA t1)(e2); }

:: (eA)(eB),
<Writeln eA>, <Writeln eB> $fail

2.10.6 Перестройки

Тропа S : P R имеет следующий смысл. Следует вычислить
источник S и сопоставить полученное объектное выражение E с
образцом P . В результате получится, вообще говоря, несколько
вариантов сопоставления. Затем следует попытаться найти са-
мый первый из вариантов сопоставления, для которого вычисле-
ние хвоста R завершается успешно.

Если хвост R является огражденной пустой тропой (всегда
вырабатывающей пустое выражение), он может быть опущен.

Для описания семантики перестройки нам понадобится сле-
дующее обозначение. Запись λ,m, σ `` Q ⇒ X,σ′ означает, что
если тропа Q находится на уровне m, то результатом ее вычисле-
ния для списка сред λ является X.

ρ,m, σ ` Q⇒ E , σ′
[ρ] _ λ,m, σ `` Q⇒ E , σ′

ρ,m, σ ` Q⇒ $fail(0), σ′

λ,m, σ′ `` Q⇒ X,σ′′

[ρ] _ λ,m, σ `` Q⇒ X,σ′′

ρ,m, σ ` Q⇒ $fail(k + 1), σ′

[ρ] _ λ,m, σ `` Q⇒ $fail(k + 1), σ′

ρ,m, σ ` Q⇒ $error(E), σ′
[ρ] _ λ,m, σ `` Q⇒ $error(E), σ′

[ ],m, σ `` Q⇒ $fail(0), σ

Теперь мы можем описать семантику перестройки.

S : P ⇒⇒ S : P,

ρ, 0, σ ` S ⇒ E , σ′
ρ ` E : P ⇒ λ
λ,m, σ′ `` R⇒ X,σ′′

ρ,m, σ ` S : P R⇒ X,σ′′

ρ, 0, σ ` S ⇒ $fail(k), σ′

ρ,m, σ ` S : P R⇒ $fail(0), σ′

ρ, 0, σ ` S ⇒ $error(E), σ′
ρ,m, σ ` S : P R⇒ $error(E), σ′

Например, вычисление приведенной ниже тропы заканчивает-
ся неуспехом. При этом печатается цепочка литер ’CBA’:

’ABC’ : $r e sX e, <Print sX> $fail

2.10.7 Огражденные тропы

Вычисление хвоста

, Q

всегда дает тот же результат, что и вычисление тропы Q.

, Q⇒⇒ Q

2.10.8 Отрицания условий

Хвост # S R означает, что следует вычислить S, а затем взять
“логическое отрицание” полученного результата.

Если хвост R является огражденной пустой тропой (всегда
вырабатывающей пустое выражение), он может быть опущен.

# S ⇒⇒ # S ,

ρ, 0, σ ` S ⇒ , σ′

ρ,m, σ ` # S R⇒ $fail(0), σ′

ρ, 0, σ ` S ⇒ $fail(k), σ′

ρ,m, σ′ ` R⇒ X,σ′′

ρ,m, σ ` # S R⇒ X,σ′′

ρ, 0, σ ` S ⇒ $error(E), σ′
ρ,m, σ ` # S R⇒ $error(E), σ′

2.10.9 Заборы

Хвост \? Q означает, что если при вычислении тропы Q воз-
никнет неуспех с ненулевой “серьезностью”, следует уменьшить
“серьезность” на единицу. При этом считается, что уровень тропы
Q на единицу больше, чем уровень всей конструкции.

ρ,m+ 1, σ ` Q⇒ E , σ′
ρ,m, σ ` \? Q⇒ E , σ′

ρ,m+ 1, σ ` Q⇒ $fail(0), σ′

ρ,m, σ ` \? Q⇒ $fail(0), σ′

ρ,m+ 1, σ ` Q⇒ $fail(k + 1), σ′

ρ,m, σ ` \? Q⇒ $fail(k), σ′

ρ,m+ 1, σ ` Q⇒ $error(E), σ′
ρ,m, σ ` \? Q⇒ $error(E), σ′

2.10.10 Отсечения

Хвост \! Q означает, что если при вычислении тропы Q воз-
никнет неуспех, нужно увеличить “серьезность” неуспеха на еди-
ницу. При этом считается, что уровень тропы Q на единицу мень-
ше, чем уровень всей конструкции.

ρ,m, σ ` Q⇒ E , σ′
ρ,m+ 1, σ ` \! Q⇒ E , σ′

ρ,m, σ ` Q⇒ $fail(k), σ′

ρ,m+ 1, σ ` \! Q⇒ $fail(k + 1), σ′

ρ,m, σ ` Q⇒ $error(E), σ′
ρ,m+ 1, σ ` \! Q⇒ $error(E), σ′

Например, в результате вычисления тропы, приведенной ни-
же, будет напечатана цепочка литер ’ABD’ и выдан результат ’2’.

{
\? {

<Print ’A’> $fail;
<Print ’B’> \! $fail;
<Print ’C’> = ’1’;
};

<Print ’D’> = ’2’;
}

2.10.11 Тупики

Вычисление хвоста $fail всегда дает $fail(0), т. е. неуспех
серьезности 0.

ρ,m, σ ` $fail⇒ $fail(0), σ

139



2.10.12 Правые части

Хвост = Q означает, что перед вычислением тропы Q следует
начать “новую жизнь”, т. е. считать, что Q находится на нуле-
вом уровне. Если в результате вычисления Q получится неуспех
$fail(k), то результатом хвоста = Q является $fail(m+ 1), где
m — уровень, на котором находится хвост = Q. Таким образом,
вырабатывается неуспех, силы которого достаточно, чтобы про-
бить все заборы, которые еще не отменены с помощью отсечений.

ρ, 0, σ ` Q⇒ E , σ′
ρ,m, σ ` = Q⇒ E , σ′

ρ, 0, σ ` Q⇒ $fail(k), σ′

ρ,m, σ ` = Q⇒ $fail(m+ 1), σ′

ρ, 0, σ ` Q⇒ $error(E), σ′
ρ,m, σ ` = Q⇒ $error(E), σ′

2.10.13 Аварии

Хвост $error Q позволяет породить ошибку $error(E), где
объектное выражение E является результатом вычисления тропы
Q.

ρ, 0, σ ` Q⇒ E , σ′
ρ,m, σ ` $error Q⇒ $error(E), σ′

ρ, 0, σ ` Q⇒ $fail(k), σ′

F - имя функции, в которой находится конструкция.
ρ,m, σ ` $error Q⇒ $error(F "Unexpected fail"), σ′

ρ, 0, σ ` Q⇒ $error(E), σ′
ρ,m, σ ` $error Q⇒ $error(E), σ′

2.10.14 Перехваты аварий

Хвост $trap Q $with Palt означает, что следует попытать-
ся вычислить Q. Если результатом этого вычисления является
$error(E), то дальше вычисляется выбор

E : Palt

и то, что получится, считается результатом всей конструкции.
Считается, что тропа Q находится на нулевом уровне.

ρ, 0, σ ` Q⇒ E , σ′
ρ,m, σ ` $trap Q $with Palt⇒ E , σ′

ρ, 0, σ ` Q⇒ $fail(k), σ′

ρ,m, σ′ ` F "Unexpected fail" : Palt⇒ X,σ′′

где F - имя функции, в которой находится конструкция.
ρ,m, σ ` $trap Q $with Palt⇒ X,σ′′

ρ, 0, σ ` Q⇒ $error(E), σ′
ρ,m, σ′ ` E : Palt⇒ X,σ′′

ρ,m, σ ` $trap Q $with Palt⇒ X,σ′′

2.10.15 Распутья

Источник \{Q1; Q2; . . . Qn; } означает, что следует вычис-
лять тропы Q1, Q2, . . . , Qn слева направо, пока не удастся найти
тропу, вычисление которой завершится успешно.

А именно, рассмотрим результат вычисления очередной тропы
Qj .

Если результат — объектное выражение E , то E считается
результатом всего распутья. Если результат — $error(E), то
$error(E) является результатом всего распутья. Если результат
— $fail(k + 1), то результатом вычисления всего распутья яв-
ляется $fail(k + 1). И, наконец, если результат — $fail(0), то
этот неуспех “перехватывается”, т. е. делается попытка вычислить
следующую тропу. Если же следующей тропы не существует (т. е.

j = n), то результатом вычисления всей тропы является неуспех
$fail(0).

{Q1; Q2; . . . Qn; }⇒⇒
\{Q1; Q2; . . . Qn; $error F "Unexpected fail";}

где F — имя функции, в которой находится конструкция.

ρ,m, σ ` \{} ⇒ $fail(0), σ

ρ,m, σ ` Q1 ⇒ E , σ′
ρ,m, σ ` \{Q1; Q2; . . . Qn; } ⇒ E , σ′

ρ,m, σ ` Q1 ⇒ $fail(0), σ′

ρ,m, σ′ ` \{Q2; . . . Qn; } ⇒ X,σ′′

ρ,m, σ ` \{Q1; Q2; . . . Qn; } ⇒ X,σ′′

ρ,m, σ ` Q1 ⇒ $fail(k + 1), σ′

ρ,m, σ ` \{Q1; Q2; . . . Qn; } ⇒ $fail(k + 1), σ′

ρ,m, σ ` Q1 ⇒ $error(E), σ′
ρ,m, σ ` \{Q1; Q2; . . . Qn; } ⇒ $error(E), σ′

2.10.16 Выборы

Вычисление источника S : \{Snt1; . . . Sntn; } всегда дает
тот же результат, что и вычисление тропы

S : V e, \{V e : Snt1; . . . V e : Sntn; }

при условии, что V e — некоторая e-переменная, которая больше
нигде не используется в программе.

S : {Snt1; . . . Sntn; }⇒⇒
S : \{Snt1; . . . Sntn;

e $error Fname "Unexpected fail";}

где F — имя функции, в которой находится конструкция.

ρ, 0, σ ` S ⇒ E , σ′
ρ,m, σ′ ` \{E : Snt1; . . . E : Sntn; } ⇒ X,σ′′

ρ,m, σ ` S : \{Snt1; . . . Sntn; } ⇒ X,σ′′

ρ, 0, σ ` S ⇒ $fail(k), σ′

ρ,m, σ ` S : \{Snt1; . . . Sntn; } ⇒ $fail(0), σ′

ρ, 0, σ ` S ⇒ $error(E), σ′
ρ,m, σ ` S : \{Snt1; . . . Sntn; } ⇒ $error(E), σ′

2.10.17 Результатные выражения как источники

Источник вида Re означает, что следует попытаться вычис-
лить Re, и то, что получится, считать результатом вычисления
источника.

ρ, σ ` Re⇒ X,σ′

ρ,m, σ ` Re⇒ X,σ′

2.11 Определения функций

$ ОпределениеФункции =
$ ИмяФункции ТелоФункции ";".
$ ТелоФункции =
$ ОбразцовоеРаспутье | Предложение.

Определение функции связывает имя функции с ее телом, т. е.
конструкцией, описывающей способ ее вычисления.

Если определение функции имеет вид F Snt; , оно эквива-
лентно определению F \{ Snt; }; .

Пусть определение функции F имеет вид

F Palt;

Тогда вычисление вызова этой функции, имеющего вид <F Re>,
сводится к вычислению источника E : Palt, где E — результат

вычисления результатного выражения Re. Если при этом полу-
чается объектное выражение E ′ или ошибка $error(E ′), они и
считаются результатом вычисления вызова. Если же получается
неуспех $fail(k), то дальнейшие действия зависят от того, была
ли функция F объявлена как откатная или безоткатная. Для от-
катной функции в качестве результата выдается $fail(0), а для
безоткатной — $error(F "Unexpected fail").

{}, 0, σ ` E : Palt⇒ E ′, σ′
σ ` <F E> ⇒ E ′, σ′

{}, 0, σ ` E : Palt⇒ $fail(k), σ′

Функция F — откатная, т. е.
объявлена как $func? F Fin = Fout;.

σ ` <F E> ⇒ $fail(0), σ′

{}, 0, σ ` E : Palt⇒ $fail(k), σ′

Функция F — безоткатная, т. е.
объявлена как $func F Fin = Fout;.

σ ` <F E> ⇒ $error(F "Unexpected fail"), σ′

{}, 0, σ ` E : Palt⇒ $error(E ′), σ′
σ ` <F E> ⇒ $error(E ′), σ′

В вышеприведенных правилах вывода предполагается, что
функция F имеет определение F Palt.

2.12 Объявления

2.12.1 Объявления констант

$ ОбъявлениеКонстант =
$ "$const" [ ОбъКонст { "," ОбъКонст } ] ";".
$ ОбъКонст =
$ ИмяКонстанты "=" КонстантноеВыражение.
$ ИмяКонстанты = Идентификатор.
$ КонстантноеВыражение =

$ { КонстантныйТерм | ИменованноеВыражение }.
$ КонстантныйТерм =
$ СтатическийСимвол |
$ "(" КонстантноеВыражение ")".

Объявления констант позволяют вводить для объектных вы-
ражений символические имена, с тем чтобы в дальнейшем исполь-
зовать их вместо самих выражений. В качестве имен используют-
ся слова. Если некоторому объектному выражению E присвоено
имя N , то в дальнейшем конструкция &N является изображением
этого выражения. Например, следующее объявление вводит три
имени объектных выражений:

$const Lf = 10, Cr = 13, ABC = A B C;

после чего &Lf обозначает 10, &Cr обозначает 13, а &ABC обозначает
A B C.

Объявление константы само может ссылаться на предшеству-
ющие объявления констант, ящиков, таблиц, каналов и функций.
Например, после объявления

$const CrLf_ABC = &Cr &Lf &ABC;

&CrLf_ABC начинает обозначать выражение 13 10 A B C.

2.12.2 Объявления ящиков, векторов, строк, таблиц
и каналов

$ ОбъявлениеЯщиков = "$box" { ИмяСсылки } ";".
$ ОбъявлениеВекторов = "$vector" { ИмяСсылки } ";".
$ ОбъявлениеСтрок = "$string" { ИмяСсылки } ";".
$ ОбъявлениеТаблиц = "$table" { ИмяСсылки } ";".
$ ОбъявлениеКаналов = "$channel" { ИмяСсылки } ";".
$ ИмяСсылки = Идентификатор.

Объявления ящика, вектора, строки, таблицы или канала
означают, что в момент загрузки программы следует породить
объект соответствующего типа а символ-ссылку на этот объект

обозначить с помощью символического имени, указанного в объ-
явлении объекта. Таким образом, в дальнейшем это имя может
использоваться для получения ссылки на объект.

Например, после объявления $box X; конструкция &X обозна-
чает символ-ссылку на ящик. Ниже приведены примеры объявле-
ний:

$box B1 B2 B3;
$vector V1 V2;
$table T1 T2;
$channel Input Output;

2.12.3 Объявления функций

$ ОбъявлениеФункции =
$ "$func" ИмяФункции
$ ВходнойФормат "=" ВыходнойФормат ";" |
$ "$func?" ИмяФункции
$ ВходнойФормат "=" ВыходнойФормат ";".
$ ВходнойФормат = ФорматноеВыражение.
$ ВыходнойФормат = ФорматноеВыражение.

$ ФорматноеВыражение = ЖесткоеВыражение.

Объявление функции вводит символическое имя для функции
и должно предшествовать как определению функции, так и всем
использованиям ее имени.

Определение функции описывает ограничения, которые на-
кладываются на внешний вид всех вызовов этой функции, на вид
входных образцов в ее определении и на вид результатных выра-
жений, генерирующих результат работы функции. Эти ограниче-
ния будут описаны подробно в дальнейшем.

Входной и выходной форматы должны быть жесткими, т. е.
на каждом уровне скобок может находиться максимум одна e-
переменная или v-переменная.

Индексы переменных в форматах могут опускаться, они не
оказывают никакого влияния на смысл программы, и обычно ис-

пользуются в качестве комментариев. Следует отметить, что хо-
тя контекстно-свободный синтаксис форматных выражений сов-
падает с синтаксисом жестких выражений, они все же считаются
разными конструкциями языка вследствие различной трактовки
индексов переменных.

Если в объявлении функции использовано ключевое слово
$func, это означает, что функция — безоткатная, т. е. ее вычисле-
ние может завершиться либо успешной выдачей результата, либо
выдачей ошибки.

Если в объявлении функции использовано ключевое слово
$func?, это означает, что функция — откатная, т. е. ее вычисле-
ние может завершиться либо успешной выдачей результата, либо
выдачей неуспеха, либо выдачей ошибки.

Например:

$func Interpreter (e.Program) (e.Input) = e.Result;
$func? Attempt t.Arg = s.Result1 t.Result2 (e.Result3);

2.12.4 Директивы отладки

$ ДирективаОтладки =
$ "$trace" { ИмяФункции } ";" |
$ "$traceall" ";".

Директива отладки $trace означает, что в процессе выполне-
ния программы для функций, имена которых перечислены в ди-
рективе, следует печатать отладочную информацию: имя функ-
ции и ее аргументы в момент обращения к ней, а также ее ре-
зультаты в момент выхода из нее. Директива $traceall означает,
что следует печатать отладочную информацию для всех последу-
ющих функций.

2.13 Контекстные ограничения

2.13.1 Устранение избыточных конструкций

В этом разделе мы описываем различные контекстные син-
таксические ограничения, которым должна удовлетворять любая

программа на Рефале Плюс.
Для того чтобы не загромождать изложение, мы будем предпо-

лагать, что предварительно была выполнена нормализация про-
граммы, в результате которой все конструкции, являющиеся со-
кращениями для других конструкций, заменены на эквивалент-
ные им полные конструкции следующим образом.

Хвосты и жесткие образцы, опущенные в присваиваниях, поис-
ках и перестройках, заменены на пустые образцы и пустые ограж-
денные тропы следующим образом:

S :: He ⇒ S :: He ,
S′ $iter S′′ ⇒ S′ $iter S′′ :: ,
S′ $iter S′′ :: He ⇒ S′ $iter S′′ :: He ,
S′ $iter S′′ R ⇒ S′ $iter S′′ :: R
S : P ⇒ S : P ,
# S ⇒ # S ,

Хвосты, опущенные в предложениях, заменены на пустые
огражденные тропы следующим образом:

P ⇒ P ,

“Непрозрачные” фигурные скобки { заменены на прозрачные
фигурные скобки \{ в распутьях и выборах следующим образом:

{Snt1; . . . Sntn; } ⇒
\{Snt1; . . . Sntn;

$error F "Unexpected fail";}

S : {Snt1; . . . Sntn; } ⇒
S : \{Snt1; . . . Sntn;

e $error F "Unexpected fail";}

где F — имя функции, в которой находится конструкция.
“Непрозрачные” фигурные скобки { заменены на прозрачные

фигурные скобки \{ в определениях функций следующим обра-
зом:

F {Snt1; . . . Sntn; } ⇒
F \{Snt1; . . . Sntn;

Fin $error F "Unexpected fail";}

где Fin — входной формат функции из объявления функции F (в
котором опущены индексы переменных).

Тела функций, являющиеся предложениями, заменены на со-
ответствующие образцовые распутья:

F Snt; ⇒ F \{ Snt; };

2.13.2 Ограничения, налагаемые объявлениями
функций

Объявления функций имеют вид

$func F Fin = Fout;
$func? F Fin = Fout;

где Fin — входной формат функции, т. е. форматное выражение,
задающее структуру аргумента функции, а Fout — выходной фор-
мат функции, т. е. форматное выражение, задающее структуру
результата функции. Как уже говорилось выше, индексы пере-
менных в форматах не играют никакой роли, и поэтому мы будем
в дальнейшем, без ограничения общности, предполагать, что они
отсутствуют.

Определения функций должны удовлетворять ограничениям,
налагаемым входными и выходными форматами функций. Что-
бы описать эти ограничения, мы введем на множестве форматов
отношение частичного порядка.

Пусть F ′ и F ′′ два формата. Запись F ′′ ⊇ F ′ будет означать,
что формат F ′ является (частным) случаем формата F ′′. Отно-
шение ⊇ определяется следующими правилами.

• F ⊇ F .

• Если F ′1 ⊇ F1 и F ′2 ⊇ F2, то F ′1 F ′2 ⊇ F1 F2.

• Если F1 ⊇ F2, то (F1) ⊇ (F2).

• e ⊇ F .

• Если F не имеет вида e e ... e, то v ⊇ F .

• t ⊇ S, для любого символа S.

• t ⊇ s.

• t ⊇ (F).

• s ⊇ S, для любого символа S.

Если в программе имеется результатное выражение Re, то его
форматом считается форматное выражение F , которое получает-
ся из Re следующим образом.

• У всех переменных, входящих в Re, отбрасываются индексы.

• Вызовы всех функций, появляющиеся в Re, заменяются на
выходные форматы этих функций. А именно, если в Re име-
ется вызов функции вида <F Re′>, a объявление функции F
имеет вид $func F Fin = Fout; либо $func? F Fin = Fout;,
то <F Re′> заменяется на Fout.

Если в программе имеется образец P , то его форматом счита-
ется форматное выражение F , которое получается из P следую-
щим образом.

• У всех переменных, входящих в P , отбрасываются индексы.

• Если P имеет указатель направления сопоставления, то этот
указатель отбрасывается.

Если в программе имеется жесткое выражение He, то его фор-
матом считается форматное выражение F , которое получается из
He в результате отбрасывания индексов у всех переменных, вхо-
дящих в He.

В дальнейшем F[Re] будет обозначать формат результатно-
го выражения Re, F[P ] — формат образца P , а F[He] — формат
жесткого выражения He.

Следует подчеркнуть, что формат результатного выражения
Re зависит не только от внешнего вида выражения Re, но и от
выходных форматов тех функций, вызовы которых содержатся

в Re. Тем не менее, для каждой конкретной программы F[Re]
определен однозначно.

Теперь мы можем описать ограничения, которым должны удо-
влетворять определения функций. Эти ограничения касаются ви-
да вызовов функций, вида входных образцов функций, и вида
результатов, вырабатываемых тропами.

Пусть объявление функции F имеет входной формат Fin, вы-
ходной формат Fout, а в ее определении

F Palt

образцовое распутье Palt имеет вид \{P1 R1; . . . Pn Rn;}.
Тогда должны быть выполнены следующие условия.

• Входные образцы функции P1, . . . , Pn должны удовлетво-
рять следующему ограничению: Fin ⊇ F[Pj ].

• Все вызовы функции F во всех определениях функций
должны удовлетворять следующему ограничению.

Пусть вызов этой функции имеет вид <F Re>. Тогда должно
выполнятся условие Fin ⊇ F[Re].

Чтобы описать ограничения, налагаемые на результаты, вы-
рабатываемые тропами, введем следующие обозначения.

Тот факт, что результаты, вырабатываемые тропой Q, удовле-
творяют формату F , мы будем записывать в виде F ` Q. По-
скольку хвосты, источники и результатные выражения являются
частными случаями троп, то же самое обозначение используется
и для них.

Точно так же, тот факт, что результаты, вырабатываемые об-
разцовым распутьем Palt, удовлетворяют формату F , мы будем
записывать в виде F ` Palt.

Теперь мы можем следующим образом описать ограничения,
накладываемые на определения функций выходными форматами
функций.

Если функция F имеет определение F Palt и выходной фор-
мат Fout, то должно быть выполнено Fout ` Palt.

Отношения F ` Q и F ` Palt определяются с помощью сле-
дующих правил.

F[ ] ` S
F ` R

F ` S R

F[He] ` S
F ` R

F ` S :: He R

F[He] ` S′′
F[He] ` S′
F ` R

F ` S′′ $iter S′ :: He R

F ` R

F ` S : P R

F ` Q

F ` , Q

F[ ] ` S
F ` R

F ` # S R

F ` Q

F ` \? Q

F ` Q

F ` \! Q
F ` $fail

F ` Q

F ` = Q
F ` $error Q

F ` Q
F ` Palt

F ` $trap Q $with Palt

F ` Qj для всех j = 1, . . . , n
F ` \{Q1; . . . Qn;}

F ` Palt

F ` S : Palt

F ⊇ F[Re]
F ` Re

F ` Rj для всех j = 1, . . . , n
F ` \{P1 R1; . . . Pn Rn;}

2.13.3 Ограничения на использование ссылок
на функции

Если в образцовом или результатном выражении появляется
ссылка на функцию F вида &F , то эта функция должна быть
объявлена как $func F e = e или $func? F e = e.

2.13.4 Ограничения на использование переменных

Переменные, появляющиеся в определении функции, должны
удовлетворять определенным ограничениям.

• Если переменная используется в результатном выражении,
она должна быть к этому моменту определена. Определе-
нием переменной является ее появление в образце, если к
этому моменту она еще не была определена.

• Если в каком-то месте являются определенными несколько
различных переменных, их индексы должны быть попарно
различны.

Сформулируем эти требования более точно. Для этого введем
следующие обозначения.

V[X] обозначает множество переменных, входящих в кон-
струкцию X.

{} обозначает пустое множество.
ϕ′ ∪ ϕ′′ обозначает объединение множеств ϕ′ и ϕ′′.
ϕ′ ⊆ ϕ′′ означает, что все элементы множества ϕ′ входят в ϕ′′

ϕ′ ∇ ϕ′′ обозначает пополнение множества переменных ϕ′

переменными из ϕ′′. А именно, ϕ′∇ϕ′′ содержит все перемен-
ные из ϕ′′, а также все переменные из ϕ′, индексы которых
отличаются от индексов всех переменных из ϕ′′. Например,
{sX, sY} ∇ {eY, eZ} = {sX, eY, eZ}.

Запись ϕ ` Q означает, что все переменные из ϕ имеют разные
индексы, и все переменные, значения которых могут понадобиться
для вычисления тропы Q, входят в ϕ. Поскольку хвосты, источ-
ники и результатные выражения являются частными случаями
троп, то же самое обозначение используется и для них.

Аналогично, запись ϕ ` Palt означает, что все переменные из
ϕ имеют разные индексы, и все переменные, значения которых
могут понадобиться для вычисления образцового распутья Palt,
входят в ϕ.

Теперь мы можем описать ограничения, налагаемые на ис-
пользование переменных в определениях функций, следующим
образом.

Пусть функция F имеет определение F Palt. Тогда должно
выполняться {} ` Palt.

Отношения ϕ ` Q и ϕ ` Palt определяются с помощью следу-
ющих правил.

ϕ ` S
ϕ ` R

ϕ ` S R

ϕ ` S
ϕ ∇ V[He] ` R
ϕ ` S :: He R

ϕ ` S′′

ϕ ∇ V[He] ` S′
ϕ ∇ V[He] ` R
ϕ ` S′′ $iter S′ :: He R

ϕ ` S
ϕ ∪ V[p] ` R
ϕ ` S : P R

ϕ ` Q

ϕ ` , Q

ϕ ` S
ϕ ` R

ϕ ` # S R

ϕ ` Q

ϕ ` \? Q

ϕ ` Q

ϕ ` \! Q

ϕ ` $fail
ϕ ` Q

ϕ ` = Q

ϕ ` Q

ϕ ` $error Q

ϕ ` Q
ϕ ` Palt

ϕ ` $trap Q $with Palt

ϕ ` Qj для всех j = 1, . . . , n
ϕ ` \{Q1; . . . Qn;}

ϕ ` S
ϕ ` Palt

ϕ ` S : Palt

Все переменные из ϕ имеют разные индексы
V[Re] ⊆ ϕ
ϕ ` Re

ϕ ∪ V[Pj ] ` Rj для всех j = 1, . . . , n
ϕ ` \{P1 R1; . . . Pn Rn;}

2.13.5 Ограничения на использование отсечений

Каждой тропе, входящей в определение функции, может быть
поставлено в соответствие целое неотрицательное число k, име-
нуемое ее уровнем. Если двигаться вдоль тропы, то прохождение
через \? увеличивает текущий уровень на 1, а прохождение через
\! уменьшает уровень на 1. Таким образом, каждому отсечению
\! должен однозначно соответствовать некоторый “парный” забор
\?.

Сформулируем это требование более точно. Для этого введем
следующие обозначения.

Пусть k — целое неотрицательное число, а Q — тропа. Запись
k ` Q означает, что тропе Q может быть приписан уровень k. По-
скольку хвосты, источники и результатные выражения являются
частными случаями троп, то же самое обозначение используется
и для них.

Аналогично, если Palt — образцовое распутье, то запись k `
Palt означает, что образцовому распутью Palt может быть при-
писан уровень k.

Теперь мы можем описать ограничения, налагаемые на ис-
пользование отсечений в определениях функций, следующим об-
разом.

Пусть функция F имеет определение F Palt. Тогда должно
выполняться 0 ` Palt.

Отношения k ` Q и k ` Palt определяются с помощью следу-
ющих правил.

0 ` S
k ` R

k ` S R

0 ` S
k ` R

k ` S :: He R

0 ` S′′

0 ` S′

k ` R

k ` S′′ $iter S′ :: He R

0 ` S
k ` R

k ` S : P R

k ` Q

k ` , Q

0 ` S
k ` R

k ` # S R

k + 1 ` Q

k ` \? Q

k ` Q

k + 1 ` \! Q
k ` $fail 0 ` Q

k ` = Q

0 ` Q

k ` $error Q

0 ` Q
k ` Palt

k ` $trap Q $with Palt

k ` Qj для всех j = 1, . . . , n
k ` \{Q1; . . . Qn;}

0 ` S
k ` Palt

k ` S : Palt

k ` Re

k ` Rj для всех j = 1, . . . , n
k ` \{P1 R1; . . . Pn Rn;}

2.14 Модули

Каждая программа на Рефале Плюс состоит из одного или
нескольких модулей. Каждый модуль состоит из двух частей: ин-
терфейса модуля и реализации модуля.

Интерфейс модуля содержит те части модуля, которые доступ-
ны из других модулей, в то время как реализация модуля содер-
жит те части модуля, которые недоступны из других модулей.

Каждый модуль MMMM занимает два файла. А именно, интер-
фейс модуля хранится в файле MMMM.rfi, а реализация — в файле
MMMM.rf.

$ ИнтерфейсМодуля =
$ { Объявление }.
$ Объявление =
$ ОбъявлениеКонстант | ОбъявлениеЯщиков |
$ ОбъявлениеВекторов | ОбъявлениеСтрок |
$ ОбъявлениеТаблиц | ОбъявлениеКаналов |
$ ОбъявлениеФункции.

$ РеализацияМодуля =
$ { Импорт } { ДирективаРеализации }.
$ ДирективаРеализации =
$ Объявление |
$ ДирективаОтладки |
$ ОпределениеФункции.

$ Импорт = "$use" { ИмяМодуля } ";".
$ ИмяМодуля = Идентификатор.

Если внутри реализации модуля XXXX требуется получить до-
ступ к именам, объявленным в интерфейсе модуля YYYY, следует
поместить внутри реализации модуля XXXX директиву $use YYYY
следующим образом:

// Интерфейс модуля XXXX.
......

Файл XXXX.rfi

// Реализация модуля XXXX.
$use ... YYYY ... ;
// С этого места доступны имена,
// объявленные в YYYY.rfi
......

Файл XXXX.rf

// Интерфейс модуля YYYY.
......

Файл YYYY.rfi

// Реализация модуля YYYY.
......

Файл YYYY.rf

Некоторые операционные системы игнорируют различие меж-
ду строчными и прописными буквами в именах файлов. Напри-
мер, StdIO.rfi, StdIo.rfi, Stdio.rfi рассматриваются как вари-
анты имени одного и того же файла. Поэтому не рекомендуется
использовать в одной и той же программе имена модулей, отли-
чающиеся только размером букв.

2.15 Исполнение программы

Программа на Рефале Плюс может состоять из нескольких
модулей, один из которых обязательно должен экспортировать
функцию Main. Эта функция называется главной функцией про-
граммы. Главная функция всегда должна иметь объявление

$func Main = e;

Если же из какого-то модуля экспортируется функция с именем
Main, имеющая другое объявление, это считается ошибкой.

Исполнение Рефал-программы заключается в вычислении вы-
зова функции Main с пустым аргументом

<Main >

Модуль, содержащий определение главной функции, может не
иметь интерфейсной части, поскольку если компилятор не нахо-
дит файл, содержащий интерфейс модуля, он считает, что интер-
фейс модуля состоит из единственного объявления функции

$func Main = e;

159



Глава 3

Библиотека функций

3.1 Использование библиотечных функций

Составной частью системы Рефал Плюс является библиотека
функций, состоящая из нескольких модулей.

Если в каком-то модуле пользователя используется библиотеч-
ная функция, входящая в библиотечный модуль MMMM, в начале
модуля пользователя следует поместить директиву

$use MMMM;

В результате этого в модуль пользователя будут импортирова-
ны объявления всех функций, входящих в библиотечный модуль
MMMM.

В настоящее время библиотека функций состоит из следую-
щих модулей:

Access — функции прямого доступа к частям выра-
жений;

Apply — функция для вызова функций, передан-
ных через параметры;

Arithm — функции целочисленной арифметики;
Bit — операции с цепочками битов;
Box — функции для работы с ящиками;

Class — предикаты для проверки принадлежно-
сти символов к различным классам сим-
волов;

Compare — предикаты для сравнения выражений;
Convert — функции для выполнения преобразова-

ний между различными типами данных;
Dos — функции для связи с операционной систе-

мой;
StdIO — объявления стандартных каналов ввода-

вывода и функции для стандартного
ввода-вывода;

String — функции для работы со строками;
Table — функции для работы с таблицами;
Vector — функции для работы с векторами.

В дальнейшем в библиотеку могут быть добавлены и другие
модули.

3.2 Access: прямой доступ к выражениям

$func Length e.Exp = s.ExpLen;
$func? Left s.Left s.Len e.Exp = e.SubExp;
$func? Right s.Right s.Len e.Exp = e.SubExp;
$func? Middle s.Left s.Right e.Exp = e.SubExp;
$func? L s.Left e.Exp = t.SubExp;
$func? R s.Right e.Exp = t.SubExp;

Эти функции используются для прямого доступа к частям вы-
ражений. Их аргументы s.Left, s.Right и s.Len должны быть
неотрицательными целыми числами. e.Exp может быть произ-
вольным объектным выражением.

Если s.Left, s.Right или s.Len не являются неотрицатель-
ными целыми числами, то результатом этих функций является
ошибка: $error(F "Invalid argument"), где F — имя функции.

Length выдает длину выражения e.Exp, измеренную в термах.
А именно, если объектное выражение E имеет вид T1 T2 . . . Tn,

то n — его длина.
Например:

<Length > => 0
<Length A B C> => 3
<Length (A B) C (D E)> => 3

Left отбрасывает от e.Exp первые s.Left термов, а затем от
того что останется берет первые s.Len термов и выдает получен-
ное выражение в качестве результата.

Right отбрасывает от e.Exp последние s.Right термов, а за-
тем от того что останется берет последние s.Len термов и выдает
полученное выражение в качестве результата.

Middle отбрасывает от e.Exp первые s.Left и последние
s.Right термов, и выдает полученное выражение в качестве ре-
зультата.

L отбрасывает от e.Exp первые s.Left термов, и выдает пер-
вый терм полученного выражения в качестве результата.

R отбрасывает от e.Exp последние s.Right термов, и выдает
последний терм полученного выражения в качестве результата.

Если e.Exp имеет недостаточно большую длину, чтобы можно
было выполнить требуемую операцию, результатом работы любой
из вышеперечисленных функций является $fail(0).

Например:

<Middle 2 3 A B C D E F> => C
<Middle 2 3 A B C D> => $fail(0)
<Middle 0 0 A B C> => A B C
<Left 2 3 A B C D E F> => C D E
<Left 2 3 A B C D> => $fail(0)
<Left 0 0 A B C> =>
<Right 2 3 A B C D E F> => B C D
<Right 2 3 A B C D> => $fail(0)
<Right 0 0 A B C> =>
<L 2 A B C D E F> => C
<L 2 A B> => $fail(0)
<R 2 A B C D E F> => D
<R 2 A B> => $fail(0)

Операции Middle, Left и Right можно изобразить в виде сле-
дующей схемы:

s.Left s.Right
<Middle s.Left s.Right e.Exp>

s.Left s.Len
<Left s.Left s.Len e.Exp>

s.Len s.Right
<Right s.Right s.Len e.Exp>

3.3 Apply: вызов функций, переданных
через параметры

$func? Apply s.Name e.Exp = e.Exp;

Apply применяет функцию, на которую указывает ссылка
s.Name к выражению e.Exp и выдает результат этого вычисле-
ния.

3.4 Arithm: целочисленная арифметика

$func Add s.Int1 s.Int2 = s.Int;
$func Sub s.Int1 s.Int2 = s.Int;
$func Mult s.Int1 s.Int2 = s.Int;
$func DivRem s.Int1 s.Int2 = s.Quo s.Rem;
$func Div s.Int1 s.Int2 = s.Quo;
$func Rem s.Int1 s.Int2 = s.Rem;
$func GCD s.Int1 s.Int2 = s.GCD;

Эти функции выполняют арифметические операции над целы-
ми со знаком неограниченной разрядности. Каждый из аргумен-
тов арифметических функций должен представлять собой один
символ-число.

Если хотя бы один из аргументов функций арифметики не
является символом-числом, результатом является ошибка:
$error(F "Invalid argument"), где F — имя функции.

Если оба аргумента — числа, эти функции выдают следующие
результаты.

Add выдает сумму двух аргументов, Sub — разность первого и
второго аргументов, Mult — произведение двух аргументов, Div
— результат деления нацело первого аргумента на второй, Rem —
остаток от деления нацело первого аргумента на второй, DivRem
— результат деления нацело и остаток от деления нацело первого
аргумента на второй, GCD — наибольший общий делитель двух
аргументов.

Если результат одной из функций Add, Sub или Mult больше,
чем допускается реализацией, результатом является ошибка:
$error(F "Size limit exceeded"), где F — имя функции.

Например:

<Add 3 5> => 8
<Add 3 -5> => -2
<Sub 3 -5> => 8
<Mult -2 3> => -6
<Div 5 2> => 2
<Rem 5 2> => 1
<DivRem 5 2> => 2 1
<Div 6 2> => 3
<Rem 6 2> => 0
<DivRem 6 2> => 3 0

Знаки частного и остатка определяются следующим образом.
Если знаки делимого и делителя совпадают — частное положи-
тельно, если различны — отрицательно. Знак остатка совпадает
со знаком делимого. Таким образом, всегда выполняется соотно-
шение

делимое = (частное ∗ делитель) + остаток

Например:

<Div 5 3> => 1
<Rem 5 3> => 2
<Div 5 -3> => -1

<Rem 5 -3> => 2
<Div -5 3> => -1
<Rem -5 3> => -2
<Div -5 -3> => 1
<Rem -5 -3> => -2

Попытка делить на ноль приводит к выдаче ошибки:
$error(F "Divide by zero"), где F — имя функции.

Например:

<Div 5 0> => $error(Div "Divide by zero")
<Rem 5 0> => $error(Rem "Divide by zero")
<DivRem 5 0> => $error(DivRem "Divide by zero")

Функция GCD, если оба ее аргумента не равны нулю од-
новременно, выдает положительное целое число, которое яв-
ляется наибольшим общим делителем аргументов. Если же
оба аргумента равны нулю, результатом является ошибка:
$error(GCD "Zero arguments"). Например:

<GCD 6 15> => 3
<GCD -6 15> => 3
<GCD 15 1> => 1
<GCD 15 0> => 15
<GCD 0 0> => $error(GCD "Zero arguments")

3.5 Bit: Операции с цепочками битов

Модуль Bit содержит функции, которые работают с цепочка-
ми двоичных цифр, представленными целыми числами.

Каждое целое число изображает цепочку двоичных цифр бес-
конечную влево, которая получится, если записать это число как
двоичное число бесконечно большой разрядности, представленное
в дополнительном коде. При этом в случае положительного числа
эта цепочка содержит конечное число единиц, а в случае отрица-
тельного — конечное число нулей. Например:

+3 ...000011
+2 ...000010
+1 ...000001
+0 ...000000
-1 ...111111
-2 ...111110
-3 ...111101

Разряды нумеруются справа налево, начиная с нуля.

$func BitOr s.Int1 s.Int2 = s.Int;
$func BitAnd s.Int1 s.Int2 = s.Int;
$func BitXor s.Int1 s.Int2 = s.Int;

BitOr выдает поразрядное “или” двух аргументов.
BitAnd выдает поразрядное “и” двух аргументов.
BitXor выдает поразрядное исключающее “или” двух аргумен-

тов.

$func BitNot s.Int = s.Int;

BitNot выдает поразрядное логическое отрицание аргумента.

$func BitLeft s.Int s.Shift = s.Int;
$func BitRight s.Int s.Shift = s.Int;

BitLeft выдает результат логического сдвига s.Int на
s.Shift разрядов влево, если s.Shift неотрицательно, или на ми-
нус s.Shift разрядов вправо, если s.Shift отрицательно.

BitRight выдает результат логического сдвига s.Int на
s.Shift разрядов вправо, если s.Shift неотрицательно, или на
минус s.Shift разрядов влево, если s.Shift отрицательно.

$func? BitTest s.Int s.Pos = ;

BitTest выдает неуспех, если разряд s.Pos числа s.Int равен
нулю, в противном случае результатом является пустое выраже-
ние.

$func BitSet s.Int s.Pos = s.Int;
$func BitClear s.Int s.Pos = s.Int;

BitSet устанавливает разряд s.Pos числа s.Int в 1 и выдает
полученное число в качестве результата.

BitClear устанавливает разряд s.Pos числа s.Int в 0 и выдает
полученное число в качестве результата.

$func BitLength s.Int = s.Len;

BitLength выдает “длину” числа s.Int. А именно, если s.Int
неотрицательно, выдается номер позиции самого правого нуля,
слева от которого нет ни одной единицы. Если же число отрица-
тельно, выдается номер самой правой единицы, слева от которой
нет ни одного нуля.

Например:

<BitLength 3> ==> 2
<BitLength 2> ==> 2
<BitLength 1> ==> 1
<BitLength 0> ==> 0
<BitLength -1> ==> 0
<BitLength -2> ==> 1
<BitLength -3> ==> 2

3.6 Box: работа с ящиками

$func Box e.Exp = s.Box;
$func Get s.Box = e.Exp;
$func Store s.Box e.Exp = ;

Box создает новый ящик, помещает в него выражение e.Exp и
выдает ссылку на этот ящик.

Get выдает содержимое ящика, на который ссылается s.Box.
Store заносит в ящик, на который ссылается s.Box, новое со-

держимое e.Expr.

3.7 Class: предикаты для распознавания
классов символов

$func? IsBox e.Exp = ;

$func? IsChannel e.Exp = ;
$func? IsChar e.Exp = ;
$func? IsDigit e.Exp = ;
$func? IsFunc e.Exp = ;
$func? IsInt e.Exp = ;
$func? IsLetter e.Exp = ;
$func? IsString e.Exp = ;
$func? IsTable e.Exp = ;
$func? IsVector e.Exp = ;
$func? IsWord e.Exp = ;

Эти функции служат для проверки, что e.Exp является сим-
волом, принадлежащим к определенному множеству символов.

Если e.Exp — не символ, результатом является $fail(0).
Если e.Exp — символ, выполняется проверка, что этот сим-

вол принадлежащит к соответствующему множеству. Если при-
надлежит — результатом является пустое выражение, если же не
принадлежит — результатом является $fail(0).

Соответствие между функциями-предикатами и множествами
символов следующее:

IsBox — ссылки на ящики;
IsChannel — ссылки на каналы;
IsChar — символы-литеры;
IsDigit — символы-литеры, которые соответствуют

десятичным цифрам;
IsFunc — ссылки на функции;
IsInt — целые числа;
IsLetter — символы-литеры, которые соответствуют

латинским прописным и строчным бук-
вам;

IsString — ссылки на строки;
IsTable — ссылки на таблицы;
IsVector — ссылки на векторы;
IsWord — слова.

3.8 Compare: сравнение выражений

$func? Eq (e.Exp1)(e.Exp2) = ;
$func? Ne (e.Exp1)(e.Exp2) = ;
$func? Ge (e.Exp1)(e.Exp2) = ;
$func? Gt (e.Exp1)(e.Exp2) = ;
$func? Le (e.Exp1)(e.Exp2) = ;
$func? Lt (e.Exp1)(e.Exp2) = ;

Эти функции сравнивают два выражения e.Exp1 и e.Exp2 и
проверяют, выполнено ли соответствующее отношение между ни-
ми: Eq — равно, Ne — не равно, Ge — больше или равно, Gt —
больше чем, Le — меньше или равно, Lt — меньше чем.

Если условие выполнено, результат работы — пусто, иначе —
$fail(0).

$func Compare (e.Exp1)(e.Exp2) = s.Res; // ’<’, ’>’, ’=’

Compare сравнивает два выражения e.Exp1 и e.Exp2. Резуль-
татом является ’<’, если e.Exp1 меньше чем e.Exp2, ’>’ если
e.Exp1 больше чем e.Exp2, и ’=’, если e.Exp1 равно e.Exp2.

Объектные выражения сравниваются в соответствии со следу-
ющим отношением линейного порядка <.

Для любых двух выражений E ′ и E ′′ всегда либо E ′ < E ′′, либо
E ′ = E ′′, либо E ′′ < E ′.

Два выражения E ′ = T ′1 . . . T ′m и E ′′ = T ′′1 . . . T ′′n сравнивают-
ся лексикографически, т. е. составляющие их термы сравниваются
попарно слева направо, пока не будет найдена пара неравных тер-
мов T ′k и T ′′k . Если при этом T ′k < T ′′k , то считается, что E ′ < E ′′.

Если окажется, что E ′ короче, чем E ′′, и все термы из E ′ равны
соответствующим термам из E ′′, то считается, что E ′ < E ′′.

Более формально: для любых объектных выражений E , E ′, E ′′
и любых объектных термов T , T ′, T ′′ имеет место

• Если E ′ < E ′′, то T E ′ < T E ′′.

• Если T ′ < T ′′, то T ′ E ′ < T ′′ E ′′.

• � < T E , где через � обозначено пустое объектное выраже-
ние.

Для объектных термов отношение порядка определяется сле-
дующим образом.

Все символы предшествуют термам вида (E), т. е. всегда

S < (E)

Сравнение термов вида (E) сводится к сравнению их содер-
жимого, т. е.

Если E ′ < E ′′, то (E ′) < (E ′′).

Каждый символ принадлежит одному и только одному из сле-
дующих множеств:

символы-литеры;
символы-слова;
символы-числа;
символы-ссылки.

Эти множества мы будем называть типами символов. Если два
символа принадлежат одному и тому же типу, они называются
однотипными, в противном случае — разнотипными. Мы счита-
ем что на множестве типов определено отношение порядка, и что
типы были перечислены выше в порядке возрастания, т. е. мно-
жество символов-литер предшествует множеству символов-слов и
т. д.

Если два символа S ′ и S ′′ принадлежат двум различным типам
Type′ и Type′′, и при этом Type′ < Type′′, то считается, что S ′ <
S ′′.

Однотипные символы сравниваются следующим образом.
Символы-литеры упорядочены в соответствии с их кодами.
Символы-слова преобразуются в цепочки символов-литер, со-

ответствующие изображениям слов, а эти цепочки сравниваются
как описано выше.

Символы-числа сравниваются в соответствии с естественным
порядком на множестве чисел.

Отношение порядка на множестве символов-ссылок зависит от
реализации.

3.9 Convert: преобразования между различ-
ными типами данных

$func ToLower e.Char = e.Char;
$func ToUpper e.Char = e.Char;
$func CharsToBytes e.Char = e.Int;
$func BytesToChars e.Int = e.Char;
$func ToChars e.Exp = e.Char;
$func ToWord e.Exp = s.Word;
$func? ToInt e.Exp = s.Int;

ToLower преобразует цепочку символов-литер в цепочку литер,
в которой все прописные латинские буквы заменены на соответ-
ствующие строчные буквы.

ToUpper преобразует цепочку символов-литер в цепочку ли-
тер, в которой все строчные латинские буквы заменены на соот-
ветствующие прописные буквы.

CharsToBytes преобразует цепочку символов-литер в последо-
вательность чисел, являющихся их кодами ASCII.

Если аргумент этих функций не является цепочкой символов-
литер, результатом является $error(F "Invalid argument"),
где F — имя функции.

Например:

<ToLower ’AbCd+’> => ’abcd+’
<ToLower 25> =>

$error(ToLower "Invalid argument")
<ToUpper ’AbCd+’> => ’ABCD+’
<ToUpper 25> =>

$error(ToUpper "Invalid argument")
<CharsToBytes ’ABC’> => 65 66 67

BytesToChars преобразует цепочку целых чисел, каждое из ко-
торых лежит в диапазоне от 0 до 255, в цепочку символов-литер,
имеющих соответствующие коды ASCII.

Например:

<BytesToChars 65 66 67> => ’ABC’

Функции ToChars, ToWord и ToInt получают на входе произ-
вольное объектное выражение и начинают свою работу с того, что
преобразуют его в цепочку литер. Это преобразование выполняет-
ся следующим образом. Символы-литеры переходят сами в себя,
круглые скобки заменяются на литеры ’(’ и ’)’, слова заменяют-
ся на цепочки литер, соответствующие их печатным именам, чис-
ла заменяются на их изображения в виде цепочек литер, ссылки
на строки заменяются на содержимое строк, все прочие ссылки
заменяются на некоторые цепочки литер, которые зависят от ре-
ализации.

Если размер цепочки литер, полученной в результате выше-
описанного преобразования, больше, чем допускается реализаци-
ей, результатом работы функций является

$error(F "Argument too large for conversion").

Дальнейшая работа функций ToChars, ToWord и ToInt проис-
ходит следующим образом.

ToChars выдает полученную цепочку литер в качестве резуль-
тата.

<ToChars "John"> => ’John’
<ToChars ’John’> => ’John’
<ToChars 326> => ’326’
<ToChars -326> => ’-326’
<ToChars (-326) "John"> => ’(-326)John’

ToWord преобразует полученную цепочку литер в слово с соот-
ветствующим печатным именем.

<ToWord "John"> => "John"

<ToWord ’John’> => "John"
<ToWord 326> => "326"
<ToWord -326> => "-326"
<ToWord (-326) "John"> => "(-326)John"

ToInt преобразует полученную цепочку литер в целое число
s.Int. Если эта цепочка литер не является правильным изобра-
жением целого числа, результатом является $fail(0).

Например:

<ToInt ’326’> => 326
<ToInt ’+326’> => 326
<ToInt "-3" ’26’> => -326
<ToInt -32 006> => -326
<ToInt ’John’> => $fail(0)

3.10 Dos: связь с операционной системой

$func Arg s.Int = e.Arg;
$func Args = e.Args;
$func GetEnv e.VarName = e.Value;
$func Time = e.String;
$func Exit s.ReturnCode = ;
$func Delay s.MSeconds = ;
$func Sleep s.Seconds = ;
$func Random s.Max = s.Rand; // 0 <= s.Rand < s.Max
$func Randomize = ;

Эти функции служат для связи с операционной системой.
Аргументы функций должны удовлетворять следующим огра-

ничениям. s.Int должно быть целым неотрицательным чис-
лом, e.VarName — цепочкой литер и слов, s.ReturnCode — це-
лым числом из интервала от 0 до 255. Если хотя бы од-
но из этих условий не выполнено, функции выдают результат
$error(F "Invalid argument"), где F — имя функции.

Arg выдает аргумент командной строки с номером s.Int в ви-
де цепочки символов-литер. Если он отсутствует, выдается пустое
выражение.

Args выдает аргументы командной строки в виде последова-
тельности, каждый элемент которой заключен в скобки и пред-
ставляет собой цепочку символов-литер.

GetEnv получает значение переменной окружения с именем
e.VarName и выдает его в виде цепочки символов-литер.

Time выдает текущую дату и время в виде объектного выра-
жения вида

DD MMM Y Y Y Y HH:MM:SS.SS

где DD — день месяца, MMM — сокращенное название месяца
("Jan", "Feb", "Mar", "Apr", "May", "Jun", "Jul", "Aug", "Sep",
"Oct", "Nov", "Dec"), Y Y Y Y — номер года, HH:MM:SS.SS —
часы, минуты, секунды и сотые доли секунды. DD, Y Y Y Y , HH,
MM , SS представлены в виде чисел, MMM — слово, все разде-
лители между ними — символы-литеры ’ ’, ’:’ и ’.’.

Exit завершает выполнение программы, при этом устанавли-
вается код завершения, равный s.ReturnCode. Если программа
завершается обычным способом, т. е. вычисление вызова главной
функции <Main> завершается, то код завершения зависит от ре-
зультата, выданного функцией Main. Если этот результат — объ-
ектное выражение, то код завершения равен 0, если же результат
имеет вид $error(E), код завершения равен 100.

Delay задерживает исполнение программы на s.MSeconds мил-
лисекунд. Точность, с которой измеряется время, не превышает
одной сотой секунды, и не может быть лучше, чем точность ча-
сов в MSDOS.

Sleep задерживает исполнение программы на s.Seconds се-
кунд. Точность, с которой измеряется время, не превышает одной
сотой секунды, и не может быть лучше, чем точность часов в
MSDOS.

Random выдает псевдослучайное целое число в интервале от 0
до s.Max минус 1.

Randomize инициализирует генератор случайных чисел слу-
чайным начальным значением.

3.11 StdIO: стандартный ввод-вывод

$channel StdIn StdOut StdErr;

StdIn, StdOut и StdErr — стандартные каналы ввода-вывода.
Они автоматически открываются перед началом работы програм-
мы и автоматически закрываются по окончании работы програм-
мы.

$func Channel = s.Channel;

Channel создает новый канал s.Channel.

$func? OpenFile s.Channel e.FileName s.Mode = ;
$func CloseChannel s.Channel = ;

OpenFile открывает канал s.Channel и связывает его с фай-
лом с именем e.FileName. s.Mode — это символ-слово, который
указывает режим, в котором будет происходить работа с файлом:
"r" или "R" — чтение, "w" или "W" — запись, "a" или "A" — добав-
ление в конец файла. Если файл открыть не удается, результатом
работы OpenFile является неуспех $fail(0).

CloseChannel закрывает канал s.Channel.

$func? IsEof s.Channel = ;

IsEof проверяет, что текущая позиция у файла, с которым
связан канал s.Channel, находится в конце файла.

$func? Read = t.Term;
$func? ReadChar = s.Char;
$func? ReadLine = e.Char;
$func Write e.Exp = ;
$func WriteLn e.Exp = ;
$func Print e.Exp = ;
$func PrintLn e.Exp = ;

Read читает из канала &StdIn очередное изображение терма.
Если не осталось ни одного терма — выдает $fail(0).

ReadChar читает из канала &StdIn очередную литеру. Если не
осталось ни одной литеры — выдает $fail(0).

ReadLine читает из канала &StdIn все литеры до ближайшего
конца строки (включительно) и выдает их в качестве результата
(не включая литеру конец строки). Если не осталось ни одной
литеры — выдает $fail(0).

Write пишет в канал &StdOut изображение выражения e.Exp
(если e.Exp не содержит динамических символов, его можно вве-
сти обратно, терм за термом, с помощью функции Read).

WriteLn делает то же, что и Write, но в конце, добавляет ли-
теру конца строки.

Print преобразует выражение e.Exp в цепочку литер так же,
как это делает функция ToChars, и выводит его в канал &StdOut.

PrintLn делает то же, что и Print, но в конце добавляет литеру
конца строки.

$func? ReadCh s.Channel = t.Term;
$func? ReadCharCh s.Channel = s.Char;
$func? ReadLineCh s.Channel = e.Char;
$func WriteCh s.Channel e.Exp = ;
$func WriteLnCh s.Channel e.Exp = ;
$func PrintCh s.Channel e.Exp = ;
$func PrintLnCh s.Channel e.Exp = ;

Эти функции отличаются от соответствующих функций без
окончания Ch тем, что операции выполняются над каналом
s.Channel.

3.12 String: работа со строками

$func String e.Source = s.String;
$func StringInit s.String s.Len s.Fill = ;
$func StringFill s.String s.Fill = ;
$func StringLength s.String = s.Len;

$func StringRef s.String s.Index = s.Char;
$func StringSet s.String s.Index s.Char = ;
$func StringReplace s.String e.Source = ;
$func Substring s.String s.Index s.Len =

s.NewString;
$func SubstringFill s.String s.Index s.Len s.Fill = ;

Эти функции используются для создания строк, изменения
их содержимого и доступа к их содержимому. Их аргументы
должны удовлетворять следующим требованиям. s.String дол-
жен быть ссылкой на строку. s.Index и s.Len должны быть неот-
рицательными целыми числами. s.Fill может быть произволь-
ным символом-литерой. e.Source должен быть последовательно-
стью ссылок на строки, символов-слов и символов-литер.

Если хотя бы одно из этих условий не выполнено, результатом
функций является $error(F "Invalid argument"), где F — имя
функции.

В каждый момент времени строка содержит конечную (мо-
жет быть пустую) последовательность символов-литер, которую
мы будем называть содержимым строки. Если строка содержит
последовательность из n + 1 символов-литер C0, C1, . . . , Cn, мы
будем говорить, что строка имеет длину n+1, а содержимое стро-
ки будем изображать в виде

C0 C1 . . . . . . Cn

Таким образом, компоненты строки C0, C1, . . . , Cn нумеруются
начиная с нуля.

String создает новую строку и выдает ссылку на нее в каче-
стве результата. При этом содержимое создаваемой строки фор-
мируется из e.Source следующим образом.

Пусть значение параметра e.Source имеет следующий вид: S1

S2 . . . Sm, где каждый из символов Sj является либо символом-
литерой, либо символом-словом, либо ссылкой на строку. Тогда
каждый из Sj преобразуется следующим образом.

Если Sj является символом-литерой C, он остается без изме-
нения.

Если Sj является символом-словом, он заменяется на цепочку
литер, которая является печатным именем этого символа.

Если Sj является ссылкой на какую-то строку, он заменяется
на содержимое этой строки (при этом состояние самой строки не
изменяется).

Преобразованное таким образом значение параметра e.Source
становится содержимым новой строки.

StringInit изменяет содержимое строки, на которую указы-
вает s.String. Старое содержимое уничтожается и вместо него
создается новое содержимое длины s.Len, причем каждая компо-
нента строки принимает значение s.Fill.

StringFill изменяет содержимое строки, на которую указы-
вает s.String. При этом длина строки не меняется, а каждая
компонента строки принимает значение s.Fill.

StringLength выдает длину строки, на которую указывает
s.String.

StringRef извлекает из строки, на который указывает
s.String, содержимое ее компоненты с номером s.Index, кото-
рое и выдается в качестве результата.

StringSet изменяет содержимое строки, на которую указыва-
ет s.String. При этом длина строки не меняется, а компонента
строки с номером s.Index принимает значение s.Char.

StringReplace изменяет содержимое строки, на которую ука-
зывает s.String. Старое содержимое уничтожается, а новое со-
держимое формируется из значения параметра e.Source точно
таким же способом, как и в случае функции String.

Substring создает новую строку и выдает ссылку на нее в
качестве результата. При этом содержимое создаваемой строки
формируется следующим образом. Рассматривается содержимое
исходной строки, т. е. строки, на которую указывает s.String.
Пусть оно имеет вид C0 C1 . . . Cn. Тогда от него отбрасываются
первые s.Index символов, а первые s.Len из оставшихся символов
считаются содержимым новой строки.

В любом случае, в результате работы функции Substring со-
стояние исходной строки не изменяется.

SubstringFill изменяет содержимое строки, на которую ука-

зывает s.String. При этом длина строки не меняется, а s.Len по-
следовательно расположенных компонент строки, начиная с ком-
поненты с номером s.Index, принимают значение s.Char.

Если размер исходной строки недостаточен для выполнения
одной из вышеуказанных операций, в качестве результата выда-
ется

$error(F "Index out of range"), где F — имя функции, а
состояние строки не меняется.

Если при выполнении одной из вышеуказанных операций воз-
никает необходимость создать содержимое строки, длина которо-
го больше, чем допускается реализацией, в качестве результата
выдается
$error(F "Size limit exceeded"), где F — имя функции, а со-
стояние строки не меняется.

3.13 Table: работа с таблицами

$func Table = s.Tab;
$func Bind s.Tab (e.Key) (e.Val) = ;
$func Unbind s.Tab e.Key = ;
$func? Lookup s.Tab e.Key = e.Val;
$func? IsInTable s.Tab e.Key = ;
$func Domain s.Tab = e.Domain ;
$func TableCopy s.Tab = s.TabCopy;
$func ReplaceTable s.TargetTable s.SourceTable = ;

Table создает новую пустую таблицу и выдает символ-ссылку
на эту таблицу.

Bind связывает ключ e.Key со значением e.Val в таблице, на
которую ссылается s.Tab.

Unbind удаляет из таблицы, на которую ссылается s.Tab, ключ
e.Key и связанное с ним значение, если такой ключ был в таблице.

Lookup находит в таблице, на которую ссылается s.Tab, значе-
ние, связанное с ключом e.Key и выдает его. Если такого ключа
в таблице нет — результатом является $fail(0).

IsInTable проверяет, что ключ e.Key находится в таблице, на
которую ссылается s.Tab.

Domain выдает список ключей, содержащихся в таблице, на
которую ссылается s.Tab. А именно, если таблица содержит мно-
жество ключей {E1, E2, . . . , En}, e.Domain имеет вид

(E1)(E2) . . . (En)

где порядок расположения ключей описанием языка не определен.
TableCopy создает новую таблицу, копирует в него содержи-

мое таблицы, на которую ссылается s.Tab, и выдает в качестве
результата ссылку на новую таблицу.

ReplaceTable уничтожает прежнее содержимое таблицы, на
которую ссылается s.TargetTable, и заменяет его на копию со-
держимого таблицы, на которую ссылается s.SourceTable.

3.14 Vector: работа с векторами

$func Vector e.Source = s.Vector;
$func VectorToExp s.Vector = e.Exp;
$func VectorInit s.Vector s.Len e.Fill = ;
$func VectorFill s.Vector e.Fill = ;
$func VectorLength s.Vector = s.Len;
$func VectorRef s.Vector s.Index = e.Exp;
$func VectorSet s.Vector s.Index e.Exp = ;
$func VectorReplace s.Vector e.Source = ;
$func Subvector s.Vector s.Index s.Len =

s.NewVector;
$func SubvectorFill s.Vector s.Index s.Len e.Fill = ;

Эти функции используются для создания векторов, измене-
ния их содержимого и доступа к их содержимому. Их аргументы
должны удовлетворять следующим требованиям. s.Vector дол-
жен быть ссылкой на вектор. s.Index и s.Len должны быть неот-
рицательными целыми числами. e.Fill может быть произволь-
ным объектным выражением. e.Source должен быть последова-
тельностью ссылок на векторы и термов вида (E).

Если хотя бы одно из этих условий не выполнено, результатом
функций является $error(F "Invalid argument"), где F — имя
функции.

В каждый момент времени вектор содержит конечную (может
быть, пустую) последовательность объектных выражений, кото-
рую мы будем называть содержимым вектора. Если вектор со-
держит последовательность из n + 1 объектных выражений E0,
E1, . . ., En, мы будем говорить, что вектор имеет длину n + 1, а
содержимое вектора будем изображать в виде

(E0)(E1) . . . (En)

Таким образом, компоненты вектора E0, E2, . . ., En нумеруются
начиная с нуля.

Vector создает новый вектор и выдает ссылку на него в ка-
честве результата. При этом содержимое создаваемого вектора
формируется из e.Source следующим образом.

Пусть значение параметра e.Source имеет следующий вид: T1
T2 . . . Tm, где каждый из объектных термов Tj либо имеет вид
(E), либо является ссылкой на вектор. Тогда каждый из Tj пре-
образуется следующим образом.

Если Tj имеет вид (E), он остается без изменения.
Если Tj является ссылкой на какой-то вектор, он заменяется

на содержимое этого вектора (при этом состояние самого вектора
не изменяется).

Преобразованное таким образом значение параметра e.Source
становится содержимым нового вектора.

VectorToExp выдает в виде объектного выражения содержи-
мое вектора, на который указывает s.Vector.

VectorInit изменяет содержимое вектора, на который указы-
вает s.Vector. Старое содержимое уничтожается и вместо него
создается новое содержимое длины s.Len, причем каждая компо-
нента вектора принимает значение e.Fill.

VectorFill изменяет содержимое вектора, на который указы-
вает s.Vector. При этом длина вектора не меняется, а каждая
компонента вектора принимает значение e.Fill.

VectorLength выдает длину вектора, на который указывает
s.Vector.

VectorRef извлекает из вектора, на который указывает
s.Vector, содержимое его компоненты с номером s.Index, кото-
рое и выдается в качестве результата.

VectorSet изменяет содержимое вектора, на который указы-
вает s.Vector. При этом длина вектора не меняется, а компонента
вектора с номером s.Index принимает значение e.Exp.

VectorReplace изменяет содержимое вектора, на который ука-
зывает s.Vector. Старое содержимое уничтожается, а новое со-
держимое формируется из значения параметра e.Source точно
таким же способом, как и в случае функции Vector.

Subvector создает новый вектор и выдает ссылку на него в
качестве результата. При этом содержимое создаваемого вектора
формируется следующим образом. Рассматривается содержимое
исходного вектора, т. е. вектора, на который указывает s.Vector.
Пусть оно имеет вид (E0)(E1) ... (En). Тогда от него отбрасы-
ваются первые s.Index термов, а первые s.Len из оставшихся
термов считаются содержимым нового вектора.

Состояние исходного вектора при этом не меняется.
SubvectorFill изменяет содержимое вектора, на который ука-

зывает s.Vector. При этом длина вектора не меняется, а s.Len
последовательно расположенных компонент вектора, начиная с
компоненты с номером s.Index, принимают значение e.Exp.

Если размер исходного вектора недостаточен для выполнения
одной из вышеуказанных операций, в качестве результата выда-
ется $error(F "Index out of range"), где F — имя функции, а
состояние вектора не меняется.

Если при выполнении одной из вышеуказанных операций воз-
никает необходимость создать содержимое вектора, длина кото-
рого больше, чем допускается реализацией, в качестве результа-
та выдается $error(F "Size limit exceeded"), где F — имя
функции, а состояние вектора не меняется.

Глава 4

Принципы реализации
Рефала Плюс

Цель данной главы — показать, что для Рефала Плюс суще-
ствует достаточно простая и естественная реализация, которая
может рассматриваться как некоторое обобщение по отношению
к реализациям Рефала-2 [КлР 1987], [Ром 1987б].

В принципе, материал данной главы может рассматриваться
как еще одно описание семантики Рефала Плюс, основанное на
компиляции, в противоположность к описанию семантики, изло-
женному в главе 2, основанному на “естественной операционной
семантике”.

В настоящее время существует несколько практических реали-
заций Рефала Плюс, каждая из которых может рассматриваться,
как конкретизация или усовершенствование схемы реализации,
изложенной в данной главе [АОППС 2004], [АО 2004].

Обзор и сравнительной анализ этих реализаций не входят в
задачу данной книги.

4.1 Общая схема реализации

Программа, написанная на Рефале Плюс, компилируется в код
виртуальной машины. Этот код в дальнейшем может как интер-

183



претироваться, так и компилироваться.
В дальнейшем мы описываем семантику виртуального кода в

терминах абстрактной виртуальной машины. Затем мы описыва-
ем алгоритм компиляции программ на Рефале Плюс в виртуаль-
ный код.

Способы отображения абстрактной виртуальной машины на
традиционную фон-Неймановскую машину в данной книге не
рассматриваются. В основном, можно использовать те же мето-
ды, которые применялись при реализации Рефала-2 [КлР 1987],
[Ром 1987а], хотя у Рефала Плюс есть и некоторые особенности.
Например, для Рефала Плюс не очень подходит тот способ пред-
ставления объектных выражений, который применялся в реали-
зациях Рефала-2 [АбР 1988].

Не обсуждаются и некоторые дополнительные оптимизации,
которые могут выполняться при компиляции с Рефала Плюс в
виртуальный код, например, оптимизации, связанные с сокраще-
нием перебора при сопоставлении с образцом [Ром 1987а].

4.2 Виртуальный код

Программа, написанная в виртуальном коде, представляет со-
бой последовательность команд, каждая из которых содержит код
операции и, может быть, дополнительные операнды.

При записи программ в виртуальном коде мы будем применять
следующие обозначения.

Пустую последовательность команд будем обозначать через [ ].
Непустую последовательность команд, начинающуюся с команды
X, за которой следует последовательность команд C, будем обо-
значать через X : C. Таким образом, последовательность команд
X1,X2, . . . ,Xn записывается в видеX1 : X2 : . . . : Xn : []. Такая за-
пись удобна при описании команд виртуальной машины. Однако,
когда мы будем приводить примеры программ в виртуальном коде
и описывать алгоритм компиляции в виртуальный код, мы будем
пользоваться также и альтернативной записью X1;X2; . . . ;Xn.

4.2.1 Состояние виртуальной машины

Состояние виртуальной машины представляет собой упорядо-
ченную четверку

〈C, S, D, E〉
где C — список исполняемых команд (Code), S — стек данных
(Stack), D — свалка (Dump), E — свалка ошибок (Error dump).

Стек данных S используется для хранения объектных выра-
жений, которые получаются в результате вычислений. Через этот
стек передаются значения параметров функций и возвращаются
результаты функций.

Свалка D представляет собой стек, содержащий упорядочен-
ные пары вида (L, S), где L — последовательность команд, а S —
стек данных. Свалка используется для сохранения текущего со-
стояния стека данных (с целью его дальнейшего восстановления),
и для запоминания последовательностей команд, которые (может
быть) потребуется выполнять в дальнейшем.

Свалка ошибок E представляет собой стек, содержащий упо-
рядоченные тройки вида (L, S,D), где L — последовательность
команд, S — стек данных, а D — свалка. Свалка ошибок предна-
значена для запоминания последовательности команд, текущего
состояния стека данных и текущего состояния свалки. Эта инфор-
мация используется в случае возникновения ошибок для восста-
новления состояния виртуальной машины.

Работа виртуальной машины описывается с помощью правил
перехода из одного состояния в другое, которые имеют вид

〈C, S, D, E〉 ⇒
〈
C ′, S′, D′, E′

〉
и означают, что если виртуальная машина видит последователь-
ность команд C, стек данных S, свалку D и свалку ошибок E,
то она переходит в новое состояние, которое содержит последова-
тельность команд C ′, стек данных S′, свалку D′ и свалку ошибок
E′.

Если переход возможен только при выполнении дополнитель-
ного условия P , правило перехода записывается в виде

〈C, S, D, E〉 ⇒ {P}
〈
C ′, S′, D′, E′

〉

4.2.2 Некоторые обозначения

При описании команд виртуальной машины мы будем исполь-
зовать следующие обозначения.

Пустое объектное выражение будет обозначаться через �.
Пустой стек будет обозначаться через [ ], а непустой стек — че-

рез X : S, где X — вершина стека, а S — стек, который получится
из X : S после выталкивания из него верхнего элемента X.

Стек X1 : . . . : Xn : [ ], содержащий элементы X1, . . . , Xn, будет
обозначаться также и через [X1 : . . . : Xn].

Если S — стек, то его k-й элемент, считая от вершины, обо-
значается через S/k. А именно, если S = X0 : . . . : Xk : S′, то
S/k = Xk.

4.2.3 Метки и адреса

Программы в виртуальном коде могут содержать метки, на
которые может передаваться управление. Если требуется поме-
тить меткой L последовательность команд C, мы записываем это
следующим образом: LABEL L;C. После этого можно передавать
управление на эту последовательность, например с помощью ко-
манды перехода JUMP L.

Если требуется пометить начало скомпилированного опреде-
ления функции, мы будем писать FUNSTART L вместо LABEL L.
Это может оказаться полезным для тех реализаций виртуальной
машины, в которых существует различие между “длинными” и
“короткими” переходами.

Смысл команды JUMP можно было бы определить с помощью
следующего правила перехода:

〈JUMP L : . . . : LABEL L : C, S, D, E〉 ⇒ 〈C, S, D, E〉

которое можно истолковать следующим образом: “найди в после-
довательности команд метку L и выполняй команды, которые
идут вслед за ней”.

Мы знаем, однако, что при исполнении программ на обычных
компьютерах никаких поисков меток не выполняется. Вместо это-
го, команда перехода содержит некоторый “адрес”, который пря-

мо указывает, где находится следующая команда, подлежащая
выполнению. Поэтому LABEL L — это “не настоящая” команда,
а скорее директива для загрузчика, которая указывает, что во
всех командах вместо метки L следует подставить адрес коман-
ды, следующей за LABEL L, а саму директиву LABEL L — выкинуть
из программы.

Таким образом, в реальном компьютере все фрагменты кода
представляются их адресами, а метки в программе обозначают
некоторые адреса. Перед выполнением команды ее адрес всякий
раз преобразуется в саму команду. Однако такая точка зрения
неудобна при описании виртуальных команд, поэтому, при записи
правил перехода мы будем считать, что все фрагменты кода пред-
ставлены самими этими фрагментами, т. е. что все адреса как бы
заранее преобразованы в соответствующие им фрагменты. С этой
точки зрения, например, команда JUMP L содержит не адрес L, а
непосредственно последовательность команд, на которую следу-
ет передать управление. При этом команда JUMP описывается с
помощью следующего правила перехода:

〈JUMP L : C, S, D, E〉 ⇒ 〈L, S, D, E〉

Мы видим, что принятое нами соглашение позволяет сделать
описание команд более кратким и наглядным.

В дальнейшем фрагменты виртуального кода обозначаются
через C или L.

4.2.4 Пустая команда

Команда NOP не выполняет никаких действий:

〈NOP : C, S, D, E〉 ⇒ 〈C, S, D, E〉

4.2.5 Команды управления неуспехами

Команда ALT L запоминает в свалке последовательность ко-
манд L и текущее состояние стека данных S. Если в дальнейшем
возникает неуспех, состояние стека данных восстанавливается и

начинает выполняться список команд L. Таким образом, ALT L
создает развилку для перехвата неуспехов.

Команда CUT n выбрасывает со свалки n последних развилок,
отменяя n перехватов неуспехов.

Команда FAIL всегда порождает неуспех: извлекает из свалки
(L′, S′), приводит стек данных в состояние S′ и начинает выпол-
нять последовательность команд L′.

〈ALT L : C, S, D, E〉 ⇒ 〈C, S, (L, S) : D, E〉
〈CUT n : C, S, (L1, S1) : . . . : (Ln, Sn) : D, E〉

⇒ 〈C, S, D, E〉
〈FAIL : C, S, (L′, S′) : D, E〉 ⇒ 〈L′, S′, D, E〉

4.2.6 Команды переходов и вызовов функций

Команда JUMP L передает управление на последовательность
команд L.

Команда CALL a L вызывает функцию, имеющую a парамет-
ров. Значения этих параметров должны находиться в a верхних
элементах стека данных. Последовательность команд, идущих за
командой CALL , запоминается в свалке вместе с состоянием стека
данных. Они используются в дальнейшем для возврата из функ-
ции.

Команда RETURN осуществляет успешный возврат из функции,
а команда RETURNFAIL осуществляет возврат из функции с выра-
боткой неуспеха.

Команда TAILCALL L осуществляет “хвостовой” вызов функ-
ции. Она может использоваться в тех случаях, когда сразу после
команды CALL следует команда RETURN.

〈JUMP L : C, S, D, E〉 ⇒ 〈L, S, D, E〉
〈CALL a L : C, Ea : . . . : E1 : S, D, E〉 ⇒
〈L, Ea : . . . : E1 : [ ], (C, S) : D, E〉

〈RETURN : C, Eb : . . . : E1 : [ ], (C ′, S′) : D, E〉 ⇒
〈C ′, Eb : . . . : E1 : S′, D, E〉

〈RETURNFAIL : C, S, (C ′, S′) : D, E〉 ⇒ 〈FAIL : [ ], S′, D, E〉
〈TAILCALL L : C, S, D, E〉 ⇒ 〈L, S, D, E〉

Примечание. На первый взгляд TAILCALL L ' JUMP L, и
RETURNFAIL ' CUT 1; FAIL. Но это справедливо только если адреса
переходов между функциями и внутри одной функции представ-
лены одинаково. Во многих реализациях виртуального кода это
не так. Поэтому лучше предусмотреть в виртуальном коде раз-
личные команды.

4.2.7 Команды обработки ошибок

Команда PUSHTRAP L запоминает текущее состояние в свалке
ошибок. Команда POPTRAP выбрасывает один элемент из свалки
ошибок. Команда ERROR извлекает состояние из свалки ошибок
и переводит машину в это состояние. При этом верхний элемент
стека данных переносится на вершину стека данных в новом со-
стоянии. Таким способом передается сообщение об ошибке.

〈PUSHTRAP L : C, S, D, E〉 ⇒ 〈C, S, D, (L, S,D) : E〉
〈POPTRAP : C, S, D, (L′, S′, D′) : E〉 ⇒ 〈C, S, D, E〉
〈ERROR : C, E : S, D, (L′, S′, D′) : E〉 ⇒ 〈L′, E : S′, D′, E〉

4.2.8 Команды построения объектных выражений

Команда PUSHQ E вталкивает в стек данных объектное выра-
жение E .

Команда PUSH m находит в стеке данных объектное выраже-
ние, находящееся на расстоянии m от вершины стека данных, и
вталкивает его в стек данных.

Команда BR надевает скобки на объектное выражение, нахо-
дящееся на вершине стека данных.

Команда CONC конкатенирует (приписывает друг к другу) два
объектных выражения, находящихся в двух верхних позициях
стека данных.

〈PUSHQ E : C, S, D, E〉 ⇒ 〈C, E : S, D, E〉
〈PUSH m : C, S, D, E〉 ⇒ {S/m = E} 〈C, E : S, D, E〉

〈BR : C, E : S, D, E〉 ⇒ 〈C, (E) : S, D, E〉
〈CONC : C, E ′′ : E ′ : S, D, E〉 ⇒ 〈C, E ′ E ′′ : S, D, E〉

4.2.9 Команды очистки стека

Команда SQUEEZE a n сдвигает a верхних элементов стека дан-
ных на n позиций вглубь стека. При этом из стека выкидываются
n элементов.

Команда SLIDE n сдвигает верхний элемент стека данных на
n позиций вглубь стека. При этом из стека выкидываются n эле-
ментов.

Команда POP n выкидывает из стека данных n верхних эле-
ментов.

〈SQUEEZE a n : C, E1 : . . . : Ea : E ′1 : . . . : E ′n : S, D, E〉 ⇒
〈C, E1 : . . . : Ea : S, D, E〉

〈SLIDE n : C, E : E ′1 : . . . : E ′n : S, D, E〉 ⇒ 〈C, E : S, D, E〉
〈POP n : C, E ′1 : . . . : E ′n : S, D, E〉 ⇒ 〈C, S, D, E〉

4.2.10 Команды анализа объектных выражений

В этом разделе, для краткости, при записи условий в правилах
перехода, используются следующие соглашения.

Условие вида X = Y , где Y содержит какие-то величины x1,
. . . , xk, которые не определены в левой части правила перехода,
означает, что “существуют и могут быть однозначно определены
такие x1, . . . , xk, что выполнено X = Y ”. При этом x1, . . . , xk

могут употребляться в правой части правила перехода.
Условие вида X 6= Y , где Y содержит какие-то величины x1,

. . . , xk, которые не определены в левой части правила перехода,
означает, что “не существует таких x1 , . . . , xk, что выполнено
X = Y ”.

Например, пусть E ′ и E ′′ не встречаются в левой части правила.
Тогда условие S/m = (E ′) E ′′ означает, что можно найти (и при
том однозначно) такие объектные выражения E ′ и E ′′, что m-й
элемент стека S равен объектному выражению (E ′) E ′′. В то же
время, условие S/m 6= (E ′) E ′′ означает, что не существует таких
объектных выражений E ′ и E ′′, для которых m-й элемент стека S
равен объектному выражению (E ′) E ′′.

Описанные ниже команды выполняют анализ объектных вы-
ражений и разложение их на составляющие подвыражения. При

этом, если выражение имеет требуемую структуру, его подком-
поненты заносятся в стек данных. Если же выражение не имеет
требуемой структуры, возникает неуспех, и выполняются те же
действия, что и в случае команды FAIL. Первый операнд всех
этих команд показывает расположение в стеке данных того вы-
ражения, которое требуется анализировать.

Например, команда LS m работает следующим образом. Рас-
сматривается объектное выражение, которое находится в стеке
данных на расстоянии m от вершины стека. Пусть это выраже-
ние равно E . Тогда, если E пусто, возникает неуспех. Если же E не
пусто и имеет вид (E ′) E ′′, то также возникает неуспех. И наконец,
если E имеет вид S E ′, то стек данных S переходит в состояние
E ′ : S : S, и работа команды успешно завершается.

Команды анализа выражений распадаются на следующие
группы: проверки условий, жесткое отщепление слева, жесткое
отщепление справа, перебор слева и перебор справа.

4.2.10.1 Проверки условий

Команды из этой группы проверяют, что выражение удовле-
творяет определенным требованиям. При этом состояние стека
данных не изменяется (исключением является только команда
CB).

〈EMPTY m : C, S, D, E〉 ⇒ {S/m = �} 〈C, S, D, E〉
〈EMPTY m : C, S, D, E〉 ⇒ {S/m 6= �} 〈FAIL : [ ], S, D, E〉

〈CV m : C, S, D, E〉 ⇒ {S/m 6= �} 〈C, S, D, E〉
〈CV m : C, S, D, E〉 ⇒ {S/m = �} 〈FAIL : [ ], S, D, E〉

〈CB m : C, S, D, E〉 ⇒ {S/m = (E)} 〈C, E : S, D, E〉
〈CB m : C, S, D, E〉 ⇒ {S/m 6= (E)} 〈FAIL : [ ], S, D, E〉

〈CQ m E : C, S, D, E〉 ⇒ {S/m = E} 〈C, S, D, E〉
〈CQ m E : C, S, D, E〉 ⇒ {S/m 6= E} 〈FAIL : [ ], S, D, E〉

〈CS m : C, S, D, E〉 ⇒ {S/m = S} 〈C, S, D, E〉
〈CS m : C, S, D, E〉 ⇒ {S/m 6= S} 〈FAIL : [ ], S, D, E〉

〈CT m : C, S, D, E〉 ⇒ {S/m = T } 〈C, S, D, E〉
〈CT m : C, S, D, E〉 ⇒ {S/m 6= T } 〈FAIL : [ ], S, D, E〉

〈CEQ m m′ : C, S, D, E〉 ⇒ {S/m = S/m′} 〈C, S, D, E〉
〈CEQ m m′ : C, S, D, E〉 ⇒
{S/m 6= S/m′} 〈FAIL : [ ], S, D, E〉

4.2.10.2 Жесткое отщепление слева

Команды из этой группы отщепляют часть выражения с ле-
вого конца. Причем, если отщепление возможно, оно делается од-
нозначно. Исходное выражение при этом разбивается на части,
которые заносятся в стек данных.

〈LB m : C, S, D, E〉 ⇒
{S/m = (E ′) E ′′} 〈C, E ′ : (E ′) : E ′′ : S, D, E〉

〈LB m : C, S, D, E〉 ⇒
{S/m 6= (E ′) E ′′} 〈FAIL : [ ], S, D, E〉

〈LQ m E : C, S, D, E〉 ⇒ {S/m = E E ′} 〈C, E ′ : E : S, D, E〉
〈LQ m E : C, S, D, E〉 ⇒ {S/m 6= E E ′} 〈FAIL : [ ], S, D, E〉

〈LS m : C, S, D, E〉 ⇒ {S/m = S E} 〈C, E : S : S, D, E〉
〈LS m : C, S, D, E〉 ⇒ {S/m 6= S E} 〈FAIL : [ ], S, D, E〉

〈LT m : C, S, D, E〉 ⇒ {S/m = T E} 〈C, E : T : S, D, E〉
〈LT m : C, S, D, E〉 ⇒ {S/m 6= T E} 〈FAIL : [ ], S, D, E〉

〈LEQ m m′ : C, S, D, E〉 ⇒
{S/m = E ′ E , где E ′ = S/m′} 〈C, E : E ′ : S, D, E〉

〈LEQ m m′ : C, S, D, E〉 ⇒
{S/m 6= E ′ E , где E ′ = S/m′} 〈FAIL : [ ], S, D, E〉

4.2.10.3 Жесткое отщепление справа

Команды из этой группы отщепляют часть выражения с пра-
вого конца. Причем, если отщепление возможно, оно делается од-

нозначно. Исходное выражение при этом разбивается на части,
которые заносятся в стек данных.

〈RB m : C, S, D, E〉 ⇒
{S/m = E ′′ (E ′)} 〈C, E ′ : (E ′) : E ′′ : S, D, E〉

〈RB m : C, S, D, E〉 ⇒
{S/m 6= E ′′ (E ′)} 〈FAIL : [ ], S, D, E〉

〈RQ m E : C, S, D, E〉 ⇒ {S/m = E ′ E} 〈C, E ′ : E : S, D, E〉
〈RQ m E : C, S, D, E〉 ⇒ {S/m 6= E ′ E} 〈FAIL : [ ], S, D, E〉

〈RS m : C, S, D, E〉 ⇒ {S/m = E S} 〈C, E : S : S, D, E〉
〈RS m : C, S, D, E〉 ⇒ {S/m 6= E S} 〈FAIL : [ ], S, D, E〉

〈RT m : C, S, D, E〉 ⇒ {S/m = E T } 〈C, E : T : S, D, E〉
〈RT m : C, S, D, E〉 ⇒ {S/m 6= E T } 〈FAIL : [ ], S, D, E〉

〈REQ m m′ : C, S, D, E〉 ⇒
{S/m = E E ′, где E ′ = S/m′} 〈C, E : E ′ : S, D, E〉

〈REQ m m′ : C, S, D, E〉 ⇒
{S/m 6= E E ′, где E ′ = S/m′} 〈FAIL : [ ], S, D, E〉

4.2.10.4 Перебор слева

Команды из этой группы отщепляют часть выражения с ле-
вого конца и устанавливают развилку. Исходное выражение при
этом разбивается на части, которые заносятся в стек данных. Если
при выполнении последующих команд возникает неуспех, он пе-
рехватывается и делается попытка удлинить отщепленную часть
выражения.

Команде LE в программе обязательно должна предшествовать
команда LEI m или LVI m.

〈LEI m : LE : C, S, D, E〉 ⇒
{S/m = E и E : � : S = S′} 〈C, S′, (LE : C, S′) : D, E〉

〈LVI m : LE : C, S, D, E〉 ⇒
{S/m = E} 〈LE : C, E : � : S, D, E〉

193



〈LE : C, T E ′′ : E ′ : S, D, E〉 ⇒
{E ′′ : E ′ T : S = S′} 〈C, S′, (LE : C, S′) : D, E〉

〈LE : C, � : E : S, D, E〉 ⇒ 〈FAIL : [ ], S, D, E〉

4.2.10.5 Перебор справа

Команды из этой группы отщепляют часть выражения с пра-
вого конца и устанавливают развилку. Исходное выражение при
этом разбивается на части, которые заносятся в стек данных. Если
при выполнении последующих команд возникает неуспех, он пе-
рехватывается и делается попытка удлинить отщепленную часть
выражения.

Команде RE в программе обязательно должна предшествовать
команда REI m или RVI m.

〈REI m : RE : C, S, D, E〉 ⇒
{S/m = E и E : � : S = S′} 〈C, S′, (RE : C, S′) : D, E〉

〈RVI m : RE : C, S, D, E〉 ⇒
{S/m = E} 〈RE : C, E : � : S, D, E〉

〈RE : C, E ′′ T : E ′ : S, D, E〉 ⇒
{E ′′ : T E ′ : S = S′} 〈C, S′, (RE : C, S′) : D, E〉

〈RE : C, � : E : S, D, E〉 〈FAIL : [ ], S, D, E〉

4.3 Компиляция Рефал-программ
в виртуальный код

При описании алгоритма компиляции Рефал-программ в вир-
туальный код будем использовать обозначения, близкие к тем,
что обычно используются в книгах по денотационной семантике
[Шмд 1986].

4.3.1 Абстрактный синтаксис

FD ∈ определения функций
Palt ∈ образцовые распутья
Snt ∈ предложения
Q ∈ тропы
R ∈ хвосты
S ∈ источники
P ∈ образцы
Pe ∈ образцовые выражения
Pt ∈ образцовые термы
He ∈ жесткие выражения
Ht ∈ жесткие термы
F , Fe ∈ форматные выражения (форматы)
Ft ∈ форматные термы
Re ∈ результатные выражения
Rt ∈ результатные термы
V ∈ переменные
V s ∈ s-переменные
V t ∈ t-переменные
V v ∈ v-переменные
V e ∈ e-переменные

E ∈ объектные выражения
T ∈ объектные термы
S ∈ объектные символы
F ∈ имена функций

FD ::= F Palt;
Palt ::= \{Snt1; . . . Sntn;} | {Snt1; . . . Sntn;}
Snt ::= P R
Q ::= S R | S :: He R | S′′ $iter S′ :: He R |

S : Snt | R | S
R ::= , Q | # S R | \? Q | \! Q | $fail | = Q |

$error Q | $trap Q $with Palt
S ::= \{Q1; . . . Qn;} | {Q1; . . . Qn;} |

S : Palt | Re
P ::= $l Pe | $r Pe
Pe ::= Pt1 . . . P tn
Pt ::= S | (Pe) | V
He ::= Hts | Hts′ V v Hts′′ | Hts′ V e Hts′′
Hts ::= Ht1 . . . Htn
Ht ::= S | (He) | V s | V t
Fe ::= Fts | Fts′ v Fts′′ | Fts′ e Fts′′
Fts ::= Ft1 . . . F tn
Ft ::= S | (Fe) | s | t
Re ::= Rt1 . . . Rtn
Rt ::= S | (Re) | V | <F Re>

4.3.2 Функции, связанные с форматами

A[[He]] — список переменных, входящих в He (с сохранением
их взаимного порядка.

A[[ ]] = [ ]
A[[SHe]] = A[[He]]
A[[V He]] = [V ] ∧ A[[He]]
A[[(He′)He]] = A[[He′]] ∧ A[[He]]

где ∧ обозначает конкатенацию списков.

196



F [[He]] — формат жесткого выражения He, который получа-
ется из He отбрасыванием индексов всех переменных.

F [[ ]] =
F [[SHe]] = S F [[He]]
F [[V sHe]] = s F [[He]]
F [[V tHe]] = t F [[He]]
F [[V vHe]] = v F [[He]]
F [[V eHe]] = e F [[He]]
F [[(He′)He]] = ( F [[He′]] ) F [[He]]

4.3.3 Компиляция результатных выражений

Предположим, что нам требуется скомпилировать результат-
ное выражение Re. В общем случае это выражение может содер-
жать символы, скобки, переменные и вызовы функций. Кроме то-
го, результат вычисления этого выражения должен удовлетворять
некоторому формату Fout.

Мы начнем с рассмотрения более простой задачи. Будем пред-
полагать, чтоRe не содержит вызовы функций и что его требуется
вычислить следующим образом: вместо всех переменных подста-
вить их значения, а получившееся объектное выражение поме-
стить на вершину стека. При этом мы считаем, что значения всех
переменных, входящих в Re, уже известны и находятся на стеке.

Таким образом, для компиляции Re нам требуется знать сле-
дующую дополнительную информацию: среду ρ, связывающую
переменные с их позициями в стеке и число d — номер последней
занятой позиции в стеке (если стек пуст, то d = −1). При этом
мы считаем, что позиции в стеке нумеруются от дна к вершине
начиная с нуля. Следовательно, если стек содержит n объектных
выражений и имеет вид

En−1 : . . . : E1 : E0 : [ ]

то содержимым позиции i является объектное выражение Ei.
Таким образом, ρ(V ) является позицией в стеке, в которой на-

ходится значение переменной V , а (d − ρ(V )) — расстояние от
вершины стека до этой позиции.

Если заданы Re, ρ и d, то результат компиляции результатного
выражения Re мы будем обозначать через Re1[[Re]] ρ d. Компили-
рующая функция Re1 определяются следующими соотношения-
ми:

Re1[[ ]] ρ d = PUSHQ �
Re1[[Re Rt]] ρ d = Re1[[Re]] ρ d; Rt1[[Rt]] ρ (d+ 1); CONC

Rt1[[S]] ρ d = PUSHQ S
Rt1[[V ]] ρ d = PUSH (d− ρ(V ))
Rt1[[(Re)]] ρ d = Re1[[Re]] ρ d; BR

Конечно, это не самый оптимальный способ компиляции, и
позднее мы обсудим возможные усовершенствования.

Теперь рассмотрим общий случай результатного выражения
Re, содержащего вызовы функций. Будем считать, что задан фор-
мат Fout, которому должен удовлетворять результат вычисления
этого выражения, и число r — позиция в стеке, куда следует по-
местить результат вычисления Re. При этом, если Fout содержит
b переменных, результатом вычисления Re должны быть b объ-
ектных выражений, соответствующих переменным формата. Эти
выражения должны быть размещены в стеке в позициях r, r + 1,
. . . , r + b− 1.

Результат компиляции результатного выражения Re может
быть описан с помощью функции Re, обращение к которой имеет
следующий вид:

Re[[Re]] [[Fout]] r ρ d

где ρ и d имеют тот же смысл, как и в случае функции Re1.
Компиляция Re состоит из следующих этапов.
Первым делом из Re извлекаются все вызовы функций и ге-

нерируется код, обеспечивающий вычисление этих вызовов. Сами
вызовы при этом удаляются из Re и заменяются на выходные
форматы этих функций. При этом переменные, входящие в фор-
маты, снабжаются индексами, в результате чего они превращают-
ся в жесткие выражения. Индексы переменных выбираются про-
извольно, но так, чтобы не возникло конфликтов с переменными

из dom ρ. (Эти новые переменные получают значения в результате
вычисления вызовов функций в Re.) Затем информация о новых
переменных заносится в среду ρ.

Вся эта деятельность осуществляется компилирующей функ-
цией

ExtractCallsRe

обращение к которой имеет вид ExtractCallsRe[[Re]] ρ d, а ре-
зультат работы имеет вид (Ccalls, Re′, ρ′, d′), где Ccalls — вирту-
альный код, производящий вычисление вызовов функций из Re,
Re′ — результатное выражение, которое получается из Re в ре-
зультате удаления вызовов функций, ρ′ — среда ρ, пополненная
сведениями о новых переменных и d′ — значение d, исправленное
с учетом того, что в результате исполнения кода Ccalls, в стеке
могут оказаться занятыми дополнительные позиции.

Интересно отметить, что функция ExtractCallsRe рекурсив-
но обращается к функции Re, поскольку вычисление вызовов
функций требует вычисления их аргументных выражений.

После извлечения вызовов функций из Re, получается резуль-
татное выражение Re′, которое является частным случаем фор-
мата Fout и может быть разбито на b результатных выражений,
каждое из которых дает значение для одной из переменных фор-
мата Fout. Это делается с помощью функции SplitRe, обращение
к которой имеет вид SplitRe[[Fout]] [[Re′]], а результатом является
список из b результатных выражений, каждое из которых теперь
может быть скомпилировано функцией Re1.

И наконец, остается сгенерировать команду SQUEEZE, сдвигаю-

щую b результатов в позицию r.

Re[[Re]] [[F ]] r ρ d =
let val (Ccalls, Re′, ρ′, d′) = ExtractCallsRe[[Re]] ρ d

val [Re1, . . . , Reb] = SplitRe[[F ]] [[Re′]]
in
Ccalls;
Re1[[Re1]] ρ′ d′;
. . .
Re1[[Reb]] ρ′ (d′ + b− 1);
SQUEEZE b (d′ + 1− r)

end

ExtractCallsRe[[ ]] ρ d = (NOP,�, ρ, d)

ExtractCallsRe[[RtRe]] ρ d =
let val (C ′, Re′, ρ′, d′) = ExtractCallsRt[[Rt]] ρ d

val (C ′′, Re′′, ρ′′, d′′) = ExtractCallsRe[[Re]] ρ′ d′

in ((C ′;C ′′), Re′Re′′, ρ′′, d′′) end

ExtractCallsRt[[S]] ρ d = (NOP,S, ρ, d)

ExtractCallsRt[[V ]] ρ d = (NOP, V, ρ, d)

ExtractCallsRt[[(Re)]] ρ d =
let val (C ′, Re′, ρ′, d′) = ExtractCallsRe[[Re]] ρ d
in (C ′, (Re′), ρ′, d′) end

ExtractCallsRt[[<F Re>]] ρ d =
let val Fin = входной формат функции F

val Fout = выходной формат функции F
val He = такое жесткое выражение, что

Fout = F [[He]] и dom ρ ∩ V[[He]] = ∅
val [V ′1 , . . . , V

′
a] = A[[Fin]]

val [V1, . . . , Vb] = A[[He]]
in

(Re[[Re]] [[Fin]] (d+ 1) ρ d; CALL a F ,
He, ρ+ {V1 = d+ 1, . . . , Vb = d+ b}, d+ b)

end

SplitRe[[FtFe]] [[RtRe]] =
SplitRt[[Ft]] [[Rt]] ∧ SplitRe[[Fe]] [[Re]]

SplitRe[[FeFt]] [[ReRt]] =
SplitRe[[Fe]] [[Re]] ∧ SplitRt[[Ft]] [[Rt]]

SplitRe[[v]] [[Re]] = [Re]
SplitRe[[e]] [[Re]] = [Re]
SplitRe[[ ]] [[ ]] = [ ]

SplitRt[[S]] [[S]] = [ ]
SplitRt[[s]] [[Rt]] = [Rt]
SplitRt[[t]] [[Rt]] = [Rt]
SplitRt[[(Fe)]] [[(Re)]] = SplitRe[[Fe]] [[Re]]

4.3.4 Генерация ошибок

В процессе работы рефал-программы во многих случаях мо-
жет вырабатываться ошибка вида

$error(F "Unexpected fail")

где F — имя функции, в которой возникла ошибка. Поэтому ре-
зультат компиляции каждой функции F содержит фрагмент вир-
туального кода вида

LABEL Lerror;
PUSHQ F "Unexpected fail"; ERROR

где Lerror — некоторая метка (для каждой функции — своя). Та-
ким образом, чтобы сгенерировать ошибку, достаточно передать
управление на Lerror.

Метка Lerror используется многими компилирующими функ-
циями, и ее следовало бы (ради “идеологической чистоты”) пере-
давать этим функциям в качестве параметра. Тем не менее, чтобы
не загромождать изложение, мы этого делать не будем, а будем
считать, что Lerror — глобальная величина, которая не меняется
на протяжении компиляции одного определения функции.

4.3.5 Компиляция троп, хвостов и источников

Результат компиляции троп, хвостов и источников может быть
описан с помощью функций Q, R и S соответственно. Обращения
к этим функциям имеют следующий вид:

Q[[Q]] [[Fout]] r k ρ d
R[[R]] [[Fout]] r k ρ d
S[[S]] [[Fout]] r k ρ d

где Fout, r, ρ и d имеют тот же смысл, что и в случае функции Re,
а k представляет собой стек альтернатив, содержимое которого
всегда имеет следующий вид:

[h1 : . . . : hn]

где h1, . . . , hn — целые неотрицательные числа, и n ≥ 1. Стек k
служит для подсчета сгенерированных команд ALT и используется
при генерации команд CUT.

А именно, если стек альтернатив имеет вид h : k и порожда-
ется команда ALT, к его вершине прибавляется 1, и он принимает
вид (h+ 1) : k.

Появление и исчезновение элементов в стеке альтернатив свя-
зано с заборами и отсечениями. А именно, если при компиляции
тропы встретился забор \?, а стек альтернатив имеет вид k, то
в стеке заводится новый элемент, и он принимает вид 0 : k. Ес-
ли же при компиляции тропы встречается отсечение \!, а стек
имеет вид h : k, то генерируется команда CUT h, а стек принима-
ет вид k. Благодаря этому, все альтернативы, скопившиеся после
последнего забора \?, отменяются.

Стек альтернатив никогда не бывает пустым, поскольку в нем
ведется учет команд ALT, сгенерированных “на нулевом уровне”,
т. е. еще до того как встретился первый забор, или после “закры-
тия” всех заборов.

Если встречается правая часть, т. е. тропа вида = Q, требуется
отменить все альтернативы (даже те, что находятся на нулевом
уровне). Поэтому, если стек альтернатив имеет вид [h1, . . . , hn],

генерируется команда CUT (h1 + . . .+hn), а стек альтернатив при-
нимает вид [0]. Таким образом, на всей предыстории “ставится
крест” и начинается “новая жизнь”.

При компиляции троп, хвостов и источников используются
функции Snt и Palt, которые служат для компиляции предложе-
ний и образцовых распутий. Эти компилирующие функции опи-
саны в разделах, посвященных компиляции предложений и образ-
цовых распутий.

4.3.6 Компиляция троп

Q[[S R]] [[F ]] r k ρ d =
S[[S]] [[ ]] (d+ 1) [0] ρ d; R[[R]] [[F ]] r k ρ d

Q[[S :: HeR]] [[F ]] r k ρ d =
let val [V1, . . . , Va] = A[[He]]

val ρ′ = ρ+ {V1 = d+ 1, . . . , Va = d+ a}
in
S[[S]] (F [[He]]) (d+ 1) [0] ρ d;
R[[R]] [[F ]] r k ρ′ (d+ a)

end

Q[[S′′ $iter S′ :: HeR]] [[F ]] r (h : k) ρ d =
let val [V1, . . . , Va] = A[[He]]

val ρ′ = ρ+ {V1 = d+ 1, . . . , Va = d+ a}
in

S[[S′′]] (F [[He]]) (d+ 1) [0] ρ d; JUMP L2;
LABEL L1;
S[[S′]] (F [[He]]) (d+ 1) [0] ρ′ (d+ a);

LABEL L2;
ALT L1;
R[[R]] [[F ]] r ((h+ 1) : k) ρ′ (d+ a)

end

Q[[S : Snt]] [[F ]] r k ρ d =
S[[S]] [[e]] (d+ 1) [0] ρ d; Snt[[Snt]] [[e]] [[F ]] r k ρ (d+ 1)

Q[[R]] [[F ]] r k ρ d = R[[R]] [[F ]] r k ρ d

Q[[S]] [[F ]] r k ρ d = S[[S]] [[F ]] r k ρ d

4.3.7 Компиляция хвостов

R[[, Q]] [[F ]] r k ρ d = Q[[Q]] [[F ]] r k ρ d

R[[# S R]] [[F ]] r k ρ d =
ALT L1;
S[[S]] [[ ]] (d+ 1) [0] ρ d;
CUT 1; FAIL;

LABEL L1;
R[[R]] [[F ]] r k ρ d

R[[\? Q]] [[F ]] r k ρ d = Q[[Q]] [[F ]] r (0 : k) ρ d

R[[\! Q]] [[F ]] r (h : k) ρ d = CUT h; Q[[Q]] [[F ]] r k ρ d

R[[$fail]] [[F ]] r k ρ d = FAIL

R[[= Q]] [[F ]] r [h1, . . . , hn] ρ d =
CUT (h1 + . . .+ hn); Q[[Q]] [[F ]] r [0] ρ d

R[[$error Q]] [[F ]] r k ρ d =
ALT Lerror;
Q[[Q]] [[e]] (d+ 1) [0] ρ d;
ERROR

R[[$trap Q $with Palt]] [[F ]] r [h1, . . . , hn] ρ d =
PUSHTRAP L1; ALT Lerror;
Q[[Q]] [[F ]] r [0] ρ d;
CUT (1 + h1 + . . .+ hn); POPTRAP; JUMP L0;

LABEL L1;
Palt[[Palt]] [[e]] [[F ]] r [h1, . . . , hn] ρ (d+ 1);

LABEL L0

4.3.8 Компиляция источников

S[[\{Q1; . . . Qn; Q0;}]] [[F ]] r (h : k) ρ d =
ALT L1;
Q[[Q1]] [[F ]] r ((h+ 1) : k) ρ d; JUMP L0;

LABEL L1;
S[[\{ . . . Qn; Q0;}]] [[F ]] r (h : k) ρ d;

LABEL L0

S[[\{Q0; }]] [[F ]] r k ρ d = Q[[Q0]] [[F ]] r k ρ d

S[[\{}]] [[F ]] r k ρ d = FAIL

S[[{Q1; . . . Qn; Q0;}]] [[F ]] r (h : k) ρ d =
ALT L1;
Q[[Q1]] [[F ]] r ((h+ 1) : k) ρ d; JUMP L0;

LABEL L1;
S[[{ . . . Qn; Q0;}]] [[F ]] r (h : k) ρ d;

LABEL L0

S[[{Q0;}]] [[F ]] r (h : k) ρ d =
ALT Lerror;
Q[[Q0]] [[F ]] r ((h+ 1) : k) ρ d

S[[{}]] [[F ]] r k ρ d = JUMP Lerror

S[[S : Palt]] [[F ]] r k ρ d =
S[[S]] [[e]] (d+ 1) [0] ρ d;
Palt[[Palt]] [[e]] [[F ]] r k ρ (d+ 1)

S[[Re]] [[F ]] r [h1, . . . , hn] ρ d =
Re[[Re]] [[F ]] r ρ d; CUT (h1 + . . .+ hn)

4.3.9 Компиляция предложений

Результат компиляции предложений может быть описан с по-
мощью функции Snt, обращение к которой имеет следующий вид:

Snt[[Snt]] [[Fin]] [[Fout]] r k ρ d

где Fout, r, k, ρ и d имеют тот же смысл, что и в случае функций
Q, R и S, а Fin — входной формат предложения.

Входной формат Fin описывает структуру объектных выра-
жений, которые должны сопоставляться со входным образцом
предложения. Переменные, входящие в Fin, соответствуют тем
компонентам объектных выражений, которые могут меняться. Ес-
ли Fin содержит a переменных, а d — номер последней занятой
позиции в стеке, то перед началом исполнения предложения ком-
поненты объектного выражения должны быть помещены в стек в
позиции d− a+ 1, . . . , d.

Компиляция предложения Snt, имеющего вид $l Pe R, про-
исходит следующим образом.

Первым делом образцовое выражение Pe разбивается на a об-
разцовых выражений Pe1, . . . , Pea, каждое из которых соответ-
ствует одной из переменных, входящих в Fin. А именно, образ-
цовому выражению Pej соответствует переменная формата Vj .
При этом, компонента объектного выражения, соответствующая
Vj , должна находиться в стеке в позиции d−a+j. Поэтому созда-
ется список упорядоченных троек (d−a+j, Vj , P ej). Для большей
легкости зрительного восприятия мы будем записывать этот спи-
сок следующим образом:

[d− a+ 1 : V1 : Pe1] . . . [d : Va : Pea]

Соответственно, тройка (d−a+j, Vj , P ej) записывается в виде [d−
a+ j : Vj : Pej ], а список этих троек записывается без наружных
квадратных скобок и без разделяющих запятых.

Этот список затем обрабатывается функцией PeInit, целью
которой является преобразовать тройки вида [i : V : Pe] в пары
вида [i : Pe], где каждая пара [i : Pe] означает, что образец Pe
должен сопоставляться с объектным выражением, хранящимся в
стеке в позиции i.

Функция PeInit анализирует каждую тройку [i : V : Pe]. Если
оказывается, что Pe — переменная, значение которой еще неиз-
вестно (т. е. Pe не входит в область определения среды ρ), и при
этом Pe имеет тот же тип, что и V — это означает, что Pe заведомо
успешно сопоставляется с объектным выражением, хранящимся в

позиции i. Поэтому тройка [i : V : Pe] удаляется из списка, а среда
ρ преобразуется в среду ρ+{Pe = i}. Если же одно из вышеуказан-
ных условий нарушено, это означает, что для сопоставления Pe с
объектным выражением заведомо требуется выполнять какие-то
действия, поэтому тройка [i : V : Pe] преобразуется в пару [i : Pe].

Таким образом, результат работы функции PeInit имеет вид

[i1 : Pe1] . . . [im : Pem]

и передается для дальнейшей обработки функции PeL, которая
осуществляет компиляцию образцов при условии, что отождеств-
ление должно выполняться слева направо.

Случай отождествления справа налево “зеркально симметри-
чен” случаю отождествления слева направо, поэтому мы его в
дальнейшем рассматривать не будем.

Snt[[$l Pe R]] [[Fin]] [[Fout]] r k ρ d =
let val [V1, . . . , Va] = A[[Fin]]

val [Pe1, . . . , P ea] = SplitPe[[Fin]] [[Pe]]
val (∆, ρ′) =

PeInit[[ ]] [[[d− a+ 1 : V1 : Pe1] . . . [d : Va : Pea]]] ρ
in PeL[[∆]] (R[[R]] [[Fout]] r) k ρ′ d end

Snt[[$r Pe R]] [[Fin]] [[Fout]] r k ρ d = . . .

PeInit[[∆]] [[ ]] ρ = (∆, ρ)

PeInit[[∆]] [[[i : V : Pe] Φ]] ρ =
if F [[Pe]] = V and Pe 6∈ dom ρ
then PeInit[[∆]] [[Φ]] (ρ+ {Pe = i})
else PeInit[[∆ [i : Pe]]] [[Φ]] ρ

SplitPe[[F ]] [[Pe]] — разбиение Pe на список образцов в соот-
ветствии с форматом F .

SplitPe[[ ]] [[ ]] = [ ]
SplitPe[[Ft Fe]] [[Pt Pe]] =

SplitPt[[Ft]] [[Pt]] ∧ SplitPe[[Fe]] [[Pe]]
SplitPe[[Fe Ft]] [[Pe Pt]] =

SplitPe[[Fe]] [[Pe]] ∧ SplitPt[[Ft]] [[Pt]]
SplitPe[[v]] [[Pe]] = [Pe]
SplitPe[[e]] [[Pe]] = [Pe]

SplitPt[[S]] [[S]] = [ ]
SplitPt[[s]] [[Pt]] = [Pt]
SplitPt[[t]] [[Pt]] = [Pt]
SplitPt[[(Fe)]] [[(Pe)]] = SplitPe[[Fe]] [[Pe]]

4.3.10 Компиляция образцов

Компиляция образцовых выражений (в случае отождествле-
ния слева направо) производится функцией PeL, обращение к
которой имеет следующий вид:

PeL[[∆]] c k ρ d

где k, ρ и d имеют тот же смысл, что и в случае функции Snt,
∆ имеет вид [i1 : Pe1] . . . [im : Pem], а c является “продолже-
нием”, т. е. функцией, которую надо вызвать после завершения
компиляции всех образцовых выражений из ∆, передав ей k, ρ и
d.

Определение функции PeL содержит правила вида

PeL[[∆]] c k ρ d || C = . . .

где C — некоторое условие. Такая запись означает, что данное
правило применимо только если выполнено дополнительное усло-
вие C.

Во многих случаях оказывается, что можно применить сра-
зу несколько правил из описания функции PeL. В таких случа-
ях делается выбор одного из правил (либо произвольно, либо из

каких-то дополнительных соображений). Конечно, во всяком ре-
альном компиляторе этот выбор делается однозначно, и может
диктоваться соображениями оптимизации.

Все правила распадаются на несколько групп.

4.3.11 Завершение компиляции образцовых
выражений

PeL[[ ]] c k ρ d = c k ρ d

4.3.12 “Закрытые” с двух сторон элементы образцов

PeL[[Γ [i :] ∆]] c k ρ d =
EMPTY (d− i); PeL[[Γ ∆]] c k ρ d

PeL[[Γ [i : (Pe)] ∆]] c k ρ d =
CB (d− i); PeL[[Γ [d+ 1 : Pe] ∆]] c k ρ (d+ 1)

PeL[[Γ [i : S] ∆]] c k ρ d =
CQ (d− i) S; PeL[[Γ ∆]] c k ρ d

PeL[[Γ [i : V ] ∆]] c k ρ d || V ∈ dom ρ =
CEQ (d− i) (d− ρ(V )); PeL[[Γ ∆]] c k ρ d

PeL[[Γ [i : V s] ∆]] c k ρ d || V s 6∈ dom ρ =
CS (d− i); PeL[[Γ ∆]] c k (ρ+ {V s = i}) d

PeL[[Γ [i : V t] ∆]] c k ρ d || V t 6∈ dom ρ =
CT (d− i); PeL[[Γ ∆]] c k (ρ+ {V t = i}) d

PeL[[Γ [i : V v] ∆]] c k ρ d || V v 6∈ dom ρ =
CV (d− i); PeL[[Γ ∆]] c k (ρ+ {V v = i}) d

PeL[[Γ [i : V e] ∆]] c k ρ d || V e 6∈ dom ρ =
PeL[[Γ ∆]] c k (ρ+ {V e = i}) d

4.3.12.1 “Жесткие” элементы слева

PeL[[Γ [i : (Pe′)Pe′′] ∆]] c k ρ d || Pe′′ 6= � =
LB (d− i); PeL[[Γ [d+ 3 : Pe′][d+ 1 : Pe′′] ∆]] c k ρ (d+ 3)

PeL[[Γ [i : S Pe] ∆]] c k ρ d || Pe 6= � =
LQ (d− i) S; PeL[[Γ [d+ 2 : Pe] ∆]] c k ρ (d+ 2)

PeL[[Γ [i : V Pe] ∆]] c k ρ d || V ∈ dom ρ, Pe 6= � =
LEQ (d− i) (d− ρ(V )); PeL[[Γ [d+ 2 : Pe] ∆]] c k ρ (d+ 2)

PeL[[Γ [i : V sPe] ∆]] c k ρ d || V s 6∈ dom ρ, Pe 6= � =
LS (d− i);
PeL[[Γ [d+ 2 : Pe] ∆]] c k (ρ+ {V s = d+ 1}) (d+ 2)

PeL[[Γ [i : V tPe] ∆]] c k ρ d || V t 6∈ dom ρ, Pe 6= � =
LT (d− i);
PeL[[Γ [d+ 2 : Pe] ∆]] c k (ρ+ {V t = d+ 1}) (d+ 2)

4.3.13 “Жесткие” элементы справа

PeL[[Γ [i : Pe′′ (Pe′)] ∆]] c k ρ d || Pe′′ 6= � =
RB (d− i); PeL[[Γ [d+ 1 : Pe′′] [d+ 3 : Pe′] ∆]] c k ρ (d+ 3)

PeL[[Γ [i : PeS] ∆]] c k ρ d || Pe 6= � =
RQ (d− i) S; PeL[[Γ [d+ 2 : Pe] ∆]] c k ρ (d+ 2)

PeL[[Γ [i : PeV ] ∆]] c k ρ d || V ∈ dom ρ, Pe 6= � =
REQ (d− i) (d− ρ(V )); PeL[[Γ [d+ 2 : Pe] ∆]] c k ρ (d+ 2)

PeL[[Γ [i : PeV s] ∆]] c k ρ d || V s 6∈ dom ρ, Pe 6= � =
RS (d− i);
PeL[[Γ [d+ 2 : Pe] ∆]] c k (ρ+ {V s = d+ 1}) (d+ 2)

PeL[[Γ [i : PeV t] ∆]] c k ρ d || V t 6∈ dom ρ, Pe 6= � =
RT (d− i);
PeL[[Γ [d+ 2 : Pe] ∆]] c k (ρ+ {V t = d+ 1}) (d+ 2)

4.3.13.1 “Открытые” ve-переменные слева

Следующие два правила компиляции имеет смысл применять
только в тех случаях, когда ни одно из предыдущих правил не
применимо. Требуется исправление стека альтернатив k, ибо ко-
манда LE неявно содержит в себе команду ALT.

PeL[[[i : V v Pe] ∆]] c (h : k) ρ d || V v 6∈ dom ρ, Pe 6= � =
LVI (d− i); LE;
PeL[[[d+ 2 : Pe] ∆]] c ((h+ 1) : k)

(ρ+ {V v = d+ 1}) (d+ 2)

PeL[[[i : V ePe] ∆]] c (h : k) ρ d || V e 6∈ dom ρ, Pe 6= � =
LEI (d− i); LE;
PeL[[[d+ 2 : Pe] ∆]] c ((h+ 1) : k)

(ρ+ {V e = d+ 1}) (d+ 2)

4.3.14 Компиляция образцовых распутий

Результат компиляции образцовых распутий может быть опи-
сан с помощью функции Palt, обращение к которой имеет следу-
ющий вид:

Palt[[Palt]] [[Fin]] [[Fout]] r k ρ d

где Fin, Fout, r, k, ρ и d имеют тот же смысл, что и в случае
функции Snt.

Palt[[\{Snt1; . . . Sntn; Snt0;}]] [[Fin]] [[Fout]]
r (h : k) ρ d =

ALT L1;
Snt[[Snt1]] [[Fin]] [[Fout]] r ((h+ 1) : k) ρ d; JUMP L0;

LABEL L1;
Palt[[\{ . . . Sntn; Snt0;}]] [[Fin]] [[Fout]] r (h : k) ρ d;

LABEL L0

Palt[[\{Snt0;}]] [[Fin]] [[Fout]] r (h : k) ρ d =
Snt[[Snt0]] [[Fin]] [[Fout]] r (h : k) ρ d

Palt[[\{}]] [[Fin]] [[Fout]] r (h : k) ρ d = FAIL

Palt[[{Snt1; . . . Sntn; Snt0;}]] [[Fin]] [[Fout]] r (h : k) ρ d =
ALT L1;
Snt[[Snt1]] [[Fin]] [[Fout]] r ((h+ 1) : k) ρ d; JUMP L0;

LABEL L1;
Palt[[{ . . . Sntn; Snt0;}]] [[Fin]] [[Fout]] r (h : k) ρ d;

LABEL L0

Palt[[{Snt0;}]] [[Fin]] [[Fout]] r (h : k) ρ d =
ALT Lerror;
Snt[[Snt0]] [[Fin]] [[Fout]] r ((h+ 1) : k) ρ d

Palt[[{}]] [[Fin]] [[Fout]] r (h : k) ρ d = JUMP Lerror

4.3.15 Компиляция определения функции

Результат компиляции определений функций может быть опи-
сан с помощью функции FD, обращение к которой имеет следу-
ющий вид:

FD[[FDec;FD]]

где FDec — объявление функции, а FD - ее определение.

FD[[$func? F Fin = Fout; F Palt;]] =
let val [V1, . . . , Va] = A[[Fin]] in
FUNSTART F ;
ALT Lfail;
Palt[[Palt]] [[Fin]] [[Fout]] 0 [0] {} (a− 1);
CUT 1; RETURN;

LABEL Lfail;
RETURNFAIL;

LABEL Lerror;
PUSHQ F "Unexpected fail"; ERROR

end

FD[[$func F Fin = Fout; F Palt;]] =
let val [V1, . . . , Va] = A[[Fin]] in
FUNSTART F ;
ALT Lerror;
Palt[[Palt]] [[Fin]] [[Fout]] 0 [0] {} (a− 1);
CUT 1; RETURN;

LABEL Lerror;
PUSHQ F "Unexpected fail"; ERROR

end

Алфавитный указатель
функций

$func Add s.Int1 s.Int2 = s.Int; Arithm
$func? Apply s.Name e.Exp = e.Exp; Apply
$func Arg s.Int = e.Arg; Dos
$func Bind s.Tab (e.Key) (e.Val) = ; Table
$func BitAnd s.Int1 s.Int2 = s.Int; Bit
$func BitClear s.Int s.Pos = s.Int; Bit
$func BitLeft s.Int s.Shift = s.Int; Bit
$func BitLength s.Int = s.Len; Bit
$func BitNot s.Int = s.Int; Bit
$func BitOr s.Int1 s.Int2 = s.Int; Bit
$func BitRight s.Int s.Shift = s.Int; Bit
$func BitSet s.Int s.Pos = s.Int; Bit
$func? BitTest s.Int s.Pos = ; Bit
$func BitXor s.Int1 s.Int2 = s.Int; Bit
$func Box e.Exp = s.Box; Box
$func BytesToChars e.Int = e.Char; Convert
$func Channel = s.Channel; StdIO
$func CharsToBytes e.Char = e.Int; Convert
$func CloseChannel s.Channel = ; StdIO
$func Compare (e.Exp1)(e.Exp2) = s.Res; Compare
$func Div s.Int1 s.Int2 = s.Quo; Arithm
$func DivRem s.Int1 s.Int2 = s.Quo s.Rem; Arithm
$func? Domain s.Tab = e.KeyList ; Table

$func? Eq (e.Exp1)(e.Exp2) = ; Compare
$func Exit s.ReturnCode = ; Dos
$func GCD s.Int1 s.Int2 = s.GCD; Arithm
$func? Ge (e.Exp1)(e.Exp2) = ; Compare
$func Get s.Box = e.Exp; Box
$func GetEnv e.VarName = e.Value; Dos
$func? Gt (e.Exp1)(e.Exp2) = ; Compare
$func? IsBox e.Exp = ; Class
$func? IsChannel e.Exp = ; Class
$func? IsChar e.Exp = ; Class
$func? IsDigit e.Exp = ; Class
$func? IsEof s.Channel = ; StdIO
$func? IsFunc e.Exp = ; Class
$func? IsInTable s.Tab e.Key = ; Table
$func? IsInt e.Exp = ; Class
$func? IsLetter e.Exp = ; Class
$func? IsString e.Exp = ; Class
$func? IsTable e.Exp = ; Class
$func? IsVector e.Exp = ; Class
$func? IsWord e.Exp = ; Class
$func? L s.Left e.Exp = t.SubTerm; Access
$func? Le (e.Exp1)(e.Exp2) = ; Compare
$func? Left Access

s.Left s.Len e.Exp = e.SubExp;
$func Length e.Exp = s.ExpLen; Access
$func? Lt (e.Exp1)(e.Exp2) = ; Compare
$func? Lookup s.Tab e.Key = e.Val; Table
$func? Middle Access

s.Left s.Right e.Exp = e.SubExp;
$func Mult s.Int1 s.Int2 = s.Int; Arithm
$func? Ne (e.Exp1)(e.Exp2) = ; Compare
$func? OpenFile s.Channel e.FileName s.Mode = ; StdIO
$func Print e.Expr = ; StdIO
$func PrintCh s.Channel e.Expr = ; StdIO
$func PrintLn e.Expr = ; StdIO
$func PrintLnCh s.Channel e.Expr = ; StdIO

219



$func? R s.Right e.Exp = t.SubTerm; Access
$func? Read = t.Term; StdIO
$func? ReadCh s.Channel = t.Term; StdIO
$func? ReadChar = s.Char; StdIO
$func? ReadCharCh s.Channel = s.Char; StdIO
$func? ReadLine = e.Char; StdIO
$func? ReadLineCh s.Channel = e.Char; StdIO
$func Rem s.Int1 s.Int2 = s.Rem; Arithm
$func ReplaceTable s.TargetTab s.SourceTab = ; Table
$func? Right s.Right s.Len e.Exp = e.SubExp; Access
$func Store s.Box e.Exp = ; Box
$func String e.Source = s.String; String
$func StringFill s.String s.Fill = ; String
$func StringInit s.String s.Len s.Fill = ; String
$func StringLength s.String = s.Len; String
$func StringRef s.String s.Index = s.Char; String
$func StringReplace s.String e.Source = ; String
$func StringSet s.String s.Index s.Char = ; String
$func Sub s.Int1 s.Int2 = s.Int; Arithm
$func Substring String

s.String s.Index s.Len = s.NewString;
$func SubstringFill String

s.String s.Index s.Len s.Fill =;
$func Subvector Vector

s.Vector s.Index s.Len = s.Vector;
$func SubvectorFill Vector

s.Vector s.Index s.Len e.Fill =;
$func Table = s.Tab; Table
$func TableCopy s.Tab = s.TabCopy ; Table
$func Time = e.String; Dos
$func ToChars e.Exp = e.Char; Convert
$func? ToInt e.Char = s.Int; Convert
$func ToLower e.Char = e.Char; Convert
$func ToUpper e.Char = e.Char; Convert
$func ToWord e.Char = s.Word; Convert
$func Unbind s.Tab e.Key = ; Table

$func Vector e.Source = s.Vector; Vector
$func VectorFill s.Vector e.Fill = ; Vector
$func VectorInit s.Vector s.Len e.Fill = ; Vector
$func VectorLength s.Vector = s.Len; Vector
$func VectorRef s.Vector s.Index = e.Exp; Vector
$func VectorReplace s.Vector e.Source = ; Vector
$func VectorSet s.Vector s.Index e.Exp = ; Vector
$func VectorToExp s.Vector = e.Exp; Vector
$func Write e.Expr = ; StdIO
$func WriteCh s.Channel e.Expr = ; StdIO
$func WriteLn e.Expr = ; StdIO
$func WriteLnCh s.Channel e.Expr = ; StdIO

